

Fortran スマートプログラミング

(2014年度版)

摂南大学 理工学部 電気電子工学科

田口 俊弘

2014年7月28日

はじめに

パソコンが登場してから今日まで、計算機とそれを取り巻く環境の劇的な進歩には目を見張るばかりです。計算機の利用目的も多様になり、それぞれの目的に応じて様々なプログラミング言語が開発されました。しかし、数値計算やコンピュータシミュレーションにおいては、コンピュータ本来の能力である「高速計算」ができればいいので、それほど新しい言語は必要ありません。

現在、数値計算をするためによく使われているプログラミング言語には、C言語とFortranがあります。最近の大学ではC言語で計算機の講義を教えるところが増えてきて、Fortranは影が薄くなっているようですが、コンピュータシミュレーションの分野ではFortranの方がよく使われています。これは、Fortranが科学技術計算用言語として発展してきているので、数値計算に便利な書式が多数採用されているからです。例えば、複素数計算や行列計算などは文法に組み込まれているので簡単に書くことができます。数値計算向けのプログラムを書くにはFortranの方が書きやすいし、完成したプログラムをそのまま大型計算機で高速に実行させることも可能です。

Fortranは、計算機の歴史の初期に開発された由緒ある言語で、“FORmula TRANslation”が語源です。当初から数値計算用言語として開発されており、筆者が計算機を利用し始めた30年前には、数値計算をするといえば、ほとんどがFortranでした。しかし、科学技術計算に特化するということはユーザーの専門分野が限られるということであり、その結果コンパイラの値段があまり下がらなかったのがFortranをマイナーにした最大の要因だと思います。C言語は誕生時点から比較的安価に供給され、これが広く使われている理由の一つです。C言語はそもそもOSを記述するために開発された言語なので、計算機のハードウェア寄りの書式が多く、必ずしも初心者向きの言語だとは思えないのですが、プログラミングの専門家からすればFortranもC言語も大して変わらないので、安い方にシフトしたのはある意味やむを得ないことだと思います。

さて、1990年代に入って無料のOSであるLinuxやFreeBSDの開発が開始され、これが非常に勢いで普及しました。普及した最大の要因は、インターネットを通じて、世界中のアマチュアプログラマーがOSやその上で動作するアプリケーションの開発に参加したことでした。コンパイラも同様で、このフリーOSには、開発環境として有志が作成した無料のC言語(gcc)が付属し、さらにこれを利用した無料のFortran(g77, gfortranなど)も付属しています。このため、Linuxなどをインストールすれば、パソコンでも無料でFortranが使えます。また、これらフリーのFortranはWindowsやMacOS上で動作するものも作られているので、価格でFortranを敬遠する理由はなくなったと言えます。

本書は、数値計算やコンピュータシミュレーションを目的として、これから計算機プログラムの勉強を始めようとする人を対象にしたFortranプログラム作成法の解説書です。単に文法だけを説明するのではなく、コンピュータの構造の話なども交えながら、効率が良くてエラーを見つけやすい、“スマートなFortranプログラム”の書き方を説明します。

長い歴史を持っているFortranは、古いバージョンとの互換性を保つ必要性から、若干緩い文法体系になっています。例えば、デフォルトでは変数宣言をしなくてもかまわないため、変数名を書き間違えたときに発見するのが難しく、思わぬエラーを引き起こす可能性があります。また、コンピュータシミュレーションのような長いプログラムを書くときには、プログラムのチェックや修正がしやすいように書いておかないと、メンテナンスに苦労します。このため、多少冗長になっても、読みやすく、後でデバッグしやすいように書くべきです。本書では、これを念頭に置いて説明をしています。

もちろん、数値計算やコンピュータシミュレーションにとっては計算速度が命ですから、できる限り計算が速くなるようなプログラムにすることも必要です。プログラムとはコンピュータでの計算手順を記述するものなので、コンピュータの構造を考慮してちょっと計算順序を変えるだけでも高速化できることがあります。本書では、このようなコンピュータの内部構造で知っておいた方が良い部分も簡単に説明しました。孫子曰く、「彼を知り己を知れば百戦殆うからず」と。

Fortranは古くからある言語なのですが、1991年に策定されたFortran90を境に大きく変貌しまし

た [1]. 配列演算などの新しい機能を導入すると同時に、現代的な書式で書けるようになったのです。それまで 1 行に 72 文字までしか書けなかったという制限は無くなったし、1 行に複数の文を書くこともできるようになりました。Fortran90 以降も何度か改版されて、現在の最新規格は Fortran2008 ですが、数値計算だけでなく、オブジェクト指向プログラミングなども可能になっています。しかし、数値計算だけならそのような新しい概念は必要ないので、本書では Fortran90 レベルの文法で説明をしました。このレベルでプログラムを書いておけばフリーの Fortran でコンパイルできるというメリットもあります。

本書には 6 章ありますが、基本的なプログラミングに関する知識は第 4 章までで十分です。そこで、第 4 章までには演習問題を付けています。ただし、第 3 章までの演習問題は各章の内容に合わせた問題になっていますが、第 4 章の演習問題はそれまでのプログラミング知識全般を応用して実用的なプログラムを書くための練習になるような問題になっています。また、第 3 章までの章末に「スマートテクニック」と称して、プログラムを書くときの心構え、高速化のこつ、知っておくと便利な文法について紹介しています。これらも合わせて利用すれば、より質の高いプログラムを書くことができると思います。

これに対し、第 5 章と第 6 章はある程度プログラミングに慣れてきた人向けです。初心者の方は、まず第 4 章までの内容を使って色々なプログラムを書き、プログラミングというものを良く理解して下さい。第 5 章以降は、ある程度の経験を積んでから必要に応じて読まれれば良いと思います。また、実際にパソコンで Fortran プログラムを動作させる方法や、その際の注意事項などをまとめて付録にしています。こちらにも必要に応じて活用して下さい。

本書は、筆者が摂南大学の卒業研究生向けに書いた Fortran のテキストがベースになっています。これを、大阪大学レーザーエネルギー学研究センターの福田優子氏が Web で公開できる Fortran テキストを探しているというので提供したところ、結構気に入ってくれて、学生に配ったり、他の計算機センターに紹介してくれました。その後、紹介先の一つである東北大学サイバーサイエンスセンターから広報誌 SENAC へテキストの内容を投稿してくれないかとの依頼があったので、加筆・修正をして 2012 年度に 3 回の連載として掲載されました。ただ、3 回の連載ではページ数が限られていたので演習問題をカットせざるを得ず、卒研究生を教育するために配る資料としてはもの足りないものになってしまいました。本書は、SENAC の原稿を再編成し、説明が不十分の箇所を補い、演習問題を加えてまとめ直したものです。

最後に、筆者が適当に書いたテキストをこのようなまとめた文書にまとめ直すきっかけを与えてくれた福田優子氏に深い感謝の意を表します。また、連載するに当たって、つたない文章を熱心に読んで多数の有益なご意見を下さった東北大学サイバーメディアセンターの方々にも深く感謝致します。

目 次

第 1 章 Fortran プログラミングの基本	1
1.1 メインプログラムの開始と終了	1
1.2 基本的な実行文の書き方	3
1.2.1 代入文と演算の書式	3
1.2.2 数値の型	4
1.2.3 変数の宣言	6
1.2.4 組み込み関数	7
1.2.5 print 文による簡易出力	9
1.3 配列	9
1.3.1 配列宣言	9
1.3.2 メモリ上での配列の並び	10
1.4 プログラミング スマートテクニック (その 1)	11
1.4.1 コンピュータで表現可能な数値の大きさ	11
1.4.2 演算の速度を考える	12
1.4.3 桁落ちに気を付ける	13
1.4.4 π の作り方	14
1.4.5 コメント文の挿入	14
1.4.6 継続行と複文	15
演習問題 1	16
(1-1) 2 次方程式の根	16
(1-2) ヘロンの公式	16
(1-3) 面積, 体積	16
(1-4) ビオ・サバルの法則	16
(1-5) 相対論的エネルギー	17
(1-6) 複素インピーダンス	17
(1-7) 回路計算	17
(1-8) 3 次元ベクトルと電磁気学	17
(1-9) 2 行 2 列の行列計算	17
第 2 章 手順の繰り返しと条件分岐	18
2.1 手順の繰り返し — do 文	18
2.2 条件分岐 — if 文	20
2.3 無条件ジャンプ — goto 文, exit 文, cycle 文	22
2.4 プログラミング スマートテクニック (その 2)	25
2.4.1 do ループのテクニック	25
2.4.2 多項式を計算する手法	27
2.4.3 do while 文と無条件 do 文	27
2.4.4 0.1 を 10 回足しても 1 に等しくない	28
2.4.5 ブロックを明確にするために字下げする	29
2.4.6 文字間や行間は適当に空ける	29
2.4.7 定数は意味のある名前をつけた変数に代入して使う	30
演習問題 2	32
(2-1) 繰り返し出力	32
(2-2) 統計計算	32

(2-3) 定積分	32
(2-4) 漸化式	32
(2-5) horner 法による多項式計算とテーラー展開	33
(2-6) 2 次方程式の根 (根の判別付き)	33
(2-7) 非線形方程式の解法：逐次代入法	33
(2-8) 非線形方程式の解法：Newton 法	34
(2-9) 常微分方程式 (初期値問題) の解法：サイクロトロン運動	34
(2-10) 常微分方程式 (境界値問題) の解法：1 次元ポワソン方程式	35
第 3 章 サブルーチン	36
3.1 サブルーチンの利用目的	36
3.2 サブルーチンの宣言と呼び出し	37
3.3 ローカル変数と引数	39
3.4 間接アドレスを用いたルーチン間におけるデータの受け渡し	41
3.5 配列を引数にする場合	44
3.6 関数副プログラム	46
3.7 モジュールを使ったグローバル変数の利用	47
3.8 サブルーチンや関数副プログラムを引数にする	49
3.9 プログラミング スマートテクニック (その 3)	51
3.9.1 拡張宣言文とそれを用いたローカル変数の使い分け	51
3.9.2 parameter 変数の利用	53
3.9.3 include 文	55
3.9.4 interface 文	56
演習問題 3	58
(3-1) 2 次方程式の根 (サブルーチン利用)	58
(3-2) ヘロンの公式 (サブルーチン利用)	58
(3-3) 回路計算 (サブルーチン利用)	58
(3-4) 3 次元ベクトルと電磁気学 (サブルーチン利用)	58
(3-5) 統計計算 (サブルーチン利用)	58
(3-6) 非線形方程式の解法：Newton 法 (サブルーチン利用)	59
(3-7) 行列計算	59
(3-8) 行列式	59
(3-9) 3 元連立方程式の解法	59
(3-10) モンテカルロ法	60
第 4 章 データ出力の詳細とデータ入力	61
4.1 データ出力先の指定	61
4.2 配列の出力, do 型出力	61
4.3 format による出力形式の指定	63
4.4 データの入力方法	67
4.4.1 入力文の一般型	67
4.4.2 入力時のエラー処理	69
4.4.3 ネームリストを用いた入力	69
4.5 書式なし入出力文によるバイナリ形式の利用	71
4.6 ファイルのオープンとクローズ	73
演習問題 4	75

(4-1) 連立常微分方程式：2 次の Runge-Kutta 法	75
(4-2) 連立常微分方程式：4 次の Runge-Kutta 法	75
(4-3) 楕円型偏微分方程式	76
(4-4) 放物型偏微分方程式：陽解差分法	77
(4-5) 放物型偏微分方程式：陰解差分法	77
(4-6) 分子動力学	78
(4-7) 1次元移流方程式	79
(4-8) 粒子密度 - 分布関数	80
(4-9) ブラウン運動 - 拡散	80
第 5 章 文字列の活用	81
5.1 文字列定数と文字列変数	81
5.2 部分文字列と文字列演算	82
5.3 出力における文字列の利用	84
5.4 数値・文字列変換	85
5.5 文字列に関する組み込み関数	86
第 6 章 配列計算式	88
6.1 基本的な配列計算式	88
6.2 部分配列	90
6.3 where 文による条件分岐	92
6.4 配列定数	93
6.5 配列に関する組み込み関数	95
6.6 配列の動的割り付け	97
付録 A gfortran を用いたコンパイルから実行までの手順	100
付録 B エラー・バグへの対処法	102
付録 C 自動倍精度化オプション	105
付録 D 数値型の精度指定	106
付録 E Big Endian と Little Endian	108
付録 F 数値計算を補助するフリーのサブルーチンライブラリ	109
F.1 高速フーリエ変換	109
F.2 多倍長計算	109
F.3 長周期乱数発生	109
F.4 数値計算ライブラリ	110
F.5 グラフィックライブラリ	110
参考文献	111

第1章 Fortran プログラミングの基本

計算機プログラム(あるいは、単にプログラム)とはコンピュータへの動作指示(命令)を記述した文の集まりのことです。コンピュータを動作させる本来の命令は“機械語”と呼ばれる数値で表されていますが、これで直接プログラムを書くことはまず不可能です。そこで、人間が理解しやすいように、単語や記号を使った書式で計算機の動作を書けるようにしたものが、Fortranのような“プログラミング言語”です。プログラミング言語を使ってプログラムを書いておけば、これを“コンパイラ”という機械語への翻訳ソフトを使って、コンピュータで実行可能な機械語(実行形式)に変換することができます。

しかし、“コンパイラ”は翻訳機に過ぎません。プログラミング言語という、我々には理解しやすい書式で書けるといっても、あくまでもコンピュータを動作させる手順になるように書かねばならないのです。本章では、Fortranにおける基本的なプログラムの書式とデータの取り扱いについて説明します。

1.1 メインプログラムの開始と終了

実行形式に変換されたプログラムを起動すると、コンピュータは、

```
実行開始処理 → プログラムに記述された命令を順に実行 → 実行終了処理
```

という流れで動作します。実行を開始した時に最初に実行する部分を“メインプログラム”といいます。言わばプログラムの本体です。プログラムには、他のプログラムの中から実行開始を指示してその機能を利用する“サブルーチン”がありますが、これについては第3章で説明します。

Fortranでは特別な手続きをせずにプログラムを書くと、それがメインプログラムと仮定されます。しかし、それではサブルーチンと区別しにくいので、program文を用いて最初にプログラムの名前を書きます。program文は以下の形式です。

```
program プログラム名
```

これに対し、メインプログラムの終了はend program文で指定します。

```
end program プログラム名
```

ここで、program文とend program文で指定する“プログラム名”は同じでなければなりません。このため、メインプログラムは次のような構造になります。

```
program code_name
  .....
  .....
  .....
end program code_name
```

この例のようにプログラム名にはアンダースコア(_)も使えるので、これをスペースの代わりに使えば単語をつないだ長いプログラム名を付けることもできます。

program文とend program文の間にFortranの文法に従った文を並べて、動作させたい計算手順を記述します。動作手順を記述する文を“実行文”といいます。しかし、プログラムに記述するのは実行文だけではありません。計算途中で必要となる変数領域を確保するための宣言文も書かねばなりません。このような計算動作に直接携わらない文を“非実行文”といいます。Fortranでは、非実行文をプログラムの最初に集約して、実行文はその後に書きます。このため、動作の開始はprogram文の次の文ではなく、最初の実行文になります。

完結したメインプログラムの一例を示します¹。

¹この例のように、プログラム内部の文は、スペースを数個入れて、program文とend program文よりも少し右にずらして書きます。そうすれば、メインプログラムの範囲が明確になります。


```

program test_code
  implicit none
  real x,y,z
  x = 5
  y = 100
  z = x + y*100
  print *,x,y
  print *,'z =',z
end program test_code

```

このプログラムにおける実行文・非実行文の区分と、動作開始から終了までの実行の流れを図 1.1 に示します。実行形式のプログラムを起動すると、最初の実行文から動作が開始し、上から順に一行ずつ実行して、end program 文に到達した段階で動作終了となります。第 2 章で述べる do 文を使った繰り返しや、if 文による条件分岐など、動作指定によっては所定の位置にバックしたり、条件に応じて実行するかどうかの選択ができますが、それでもそれぞれの文が終了した後に次の文が実行されるという基本的動作は変わりません。

これは計算機が 1 回に一つの動作しかできないためです。プログラム全体を見渡して実行するのではなく、1 行ずつ順に実行していくので、バックするような動作指定をしない限り、下方に書いた実行文は上方に影響を及ぼしません。プログラムはこのことを常に念頭に置いて書かなければなりません。




図 1.1 プログラム開始から終了までの流れ

なお、2.2 節で述べる if 文を使って、条件によって途中でプログラムを終了するときや、第 3 章で説明するサブルーチン内でプログラムを終了するときには stop 文を用います。stop 文を実行すると、その時点でプログラムが終了します。例えば、

```

program fluid_code
  implicit none
  real x,y
  integer m,n
  .....
  if (m < 0) stop
  .....
end program fluid_code

```

と書けば、 $m < 0$ の時にプログラムは終了し、それ以降は実行しません。

1.2 基本的な実行文の書き方

数値計算における基本的要素は“定数”と“変数”で、実行文はこれらを使って数学的に記述します。ただし、あくまでもコンピュータを実行させるための記述ですから、数学における表現とは異なる意味を持つこともあります。そのあたりの区別が理解できれば、本節の知識だけでもプログラム電卓的な計算をさせることが可能です。

1.2.1 代入文と演算の書式

実行文には、stop文のように常に同じ動作を指示する文と、いくつかの“数値”を伴って、その数値に応じた動作を指示する文とがあります。プログラム中で“数値”を与える方法は3種類あります。-500とか3.14のような数字を直接書く“定数”，xとかyzのような単語で示した“変数”，およびそれらを使ってx+3やsin(y-5)のような計算手順を表した“計算式”です。計算式に書かれている手順も計算機の動作なのですが、プログラム中の計算式は、その計算結果を動作命令に与える“数値”として位置づけられています。このため、計算式を書いただけでは実行文になりません。

プログラムで最も良く使う実行文は“代入文”です。これは、

```
変数 = 計算式
```

という形をしています。計算式の代わりに、定数や変数を書くこともできます。プログラムにおける“変数”とはコンピュータのメモリ領域のことで、数値を入れる箱だと考えればいいでしょう。代入文は右辺の“数値”を左辺で示した変数メモリに格納する動作を表します。よって、

```
計算式 = 計算式      !   これはエラー
```

という形は使えません。Fortranにおいて“=”という記号は、“左辺と右辺が等しい”という意味ではないからです。例えば、

```
x = 4 + 2
y = 9 - x
y = y + 1
```

のようなプログラムを考えてみましょう。プログラムは上から順に実行されるので、まず4+2の結果である6が変数xに代入されます。次に、9から変数xに保存されている数値を引いた値が変数yに代入されるので、yには3が代入されます。最後の文は数学の等式と見れば変ですが、イコールが“代入する動作”だと理解できれば特に不思議な文ではありません。yはその前の文で3が代入されていますから、最後の文によって変数yにy+1の結果である4が代入されます。

Fortranにおける基本演算の書き方と使い方を表1.1に示します。

表 1.1 演算記号の書き方と使い方

演算記号	演算の意味	使用例	使用例の意味
+	足し算	x+y	$x + y$
-	引き算	x-y	$x - y$
*	掛け算	x*y	$x \times y$
/	割り算	x/y	$x \div y$
**	べき乗	x**y	x^y
-	マイナス	-x	$-x$

べき乗までの二項演算には以下のような優先順位があり、優先順位の高い方が低い方より先に計算されます。これは数学における演算順序と同じです。

べき乗 > 掛け算または割り算 > 足し算または引き算

掛け算と割り算のような同レベルの演算は左から順に計算されます。さらに、かっこを使えばかっこの中が優先的に計算されます。例えば、

$$f = x + (y-3)/z**2$$

という文では、一番最初に $y-3$ を計算し、次に z の 2 乗を計算し、次に $y-3$ の結果を z の 2 乗の結果で割り、その結果を x に加えて、最後に f に代入する、という動作になります。

なお、時々見かける間違いに、

$$f = x + y/ab$$

という数式を次のようなプログラムにするものがあります。

$$f = x + y/a*b$$

数学では、 y/ab の記述を $\frac{y}{ab}$ と解釈することが多いのですが、プログラムでは掛け算と割り算が同じレベルなので、 $y/a*b$ は“ y を a で割って、その結果に b を掛ける”です。よって、数学的解釈になるようにするには、

$$f = x + y/(a*b)$$

と書かなければなりません。

1.2.2 数値の型

コンピュータという機械が取り扱う数値は 2 進数で表されていて、基本的には整数です。例えば、1byte の数は、8bit、つまり 0 か 1 のどちらかを選択できる数が 8 個並んだもので、10 進数では $0 \sim 255$ ($=2^8-1$) までの整数を表すことができます²。Fortran では小数点のない数字、56 とか -1112 などの数字を“整数型”といいます。Fortran の整数型数は 4byte (32bit) で表現されていて、 $-2^{31} \sim 2^{31} - 1$ の整数を扱うことができます。

しかし、数値計算やシミュレーションをする場合には、3.14 のように小数点以下を含んだり、 3×10^{19} とか、 1.6×10^{-27} とかいった様々なスケールの数値を取り扱うため、整数型だけでは不便です。特に、整数型数は小数点以下の表現ができないので、割り算をすると全て切り捨てになります。これを忘れてプログラムを書くと、よく間違いを起こします。例えば、

$$\begin{aligned}x &= (5/2)**2 \\y &= x**(3/2) \\t &= 2/3*y*x\end{aligned}$$

と書くと、 x 、 y 、 z はいくつになると思いますか？ 答えは $x=4$ 、 $y=4$ 、 $t=0$ です。これは整数の割り算が切り捨てになるため、 $5/2 \rightarrow 2$ 、 $3/2 \rightarrow 1$ 、 $2/3 \rightarrow 0$ になるからです。

そこで、整数型の他に、3.14 のような少数点以下を含んだり、 1.6×10^{-27} のような、 $A \times 10^B$ という形式の数値を取り扱うことができます。これを“実数型”といい、 A の部分を“仮数部”、 B の部分を“指数部”といいます。通常、数値計算は実数型で計算しなければなりません。

実数型には有効数字の違う 2 種類が用意されています。単精度実数と倍精度実数です²。単精度実数とは 4byte で表される実数のことで、仮数部の有効数字は 7 桁程度です。これに対し、倍精度実数とは 8byte (64bit) で表される実数のことで、仮数部の有効数字は 15 桁程度です。7 桁あれば問題ないと思われるかもしれませんが、何百万個もの数字の合計を計算したり、次数の高い多項式の計算をするときな

²bit や byte の詳細と実数型の表現については 1.4.1 節参照。

どでは、計算回数の増加につれて有効桁数が減少することを考慮しなければなりません。このため、通常は倍精度実数で計算をします。

Fortran における基本的な実数型は単精度であり、倍精度実数型を使用する時にはそれを意識した書式で書かなければなりません。しかし、最近のコンパイラはオプションを指定すればデフォルトの実数型を倍精度にする“自動倍精度化機能”を持っているので、本書では、単精度実数と倍精度実数を使い分けることはせず、単に“実数型”と表現します³。よって、特に単精度で計算する必要がなければ、コンパイル時に自動倍精度化オプションを付加して倍精度で計算して下さい。多くの計算機は倍精度実数計算を高速で行えるハードウェアを内蔵しているので、倍精度計算をしてもさほど実行速度は下がりません。

プログラム上で、整数型の定数と実数型の定数は小数点の有無で区別します。例えば、100 とか、-12345 と書けば整数型ですが、100.0 とか、-12345. とか、-0.0314 とか書けば実数型になります。また、 1.6×10^{23} を入力したい時には、1.6e23 と書きます。すなわち、 $A \times 10^B$ は AeB と書きます。B が負の場合でも 1.6e-19 のように e の後に続けて書きます。また、6e20 のように eB を付加した場合には小数点が無くても実数型になります。例えば、

$$a = 3.141592r^2 + 3x^5 + 6.5 \times 10^{-5}x - 10^5$$

という式をプログラムで書けば以下のようになります。

```
a = 3.141592*r**2 + 3.0*x**5 + 6.5e-5*x - 1e5
```

この例のように、r**2 とか x**5 のように整数べき乗の指数 (2 とか 5) には整数型を使います。

先ほど整数型を使ったら期待どおりの結果が出ない例を挙げましたが、以下のように実数型を使えば正しい結果が得られます。

```
x = (5.0/2.0)**2
y = x**(3.0/2.0)
y = 2.0/3.0*y*x
```

Fortran の便利な機能の一つは、複素数が使えることです。複素数にも単精度複素数型と倍精度複素数型がありますが、自動倍精度化機能を使えばデフォルトが倍精度複素数形になるので、本書では単に“複素数型”と表現します。複素数型の定数は、

```
(0.0,1.0)
(1e-5,-5.2e3)
(-3200.0,0.005)
```

のように、2 個の実数をコンマでつないで、かっこで囲みます。前半が実部、後半が虚部です。つまりこの例は、それぞれ、 i 、 $10^{-5} - 5.2 \times 10^3 i$ 、 $-3200 + 0.005 i$ を表した複素数型定数です。

なお、ここまで読むと整数型は必要がないと思われるかもしれませんが、そうではありません。1.3 節で説明する配列の要素を指定する数や、2.1 節で説明する do 文のカウンタ変数は整数型でなければなりません。また、オン・オフを表すだけの変数を作るときにも整数型を使います。すなわち、整数型数は“順番”や“区別”を表すときに不可欠です。また実数の整数部を取り出したいときや、剰余 (あまり) を計算するときも整数型を利用します。数値計算プログラムの中でも整数型の使い道は多いのです。

計算式中に異なる数値型が混在する時は、精度の高い方の型に合わせて計算し、その型の値が結果になります。例えば、実数型と整数型の計算は整数型の数値を実数型に変換して実数型と実数型の計算を行い、実数型の結果になります。複素数型と実数型の計算の場合にはその実数型の数値を実部とした複素数型にして複素数計算をし、結果は複素数型になります。ただし、掛け算と割り算の計算は左から順に行うので注意が必要です。最初の整数型を使った計算例で、

³自動倍精度化については付録 C 参照。精度の変更については付録 D で説明しています。

```
t = 2/3*y*x
```

が0になるのは、最初に計算されるのが2/3という整数型÷整数型だからです。これを、

```
t = y*x*2/3
```

と書けば、切り捨ては起こりません。

1.2.3 変数の宣言

次に、数式の計算結果を保存する変数の使い方を説明します。変数の名前は、頭文字がa~zのどれかであれば、後はa~z, 0~9をどのような順序で並べたものでも使えます。例えば、abcとかk10xy等です。Fortranでは大文字と小文字を区別しないため、abcとABCは同じ変数になります。変数にも型があり、計算結果に応じた型の変数を使わなければ正確に保存することができません。この変数の型を決める文を“宣言文”といいます。型を決めると同時に、変数のメモリ領域を確保します。このため、計算に用いる変数は全て宣言しなければなりません。宣言は一度しかできず、プログラムの実行時に変更することはできません。宣言文は非実行文です。

ところがFortranには“暗黙の宣言”があり、通常は宣言しなくても文法的な間違いにならないので、タイプミスなどで思わぬエラーが発生する可能性があります。これを防ぐため、プログラムの2行目、すなわちprogram文の次の行に必ず以下のimplicit文を書いておきます。

```
implicit none
```

この文を入れておけば、宣言せずに使用した変数があるとコンパイルエラーになり、タイプミスのチェックができます。

数値計算は基本的に実数で行いますが、実数型変数は次のように宣言します。

```
real 変数 1, 変数 2, ...
```

これに対し、整数型の変数は、次のように宣言します。

```
integer 変数 1, 変数 2, ...
```

宣言文は非実行文なので、全ての実行文より前に書かなければなりません。例えば、メインプログラムの始まりは以下のようになります。

```
program test1
  implicit none
  real x,y,z,omega,wave,area
  integer i,k,n,imin,imax,kmax
  .....
```

realやinteger等の型宣言文は何行書いても良いし、順番も無関係です。

Fortranの暗黙宣言では、名前の頭文字がa~hかo~zの変数は実数型で、i~nの変数は整数型でした。このため、整数型変数の頭文字をi~nにする習慣があります。全てを宣言する以上、基本的に変数名の付け方に制限はないのですが、整数型は用途が限られているので、頭文字を限定する方が良いと思います。実数型数の名前はあまり頭文字にこだわる必要はありませんが、原則として整数型をイメージするi,j,k,l,m,nの1文字変数は使わない方が無難です。

複素数型変数の宣言文は、

```
complex 変数 1, 変数 2, ...
```

です。複素数型も用途が限定されているので名前の付け方に規則をつける方が良いでしょう。複素数型変数名は、頭文字をcかzにすることが多いようです。

プログラムにおいて、数値を変数に保存する意義は大別して二つあります。一つはプログラム全域にわたって共通してその値を利用するためであり、もう一つは狭い範囲で計算結果を一時的に保存するためです。この二つは意識して使い分けるようにします。特に、前者の大域的に利用する変数には長めで意味のある名前を付けるべきです。これは、1文字のような短い変数名を使うと、プログラムが長くなるにつれて、どこでその変数を使っているかを検索するのが難しくなるからです。

変数に数値を代入する時は、右辺の計算結果を左辺の変数の型に変換して代入します。このため、実数の計算結果を整数型の変数に代入すると、小数点以下は切り捨てられます。例えば、

```
real x
integer n
x = 3.14
n = x + 6
```

の結果、`n` に代入される値は9です。

また、複素数型の計算結果を実数型の変数に代入すると、その複素数の実部が代入されます。例えば、

```
real x
complex c
c=(1.0,-2.0)
x=c**2
```

とすると、`x` = -3.0 になります。逆に、複素数型の変数に実数型の数値を代入すると、実部に結果が代入され、虚部は0になります。例えば、

```
real x
complex c
x=5
c=x**2
```

とすると、`c` = (25.0,0.0) になります。

1.2.4 組み込み関数

数値計算上よく使う数学関数はあらかじめ用意されています。この用意された関数を“組み込み関数”といいます。表 1.2 に、代表的な組み込み数学関数を示します。

表 1.2 数学関数の書き方と使い方

組み込み関数	名称	数学的表現	必要条件	関数値 f の範囲
<code>sqrt(x)</code>	平方根*	\sqrt{x}	$x \geq 0$	
<code>sin(x)</code>	正弦関数*	$\sin x$		
<code>cos(x)</code>	余弦関数*	$\cos x$		
<code>tan(x)</code>	正接関数	$\tan x$		
<code>asin(x)</code>	逆正弦関数	$\sin^{-1} x$	$-1 \leq x \leq 1$	$-\frac{\pi}{2} \leq f \leq \frac{\pi}{2}$
<code>acos(x)</code>	逆余弦関数	$\cos^{-1} x$	$-1 \leq x \leq 1$	$0 \leq f \leq \pi$
<code>atan(x)</code>	逆正接関数	$\tan^{-1} x$		$-\frac{\pi}{2} < f < \frac{\pi}{2}$
<code>atan2(y,x)</code>	逆正接関数 ⁴	$\tan^{-1}(y/x)$		$-\pi < f \leq \pi$
<code>exp(x)</code>	指数関数*	e^x		
<code>log(x)</code>	自然対数*	$\log_e x$	$x > 0$	
<code>log10(x)</code>	常用対数	$\log_{10} x$	$x > 0$	

⁴逆正接関数 `atan2(y,x)` は、座標点 (x,y) の偏角を計算する関数です (x と y の順序に注意)。よって、表の数学的表現には“ $\tan^{-1}(y/x)$ ”と記述してありますが、実際には x と y の値に応じて $-\pi$ から π の間の角度が結果になります。

関数名の後のかっちは必ず必要です。かっこの中の x や y を“引数(ひきすう)”といいます。関数を計算式中に記述すると、引数に対する関数値が結果となってその計算式を実行します。例えば、

```
c = exp(-x**2) + sin(10*x+3) - 2*tan(-2*log(x))**3
```

のように書くことができます。この例のように関数の引数に関数を使った計算式を与えることも可能です。

表 1.2 の数学関数は、引数に実数型数を与えると実数型の結果になる“実数型関数”ですが、名称に“*”の付いている関数は、引数が複素数型でも使えます。Fortran の組み込み関数には、引数の型や精度に応じて計算をする“総称名”機能があるので、複素数を引数に与えた場合の関数値は複素数型の結果になります⁵。

ただし、総称名機能では引数の数値型に対応した関数が用意されていないとコンパイルエラーになります。例えば、 $\sqrt{2}$ を計算したくて、`sqrt(2)` と書くとエラーです。“2”は整数型の定数であり、整数型数の平方根は用意されていないからです。`sqrt(2.0)` のように実数型を引数に書かなければなりません。

この他、絶対値や余りを計算するのも組み込み関数を使うし、実数型を複素数型に変換するなどの型変換も組み込み関数を使って処理します。その中の代表的なものを表 1.3 に示します。

表 1.3 型変換関数などの組み込み関数

組み込み関数	名称	引数の数値型	関数値の数値型	関数の意味
<code>real(n)</code>	実数化	整数	実数	実数型へ変換する
<code>abs(n)</code>	絶対値	整数	整数	n の絶対値
<code>mod(m,n)</code>	剰余 ⁶	2 個の整数	整数	m を n で割った余り
<code>int(x)</code>	整数化	実数	整数	整数型に変換する (切り捨て)
<code>nint(x)</code>	整数化	実数	整数	整数型に変換する (四捨五入)
<code>sign(x,s)</code>	符号の変更	実数	実数	$s \geq 0$ なら $ x $, さもなくば $- x $
<code>real(z)</code>	複素数の実部	複素数	実数	z の実部
<code>imag(z)</code>	複素数の虚部	複素数	実数	z の虚部
<code>abs(z)</code>	絶対値	実数または複素数	実数	z の絶対値
<code>cmplx(x,y)</code>	複素数化	2 個の実数	複素数	$x + iy$
<code>conjg(z)</code>	共役複素数	複素数	複素数	z の共役複素数

型変換関数は、数値型が限定されている場所に、その数値型以外の値を指定したいときなどに使います。例えば、整数型の変数 n の平方根を計算したいときには、上記のように `sqrt(n)` と書くことはできません。このときは、`sqrt(real(n))` と書きます。また、整数化は切り捨て演算を含んでいるので、それを積極的に利用するときにも使います。例えば、正の実数 x の小数点以下を四捨五入した整数は、`int(x+0.5)` で計算することができます。

複素数型の定数は、 $(1.0, -2.0)$ のような表現で指定することができますが、変数や計算式で実部と虚部を指定した複素数を作る時には関数 `cmplx` を使います。例えば、

```
complex z,z1
real x,y
x = 10.0
y = -3.5
z = cmplx(x,y)
z1 = cmplx(x**2,y+2.0)
```

⁵表 1.2 の必要条件と関数値の範囲は引数を実数型の場合です。この必要条件を満たさない実数型数を与えると実行時エラーになります。しかし複素数型を与えた場合には必ずしもエラーにはなりません。

⁶剰余を計算する関数 `mod` の利用例として繰り返しの制御があります。2.2 節を参照して下さい。

のように書きます。この場合、 $z = x + iy$ であり、 $z1 = x^2 + i(y + 2)$ になります。

1.2.5 print 文による簡易出力

数値計算を目的としたプログラムでは、得られた計算結果を表示したりファイルに保存する必要があります。入出力の詳細については第4章で説明しますが、最低限、`print` 文を使った標準形式による数値出力は覚えておいて下さい。`print` 文の一例を以下に示します。

```
integer n
n = 3
print *, 4+5, n, n*2, 2*n-11
```

このように、“`print *`,”に続いて変数や計算式をコンマで区切って並べると、それらの計算結果が横に並んで出力されます。上例の場合には、`4+5`の結果である9から`2*n-11`の結果である-5までが以下のように出力されます。

```
          9          3          6          -5
```

なお、“`print *`,”の“`*`”は、標準形式で出力することを意味しているのですが、とりあえずは形式的に書くものと覚えて下さい⁷。

複数の数値を出力するときに数字だけを出力すると、どれがどの数値かわからなくなる可能性があります。このような場合は、文字列を併用して変数の意味を同時に出力します。Fortranにおける“文字列”とは、2個のアポストロフィ「`'`」ではさんだ文字の並びのことで、`print` 文中で数値といっしょに並べると、その文字の並びがそのまま出力されます。例えば、

```
real x
x = 3
print *, 'x = ', x, ' x**3 = ', x**3
```

というプログラムの出力は、

```
x =      3.0000000000000000      x**3 =      27.000000000000000
```

となります。

1.3 配列

ここまで出てきた変数は型に応じた1個の数値を記憶することしかできませんでした。しかし、数値計算やシミュレーションでは、数万個のデータを保存して、それぞれを条件に応じて変化させていく、なんていうのが当たり前のように行われます。そこで、数学で a_1, a_2, \dots, a_n のように変数に添字を付けて区別するように、数字で区別した変数を作ることができます。これを“配列”といいます。本節では配列の使い方について説明します。

1.3.1 配列宣言

配列は変数の一種なので、型宣言文を使って宣言しておかねばなりません。単一変数と異なるのは、宣言時に添字の範囲を示す整数値をカッコを使って付加することです。例えば、次のように宣言します。

```
real a(10), b(20,30)
complex cint(10,10)
integer node(100)
```

⁷出力形式の指定方法は4.3節参照。

ここで、aやnodeのように数字が1個の配列を1次元配列、bやcintのように数字が2個の配列を2次元配列といいます。3次元以上の配列を作ることもできます。配列の名前の付け方、頭文字の選択などの規則は単一変数名の付け方と同じにします。宣言した配列の各部の名称を表1.4に示します。

表 1.4 配列の各部の名称

real a(10) と宣言した配列について	
記号	名称
a	配列名
a(3) など	配列要素
かっこ中の数字	要素番号, 添字

配列宣言で指定した数値は要素番号の最大値を表し、要素番号の使用可能範囲は1から指定した最大値までです。例えばa(10)の宣言ではa(1)からa(10)までの10個の配列要素が使用可能になります。また、2次元以上の配列の場合には各次元ごとの最大値を指定しているので、b(20,30)の宣言では全部で $20 \times 30 = 600$ 個の配列要素が使用可能になります。

問題によっては、要素番号として0や負数を使いたい場合があります。このような時には“:”を間にに入れて、使用可能な要素番号の最小値と最大値を同時に指定します。例えば、

```
real ac(-3:5),bc(-20:20,0:100)
```

と宣言すると、1次元配列acは、ac(-3)からac(5)までの9個が使用可能であり、2次元配列bcは、bc(-20,0)からbc(20,100)までの $(20 \times 2 + 1) \times (100 + 1) = 4141$ 個が使用可能です。

1.3.2 メモリ上での配列の並び

配列はコンピュータ内部において連続したメモリ領域で実現されています。例えば、real a(10)と宣言された配列は、図1.2左のように、a(1),a(2),...,a(10)の順で並んだ実数型のメモリです。




図 1.2 1次元配列のメモリ並び

図からわかるように、a(3)とはa(1)から数えて3番目のメモリのことです。また、real a(10)の宣言では10個しかメモリを確保していないのですから、a(10000)のように範囲外の要素番号を指定すると問題が起こります。10000番目の要素a(10000)がどのメモリを示すのか不明だし、そもそも存在するという保証さえありません。

Fortranでの配列名は配列を代表する名称であると同時に、配列の先頭メモリを示します。たとえば、配列名aは図1.2左のようにa(1)を示します。また、ac(-3:5)のように最小値も指定して宣言した場

合には、図 1.2 右のように並んでいて、配列名 ac は ac(-3) を示します。配列名が先頭要素を示すことは配列をサブルーチンの引数として与える時に意識する必要があります。

2次元以上の配列の場合は、左の方の要素番号から先に進むようにメモリ上で並んでいます。例えば、

```
real b(3,2)
```

と宣言した場合、メモリ上での並びは図 1.3 のようになります。2次元以上の配列も配列名は先頭要素を示しています。




図 1.3 2次元配列のメモリ並び

2次元と3次元の配列の並び方を式で表せば、

```
real b(m,n) の配列宣言で、b(i,j) は先頭から数えて i + m*(j-1) 番目
real b(l,m,n) の配列宣言で、b(i,j,k) は先頭から数えて i + l*(j-1 + m*(k-1)) 番目
```

となります。どちらの公式も一番右側の要素数 n に依存していません。一般的に、どんな次元の配列の順番を与える公式も一番右側の要素数には依存しません。

1.4 プログラミング スマートテクニック (その1)

計算手順が複雑になったり大量のデータを処理するようになると、数式を単純にプログラムに置きかえるのではなく、計算機の特徴を考慮したプログラムにすることで計算効率や精度を高めることができます。ここでは、計算速度を向上させるテクニックや、コメントを入れてプログラムのメンテナンスを容易にする手法などについて説明します。

1.4.1 コンピュータで表現可能な数値の大きさ

我々が日常的に使っている 10 進数は、10 のべき乗が基本です。すなわち、数字を 10 の多項式で表したときの係数を次数の大きい方から並べたものです。ただし、係数は 10 より小さい整数 (0~9) でなければなりません。例えば、1203 という数は、10 の多項式で表せば、

$$1203 = \underline{1} \times 10^3 + \underline{2} \times 10^2 + \underline{0} \times 10^1 + \underline{3} \times 10^0$$

となるので、表現が“1203”なのです。

これに対し、コンピュータ内部で使われている 2 進数は、2 のべき乗を基本として表された数です。この場合、多項式の係数は 2 より小さい整数でなければならないので、0 か 1 です。例えば、10 進数で 123 と表示される数を 2 の多項式で表せば、

$$123 = \underline{1} \times 2^6 + \underline{1} \times 2^5 + \underline{1} \times 2^4 + \underline{1} \times 2^3 + \underline{0} \times 2^2 + \underline{1} \times 2^1 + \underline{1} \times 2^0$$

となるので、2進数で表すと、1111011 となります。2進数は0または1だけで表すことができるので、電子回路のON/OFF回路を使って実現することができます。これがコンピュータで2進数が使われている理由です。2進数の一桁を bit といいます。現在のコンピュータは2進数8桁(8bit)を基本に作られていて、これを byte といいます。1byteは8桁の2進数ですから $2^8 = 256$ 個の数字をあつかうことができます。これを0以上の整数と考える場合には、0~255 となります。

さて、Fortranの整数型は4byteなので、32桁(32bit)の2進数です。すなわち、 2^{32} 個の数字があつかえます。しかし負の整数を含ませるために、整数型の範囲は $-2^{31} \sim 2^{31} - 1$ になります。 2^{31} は2,147,483,648 ですから、取り扱える数の上限はおよそ21億です。整数型の計算をするときは、絶対値がこれを超えないようにしなければなりません。

これに対し、実数型数は数値の bit を仮数部 f と指数部 e 、および正負を決める符号 bit に分割して、 $\pm f \times 2^e$ のような形式で表現します。指数部 e は負数も含んでいるので、 10^{-19} のような絶対値の小さい数も表現できます。実数型における bit 分割とそれによる数の表現には規格がいくつかありますが、現在最もよく使われているのは IEEE 754 と呼ばれている規格です。IEEE 754 における実数型の bit 分割数と、それによる表現可能な絶対値の上限を表 1.5 に示します⁸。

表 1.5 実数型の bit 分割数と表現可能な絶対値の上限 (IEEE 754)

数値型 (bit)	符号 bit	指数部 bit	指数部 e の範囲	仮数部 bit	絶対値の上限
単精度 (32)	1	8	-126~127	23	3.4×10^{38} 程度
倍精度 (64)	1	11	-1022~1023	52	1.7×10^{308} 程度
4倍精度 (128)	1	15	-16382~16383	112	1.1×10^{4932} 程度

例えば、倍精度実数型は仮数部が52bitなので、 $2^{52} \approx 4.5 \times 10^{15}$ 個の区別ができ、これが“有効数字15桁程度”の根拠です。また、指数部の最大が1023なので、最大 10^{308} 程度まで表現することができます。これに対し、単精度は指数部の最大が127なので、最大は 10^{38} 程度です。 10^{38} のような大きな数字を扱うことはあまりないかもしれませんが、計算の途中で現れる可能性は考慮する必要があります。例えば、プランク定数は $h \approx 6.6 \times 10^{-34}$ Js なので、 h の逆2乗が公式の中に入っていると、途中計算で 10^{38} を越えてしまいます。倍精度実数を使う意義はここにもあります。

なお、IEEE 754 での指数部には 2^e という以外にも意味があり、数字の組み合わせによっては、“無限大”や“非数”を表します。もし、print 文で数値を出力したときに、Infinity や INF という文字が出力された場合には無限大です。つまり、計算した結果が表現できる上限を超えたという意味です。これを“オーバーフロー”といいます。

これに対し、NaN と出力された場合は非数です。非数 (Not a Number) というのは、「数字じゃない」という意味ですが、具体的には、0を0で割ったときや、-1の平方根を計算したときの結果がこれに相当します。ただ、結果が非数になっても計算が継続することもあるので、どの時点で発生したのかを特定するのは結構面倒です。

1.4.2 演算の速度を考える

コンピュータの計算動作は $a+b$ のような加算が基本です。 $a-b$ のような減算は b を負数に変換して加算するだけなので加算とそれほど実行時間は変わりませんが、 $a*b$ のような乗算は加算の繰り返し動作ですから、加減算よりかなり遅いです。 a/b のような除算にいたっては、減算の繰り返しを条件付きで行うので、さらに時間がかかります。この演算速度の比較を書けば次のようになります。

⁸IEEE 754 に関しては「<http://ja.wikipedia.org/wiki/浮動小数点数>」を参考にしました。詳しくは、この Web ページを見て下さい。なお、表現できる正数の下限は、およそ「1/上限数」程度、という見当が良いと思いますが、IEEE 754 の規格ではもう少し小さい桁の数まで表せます。もっとも、実際にパソコンで上限と下限をチェックしてみたところ、下限は動作環境やコンパイラによって異なりました。また、4倍精度に関しては上限が倍精度と同じ 10^{308} 程度しかないものもありました。すなわち、必ずしも IEEE 754 の規格通りに実装されているとは限らないようです。

加減算 ≧ 乗算 ≧≧ 除算

このため、割り算ができるだけ少なくなるような計算手順にします。例えば、

```
x = a/b/c
```

と書くより、

```
x = a/(b*c)
```

と書く方が速くなります。最近の計算機はかなり高速に計算することができるので、数回の計算だけならこのような書き換えはあまり意味がありませんが、大量に繰り返し処理をする時には価値があります。べき乗算はもっと遅いので、2乗や3乗程度のときは掛け算にする方が速くなります。例えば、

```
x = a**2 + b**3
```

と書くより、

```
x = a*a + b*b*b
```

と書く方が速くなります。ただし、指数が大きいときには意味がないし、プログラムもわかりにくくなるので3乗程度までが良いと思います。

実数のべき乗(1.2乗とか、3.7乗など)は、対数関数と指数関数を使って計算するので時間がかかります。このため、 $\frac{1}{2}$ 乗に関しては、組み込み関数 `sqrt` を利用する方が速いです。例えば、

```
y = x**0.5  
z = y**1.5
```

と書くより、

```
y = sqrt(x)  
z = y*sqrt(y)
```

と書く方が速くなります。

1.4.3 桁落ちに気を付ける

実数型数の有効数字は倍精度でも15桁程度です。よって、値の接近した2個の実数の引き算をするときは気を付けなければなりません。例えば、

```
2000.06 - 2000.00 = 0.06
```

ですが、計算結果0.06は元の2000.06と比べると有効数字が5桁も少なくなっています。これを“桁落ち”といいます。桁落ちするような引き算はできるだけ避けねばなりません。

例えば、次のように変数 `x1` に小さい正の数 `h1` を加えて `x2` を計算し、`x2` に小さい正の数 `h2` を加えて `x3` を計算したとします。

```
real x1,x2,h1,h2  
x1 = 1.0  
h1 = 1e-7  
h2 = 2e-7  
x2 = x1 + h1  
x3 = x2 + h2  
print *,x3-x1,h1+h2
```

最後の print 文の出力を見ればわかりますが、x3-x1 の値が必要な時には、引き算で直接計算するよりも h1+h2 で計算する方が精度が上がります。

2 次方程式 $ax^2 + bx + c = 0$ の根を求めるときにも注意が必要です。この方程式の 1 根は、

$$x_1 = \frac{-b + \sqrt{b^2 - 4ac}}{2a}$$

ですが、 $b > 0$ かつ $b^2 \gg |4ac|$ の時には、 b と $\sqrt{b^2 - 4ac}$ がほぼ等しくなるので、 $-b + \sqrt{b^2 - 4ac}$ の計算をすると桁落ちする可能性があります。そこでこの根を計算する時は、分子の有理化をした公式を使います。すなわち、分母分子に $-b - \sqrt{b^2 - 4ac}$ を掛けると、

$$x_1 = \frac{(-b + \sqrt{b^2 - 4ac})(-b - \sqrt{b^2 - 4ac})}{2a(-b - \sqrt{b^2 - 4ac})} = \frac{2c}{-b - \sqrt{b^2 - 4ac}}$$

となりますが、最後の公式を使えば桁落ちする心配がありません。2 根とも計算する時には、2 根の積が c/a であることを利用して、まず桁落ちしない方の根 x_1 を計算した後で、もう一方の根を $c/(ax_1)$ で計算すると良いでしょう。

1.4.4 π の作り方

物理や数学の計算をするときには円周率 π を使うことが多いです。 π の値を倍精度計算用に有効数字 16 桁で表せば、

```
real pi
pi = 3.141592653589793
```

となります。これをそのまま使えば良いのですが、計算機によっては有効数字が異なるし、数値を覚えるのも面倒です。そこで逆三角関数を使って計算で求める方法がよく使われています。例えば、

```
real pi
pi = acos(-1.0)
pi = 2.0*asin(1.0)
pi = 4.0*atan(1.0)
```

などのように書くことができます。関数計算には時間がかかりますが、 π の値は変化しないので計算するのは 1 回だけです。問題になるほどではありません。

1.4.5 コメント文の挿入

Fortran では “!” の後に続く文字は全て無視されます。すなわち何を書いても実行とは無関係です。これをコメント文といいます。コメント文を機会あるごとにプログラム中に入れて、書かれている内容を表示するように心がけましょう。さもなくば、プログラムが長くなるにつれて、自分でも何を書いたのか忘れてしまいます。例えば、

```
! area of circles
s = pi*r*r
```

のように、1 列目に “!” を書けば、その行はコメント行になります。また、

```
s = pi*r*r      ! 円の面積
v = 3*pi*r*r*r/4 ! 球の体積
```

のように、実行文の末尾に書き込むこともできるし、日本語を書くこともできます。ただし、日本語環境によっては正しく認識できずに文字化けすることがあります。可般性を考慮するならローマ字つづり

でも良いから半角英数字だけでコメントを書いた方が無難です。

1.4.6 継続行と複文

計算式が非常に長くて、作成に使っているエディタウィンドウの横幅を越えると読みづらくなります。また、コンパイラによっては1行に書ける文字数に制限があるので、そのままではエラーになることもあります。そこで、1行の文を途中で改行して、複数行に分割して書くことができます。このとき、次の行に継続するという意志を示す印として、行末に“&”の文字を書きます。例えば、

```
print *,alpha,beta,gamma,delta,epsilon,zeta,eta,iota,kappa,lambda,mu
```

という1行は、

```
print *,alpha,beta,gamma,delta,epsilon &  
      ,zeta,eta,iota,kappa,lambda,mu
```

と分割して書くことができます。継続行は何行にわたってもかまいません。

```
print *,alpha,beta,gamma &  
      ,delta,epsilon &  
      ,zeta,eta,iota &  
      ,kappa,lambda,mu
```

と書いても同じです。最後の行に“&”を付加するとエラーになるので注意して下さい。

逆に、1行に複数の文を書くことも可能です。これを“複文”といいます。2個以上の文を1行で書くには、文と文を分離する記号として“;”の文字を書きます。例えば、

```
x = 1; y = 2; z = 3
```

のように書くことができます。ただし、複文の使用はこのような短い代入文を書く場合に限定して下さい。プログラムは1行で一つの完結した動作を表すのが基本です。1行に複数の文を書くと、動作の流れが見えにくくなるので、多用しない方が良いでしょう。

演習問題 1

(1-1) 2次方程式の根

3個の実変数 a, b, c に適当な値を代入し、それから、

$$ax^2 + bx + c = 0$$

の2根を計算して出力するプログラムを作成せよ。さらに、その根を $ax^2 + bx + c$ に代入して0に近い値になることもプログラムして確かめよ。また、 a, b, c の値が不適切だと、エラーが出ることも確認せよ。(例えば、 $a = b = c = 1$ にしてみよ)

(1-2) ヘロンの公式

三角形の3辺の長さを a, b, c とすると、その三角形の面積 S は次式(ヘロンの公式)で表される。

$$S = \sqrt{p(p-a)(p-b)(p-c)}$$

ただし、 $p = \frac{a+b+c}{2}$ である。

a, b, c に適当な数値を与え、この公式を使って三角形の面積を計算するプログラムを作成せよ。さらに、一辺 a の正三角形の面積を別の方法で計算するプログラムを作成し、ヘロンの公式から求めた面積と一致するかどうか確かめよ。

(1-3) 面積, 体積

1個の実変数 a と整数 n に適当な数値を与え、これから、半径 a の円の面積、半径 a の球の体積、一辺 a の正 n 角形の面積を計算するプログラムを作成せよ。ただし、 π の値はできるだけ正確な値(有効数字15桁程度)を使うこと。

(1-4) ビオ・サバールの法則

右図のように線分導体 PQ に電流 I [A] が流れている時、その線分から距離 r [m] の点 A にできる磁場の磁束密度 B [T] は次の公式で与えられる。

$$B = \frac{\mu_0 I}{4\pi r} (\cos \alpha + \cos \beta)$$

ここで、 μ_0 は真空の透磁率 ($4\pi \times 10^{-7}$ H/m)、 α, β はそれぞれ点 P、点 Q における三角形 APQ の内角である。

この公式を使って、 I, r, α, β を入力して磁束密度 B を計算するプログラムを作成せよ。ただし、角度は“度”の単位の数値で与えて、内部でラジアンに換算するようにせよ。

次に、一辺 a [m] の正三角形の導線に電流 I [A] が流れている時、中心にできる磁束密度 B [T] を計算するプログラムを作成せよ。

さらに、一辺 a [m] の正方形の導線に電流 I [A] が流れている時、中心にできる磁束密度 B [T] を計算するプログラムを作成せよ。




図 1.4 線分電流

(1-5) 相対論的エネルギー

相対性理論によれば、質量 m [kg] の物質は、真空中の光の速度を $c (=2.99792458 \times 10^8$ [m/s]) とすると、 mc^2 [J] の静止エネルギーを持っている。この公式を使って電子と陽子の静止エネルギーを計算するプログラムを作成せよ。ここで電子の質量は $9.10938291 \times 10^{-31}$ [kg]、陽子の質量は $1.672621777 \times 10^{-27}$ [kg] である。なお、エネルギーの単位はエレクトロンボルト (eV) に換算して表示せよ。1eV は $1.602176565 \times 10^{-19}$ [J] である。

また、物質が速度 v [m/s] で運動しているときの運動エネルギーは $m(\gamma - 1)c^2$ である。ここで、 $\gamma = \frac{1}{\sqrt{1 - v^2/c^2}}$ である。電子と陽子が光速の 1% で動いているときの運動エネルギーと、90% で動いているときの運動エネルギーをそれぞれ計算するプログラムを追加せよ。こちらも単位はエレクトロンボルトとする。

(1-6) 複素インピーダンス

インダクタンス $L=0.04$ [mH]、抵抗 $R_L=10$ [Ω] のコイルと、 $C=100$ [μ F] のキャパシタンス、および抵抗 $R=100$ [Ω] を並列に接続し、これに周波数 $f=60$ [Hz] の電圧 $E=100$ [V] を加えたとき、流れる複素電流 I 、力率 $\cos \theta$ および有効電力 P を計算して出力するプログラムを複素変数を使って作成せよ。この場合も π の値はできるだけ正確な値を使うこと。ここで、抵抗 R 、インダクタンス L 、キャパシタンス C 、に対する複素インピーダンスは、それぞれ、

$$Z_R = R, \quad Z_L = i\omega L, \quad Z_C = \frac{1}{i\omega C}$$

である。ただし、 $\omega = 2\pi f$ である。

また、それら 3 つのインピーダンスを直列につないだときに流れる複素電流 I 、力率 $\cos \theta$ および有効電力 P を計算するプログラムも作成せよ。

(1-7) 回路計算

5 個の抵抗 R_1, R_2, R_3, R_4, R_5 がある時、この 5 個を直列にした時の合成抵抗 R_s 、5 個を並列にした時の合成抵抗 R_p 、および R_1 と R_2 を並列にしたものと R_3 と R_4 を並列にしたものと R_5 とを直列にした時の合成抵抗 R_{ps} を計算するプログラムを作成せよ。ただし、5 個の抵抗値は、要素数 5 の 1 次元配列 $R(5)$ で表すとする。

(1-8) 3次元ベクトルと電磁気学

3次元ベクトル $\mathbf{A} = (A_1, A_2, A_3)$ を要素数 3 の 1次元配列 $\mathbf{a}(3)$ で表すとする。まず電場ベクトル \mathbf{E} 、磁場ベクトル \mathbf{B} を配列で表して適当な数値を代入する。次に荷電粒子の速度ベクトル \mathbf{v} を配列で表して適当な数値を代入し、これらの配列から、内積 $\mathbf{v} \cdot \mathbf{E}$ 、および外積ベクトル $\mathbf{v} \times \mathbf{B}$ を計算するプログラムを作成せよ。外積ベクトルも配列にすること。

(1-9) 2行2列の行列計算

2行2列の行列、 $\mathbf{A} = \begin{pmatrix} a_{11} & a_{12} \\ a_{21} & a_{22} \end{pmatrix}$ を、2次元配列 $\mathbf{a}(2,2)$ で表すとする。2個の2行2列の行列 \mathbf{A} と \mathbf{B} を表現する配列、 $\mathbf{a}(2,2)$ と $\mathbf{b}(2,2)$ の各要素に適当な数値を代入して、 \mathbf{A} と \mathbf{B} の足し算の行列 $\mathbf{C} = \mathbf{A} + \mathbf{B}$ 、掛け算の行列 $\mathbf{M} = \mathbf{A}\mathbf{B}$ 、および、それぞれの行列式 D_A, D_B を計算するプログラムを作成せよ。

第2章 手順の繰り返しと条件分岐

第1章で説明した基本的プログラミングは、単純な計算動作を並べたものであり、言ってみれば関数電卓の操作をプログラムという書式で表しただけに過ぎません。コンピュータの能力を使いこなすには、同じような動作を何度も繰り返したり、条件に応じて異なる動作をさせる手法の知識が必要です。本章ではこれらの書き方について説明します。

2.1 手順の繰り返し — do 文

実行文は基本的に上から下へ順に実行されますが、それだけでは類似した手順を繰り返す時に、必要な回数だけ同じ文を書かねばなりません。そこで、ある範囲の手順を必要な回数だけ繰り返し行わせる手段として do 文があります。do 文を使う時の基本形は、

```
do 整数型変数 = 初期値, 終了値
    .....
    .....
enddo
```

です。最初の do の行が do 文で、do 文と最後の enddo 文で範囲を指定された一連の実行文が繰り返し実行されます。この範囲を“do ブロック”といいます。また、プログラムの流れが循環するという意味で“do ループ”ともいいます。do 文の整数型変数を“カウンタ変数”と呼びます。do ブロックの繰り返しは、カウンタ変数に“初期値”を代入することから開始し、do ブロック内の手順を1回実行するごとにカウンタ変数を1増加します。そして、カウンタ変数が“終了値”より大きくなった時点で繰り返しは終了し、enddo 文の次の文に実行が移ります。例えば、

```
do m = 1, 10
    a(m) = m
enddo
```

と書けば、a(1)~a(10)までの配列要素に、1~10までの数値が順に代入されます。do 文における、“カウンタ変数>終了値”の判定は do ブロックの開始時にも行うため、初期値が終了値より大きい場合にはブロック内部が一度も実行されずに enddo 文の次に実行が移ります。

次の do 文のように、整数値をもう一つ追加することで増加量を指定することもできます。

```
do 整数型変数 = 初期値, 終了値, 増分値
    .....
    .....
enddo
```

このときカウンタ変数は初期値から開始して、“増分値”ずつ増加しながら do ブロック内の手順を繰り返し、“カウンタ変数>終了値”の時点で終了します。

例えば、m が 10 以下の奇数の時のみ計算をしたい時は、

```
do m = 1, 10, 2
    .....
    .....
enddo
```

と書きます。この時の終了値は 10 ですが、10 は奇数ではないので計算しません。

増分値は負数を指定することもできます。負数の時にはカウンタ変数が減少していくので、“カウンタ変数<終了値”になった時点で終了します。例えば、100 から順に下って 1 まで繰り返す時には以下のように書きます。

```
do m = 100, 1, -1
  .....
  .....
enddo
```

増分値が負数の時には、“初期値<終了値”ならば do ブロック内部は一度も実行されません。do 文のカウンタ変数に増分値を加えるタイミングは、enddo 文の実行時です。例えば、

```
do m = 1, 3
  a(m) = m**2
enddo
```

という文は、

```
m = 1
  [m > 3 の判定をする。満足しないので、do ブロックを実行]
a(m) = m**2
m = m + 1
  [m > 3 の判定をする。満足しないので、do ブロックを実行]
a(m) = m**2
m = m + 1
  [m > 3 の判定をする。満足しないので、do ブロックを実行]
a(m) = m**2
m = m + 1
  [m > 3 の判定をする。満足するので、do ブロックを終了]
```

という動作になります。この展開からわかるように、do ブロックを終了した時点で、カウンタ変数には終了値より大きい値が代入されています。上例では、do ブロック終了後、 $m=4$ です。do ブロック終了時の m の値を利用する時は、このことを考慮しなければなりません。

次のように、do ブロックの中に別の do ブロックを入れて多重にすることもできます。ただし、カウンタ変数は異なるものを使わなければなりません。これは、do ブロック内でカウンタ変数を変更することが禁止されているからです。

```
do k = 1, 100
  a(k) = k**2
  do m = 1, 10
    b(m,k) = m*a(k)**3
    c(m,k) = b(m,k) + m*k
  enddo
  d(k) = a(k) + c(10,k)
enddo
```

よく使うので覚えておくと便利なのが、合計を計算する do 文です。例えば、 n 要素の 1 次元配列、 $a(1)$, $a(2)$, ..., $a(n)$ の中に数値データが入っている場合、これらの合計 sum を求めるには do 文を使って以下のように書きます。

```
sum = 0
do m = 1, n
  sum = sum + a(m)
enddo
```

この文が合計を計算していることを理解するには、やはり以下のように手動展開してみると良いでしょう。

```

sum = 0
m = 1
sum = sum + a(m)
m = m + 1
sum = sum + a(m)
m = m + 1
sum = sum + a(m)
m = m + 1
.....

```

ここで判定文は省略しました。このプログラムで重要なことは、do文の前で変数 sum に 0 を代入していることです。これがないと正しい結果が得られないことがあります。このようなプログラムならではの書き方はパターンとして覚えておくと良いと思います。

2.2 条件分岐 — if 文

条件に応じて異なる手順を行わせることを“条件分岐”といい、if 文を使って指定します。もっとも単純に、一つの条件に応じて一つの文を実行するかしないかを決めるだけの時は単純 if 文を使います。単純 if 文は以下の形式です。

```
if (条件) 実行文
```

単純 if 文はカッコ内の“条件”が満足されれば、その右の“実行文”を実行し、条件が満足されなければ何もしないで次の文に実行が移ります。例えば、

```

a = 5
if (i < 0) a = 10
b = a**2

```

と書くと、 $i < 0$ の場合には $a = 10$ となり、それ以外は $a = 5$ のままなので、それに応じて b に代入される値が異なります。

しかし単純 if 文には実行文が一つしか書けないので、実行したい文が複数あるときには使えません。また、条件に合った時の動作指定しかできないので、合わなかった時の動作を別に指定したい時には不便です。そこで、通常は単純 if 文ではなく、ブロック if 文を使います。ブロック if 文とは、if 文の実行文のところを then にした文のことで、以下のようにブロック if 文と endif 文で一連の実行文の範囲を指定します。

```

if (条件) then
.....
.....
endif

```

この指定された範囲を“if ブロック”といい、ブロック if 文に書かれた条件を満足した時のみ、if ブロック内の実行文が実行されます。例えば、

```

a = 5
b = 2
if (i < 0) then
  a = 10
  b = 6
endif
c = a*b

```

と書くと、 $i < 0$ の場合には $a=10$, $b=6$ の代入文を実行してから $c=a*b$ を計算しますが、それ以外、すなわち $i \geq 0$ の場合には if ブロック内を実行しないので、 a も b も変化せず、 $a=5$, $b=2$ のままで $c=a*b$

を計算します。

さて、この例では、 $i < 0$ という条件を満足しないときには、あらかじめ $a=5$, $b=2$ という代入をする必要がありません。このように“条件を満足しない場合”に別の動作をさせたいときには `if` ブロック内に `else` 文を挿入します。 `else` 文を挿入すると、ブロック `if` 文で指定した条件を満足しない場合に、`else` 文から `endif` 文までの実行文が実行されます。例えば上記のプログラムは、

```
if (i < 0) then
  a = 10
  b = 6
else
  a = 5
  b = 2
endif
c = a*b
```

と書くことができます。この `if` ブロックでは、 $i < 0$ の場合には $a=10$, $b=6$ を実行、それ以外の場合には $a=5$, $b=2$ を実行します。

また、条件を満足しない場合に、さらに別の条件を指定したいときには `else if` 文を使います。 `else if` 文も条件はかっこで指定し、その後に `then` を書きます。

```
if (i < 0) then
  a = 10
  b = 6
else if (i < 5) then
  a = 4
  b = 7
else
  a = 5
  b = 2
endif
c = a*b
```

この場合、 $i < 0$ の場合には $a=10$, $b=6$ を実行、 $0 \leq i < 5$ の場合には $a=4$, $b=7$ を実行、それ以外 ($i \geq 5$) の場合は $a=5$, $b=2$ を実行、となります。 `else if` 文による新たな条件は `if` ブロック内にいくつ入れてもかまいません。その場合は、“その `else if` 文より以前の条件を全て満足しない場合に、その条件を満足すれば”という意味になります。これに対し、`else` 文は最後の一回しか使えません。

数値計算でよく使う代表的な比較条件の書き方を表 2.1 に示します。

表 2.1 比較条件の書き方

比較条件記号	記号の意味	使用例
<code>==</code>	左辺と右辺が等しいとき	<code>x == 10</code>
<code>/=</code>	左辺と右辺が等しくないとき	<code>x+10 /= y-5</code>
<code>></code>	左辺が右辺より大きいとき	<code>2*x > 1000</code>
<code>>=</code>	左辺が右辺以上のとき	<code>3*x+1 >= a(10)**2</code>
<code><</code>	左辺が右辺より小さいとき	<code>sin(x+10) < 0.5</code>
<code><=</code>	左辺が右辺以下のとき	<code>tan(x)+5 <= log(y)</code>

使用例のように、比較条件を指定する場合には両辺どちらにも計算式を書くことが可能です。

さらに表 2.2 の論理演算記号を使えば、これらの条件を論理的につないだ条件や、否定した条件を与えることもできます。

表 2.2 論理式の書き方

論理演算記号	演算の意味	使用例
“条件1”.and.“条件2”	“条件1”かつ“条件2”のとき	$x > 10$.and. $2*x \leq 50$
“条件1”.or.“条件2”	“条件1”または“条件2”のとき	$x > 0$.or. $y > 0$
.not.“条件”	“条件”を満足しないとき	.not.($x < 0$.and. $y > 0$)

例えば,

```
if (i > 0 .and. i <= 5) then
  a = 10.0
else
  a = 0.0
endif
```

と書くと, i が 0 より大きく “かつ” 5 以下の時, すなわち, $0 < i \leq 5$ のときに $a = 10$, それ以外は $a = 0$ となります. なお, 横着してこのブロック if 文を,

```
if (0 < i <= 5) then      ! これはエラー
```

と書くことはできないので注意しましょう.

do 文で繰り返し計算をする場合, “ n 回ごとに出力する” というように, 一定間隔で動作を変更したいことがあります. この場合には, 表 1.3 に示した剰余を計算する組み込み関数 mod を利用して条件分岐をします. 例えば, 10 回に 1 回出力するなら,

```
do m = 1, 10000
  .....
  if (mod(m,10) == 0) print *,m,x,y
  .....
enddo
```

のように書きます. プログラムというのは, 基本的に全て計算であり, 流れのコントロールも計算結果を使って行わなければなりません. このようなケースで mod を使うのは, パターンの一つとして覚えておくと良いでしょう.

2.3 無条件ジャンプ — goto 文, exit 文, cycle 文

プログラムというのは基本的に上から順に実行していくものです. do 文を使えば, do ブロックで指定した範囲の実行文を所定の回数繰り返すことができますが, 繰り返す範囲は固定されているし, 繰り返しを終了する条件はカウンタ変数の増加で決まります.

これに対し, より一般的にプログラムの流れを変えたい時, 例えば途中で計算を中断してプログラムの最初からやり直す, とか, 最後の文に一気に移動して終了する, とかいう時には goto 文を使います. goto 文を使えば, 指定した行へ強制的に移動することができます. 計算機的には, これを “ジャンプする” といいます.

goto 文とは, 以下のように goto の後に “文番号” と呼ばれる整数を指定した文です.

```
goto 文番号
```

これに対し, この goto 文でジャンプしたい先の行には, 以下のように実行文の前にスペースを 1 個以上空けて文番号を書きます.

```
文番号 実行文
```

goto 文を使ったプログラム例を以下に示します。

```
cd = 10
goto 11          ! 文番号 11 の行へジャンプ
cd = 50
ab = 20
ij = 1
11 ab = 1000    ! この行へジャンプ
```

最後の行で ab=1000 の前に書かれた 11 が文番号です。この例では、最初の cd=10 の実行後、cd=50 から ij=1 までの文は実行されず、直ちに ab=1000 が実行されます。すなわち、cd=50, ab=20, ij=1 の文は書いていないのと等価です。このように有無をいわずにジャンプすることを“無条件ジャンプ”といいます。

goto 文でバックすることも可能です。例えば、次のように書けば、指定した文番号 22 の行と goto 文の間の動作を繰り返し実行します。

```
cd = 50
22 ab = 200      ! この行へジャンプ
cd = cd + ab - ef
ef = 10
goto 22         ! 文番号 22 の行へジャンプ
ij = 1
```

もっとも、この例ではいつまでたっても goto 文の次の ij=1 は実行されません。こういうのは“無限ループ”と呼ばれ、プログラムエラーの一つです。計算結果に応じて条件分岐し、goto 文より下の行へジャンプする別の goto 文や、プログラム自体を終了させる stop 文を挿入しなければ、プログラムは永遠に終了しません。

文番号は行の指定に使うだけなので、重複さえしなければどの行にどんな数値を付けても良いのですが、なるべく下に行くほど大きくなるように数値を選びます。また、文番号を特定の実行文に付けると、変更する時に付け替えが面倒だと思う時には continue 文を挿入します。continue 文に動作は無く、文番号で位置を指定するためだけに用います。例えば、上記のプログラムは、

```
cd = 50
22 continue     ! この行は動作がない
ab = 200
cd = cd + ab - ef
ef = 10
goto 22
ij = 1
```

と等価です。

goto 文は、多用するとプログラムの流れがわかりにくくなるし、無限ループに陥る可能性もあるので使用はできるだけ避けるべきです。基本的な繰り返しや、条件に応じたジャンプは do ブロックと if ブロックでほとんどすべて書くことができます。

do ブロックを使って繰り返し計算をする時、条件によって途中で繰り返しを終了したい場合があります。この場合、原理的には goto 文を使わなければなりません。goto 文の使用を極力避けるという方針から、exit 文が用意されています。do ブロック中で exit 文を実行すると、その do ブロックの外に飛び出して enddo 文の直後から実行を継続します。

例えば、要素数 n の配列 a(n) に代入されている値を条件付きで平均するため、

```

sum = 0
do m = 1, n
  sum = sum + a(m)
  if (sum > 100) exit
enddo
ave = sum/n

```

のように書いたプログラムは、次の goto 文を使ったプログラムと等価です⁹。

```

sum = 0
do m = 1, n
  sum = sum + a(m)
  if (sum > 100) goto 10
enddo
10 ave = sum/n

```

なお、do ブロックの中から外へのジャンプはできますが、外から中に入るジャンプは禁止されています。do ブロックの途中で実行を中断し、残りの部分をスキップしてカウンタ変数を進める時には cycle 文を使います。例えば、

```

do m = 1, n
  sum = sum + a(m)
  if (sum > 100) cycle
  sum = sum*2
enddo

```

のように書いたプログラムは、次の goto 文を使ったものと同じです。

```

do m = 1, n
  sum = sum + a(m)
  if (sum > 100) goto 10
  sum = sum*2
10 enddo

```

すなわち、cycle 文の実行は enddo 文にジャンプするのと等価です。enddo 文にジャンプすれば、カウンタ変数 m を増加して、 $m > n$ かどうかをチェックした後、条件を満足しなければ再び do ブロックを最初から実行することになります。ただし、cycle 文は do ブロック内部の制御なので次のように if ブロックを使って同じ動作をさせることができます。

```

do m = 1, n
  sum = sum + a(m)
  if (sum <= 100) then
    sum = sum*2
  endif
enddo

```

この方が、条件に応じた動作を明示している点が良いと思います。goto 文を多用しない方がよい、という意味合いと同じで、cycle 文を使うとプログラムがわかりにくくなるのであまり使わない方がよいと思います。

⁹このプログラムを修正して、合計した要素 $a(1) \sim a(m)$ の平均値を計算したい場合には、単に n の代わりに m で割るよう書き換えるだけでは不完全です。なぜだか考えてみましょう。

2.4 プログラミング スマートテクニック (その2)

本節では do ループを使うときのテクニックや、プログラムを読みやすくするためのヒントをいくつか紹介します。do 文や if 文がマスターできれば、本格的なプログラムが書けるようになります。大量のデータを処理する計算ができるし、色々な計算処理機能を複合して、条件に応じて使い分けることも可能です。それに伴ってプログラムが長くなりますから、文法的に間違いがなければ良い、という考えで書くのではなく、読みやすいプログラムになるよう心がける必要があります。ここで、“読みやすいプログラム”というのは、チェックがしやすいプログラムのことです。読みやすく書いておけば、書き間違いに気がつきやすいし、実行時にエラーが出ても早期に誤りを発見することができます。以下のヒントは、プログラミングの初心者が読むと「細かいことだから無視してもいいや」と感じるかもしれませんが、慣れればそれほど手間ではありません。面倒がらずに心がけましょう。

2.4.1 do ループのテクニック

do ブロック内部に同じ計算を繰り返す記述があるときには、あらかじめ計算をしておきます。例えば、

```
do i = 1, 10000
  a(i) = b(i)*c*f**2
  b(i) = sin(x)*a(i)
enddo
```

と書くと、繰り返しごとに $c*f**2$ や $\sin(x)$ を計算することになりますが、これらは do ブロック内部では変化しないのですから、あらかじめ計算しておくことで速くなります。例えば、

```
cf = c*f**2
sx = sin(x)
do i = 1, 10000
  a(i) = b(i)*cf
  b(i) = sx*a(i)
enddo
```

のように書き換えると速くなります。

また、do ブロック内で何度も同じ割り算をするときには、逆数を掛けるように書き換えると速くなります。これは、割り算が遅いからです。例えば、

```
do i = 1, 10000
  a(i) = b(i)/c
  x(i) = a(i)/10.0
enddo
```

と書くより、

```
d = 1.0/c
do i = 1, 10000
  a(i) = b(i)*d
  x(i) = a(i)*0.1
enddo
```

と書く方が速くなります。もっとも、プログラムがわかりにくくなるという欠点もあるので、割り算の箇所が少なく、繰り返し回数がさほど多くない時にはそれほど神経質に変形する必要はないと思います。

配列を使った繰り返し計算をするときは、メモリをできるだけ連続的に読み書きする方が速くなります。このため、多重 do ブロックを使って2次元以上の配列計算をするときは、左の要素から先に進めるようにします。例えば、


```

real b(10,100)
integer m,n
do n = 1, 100
  do m = 1, 10          ! 左の要素を先に進めると速い
    b(m,n) = m*n
  enddo
enddo

```

のように、2次元配列 $b(m,n)$ に計算結果を代入するときは、左側の要素 m に関する `do` ブロックを右側の要素 n の `do` ブロックの内側に持ってくる方が高速です。これは、内側の `do` ブロックのカウンタ変数が先に進むので、 $b(1,1), b(2,1), b(3,1) \dots$ のように、メモリの並んでいる順に格納していくからです。これを、

```

real b(10,100)
integer m,n
do m = 1, 10
  do n = 1, 100        ! 右の要素を先に進めると遅い
    b(m,n) = m*n
  enddo
enddo

```

のように書くと、 $b(1,1), b(1,2), b(1,3) \dots$ のように飛び飛びに格納していくので効率が悪くなり、スピードダウンします。

大きな数に小さい数を加えると、小さい数が桁落ちして情報が失われる可能性があります。例えば、1次元配列 $a(i)$ の合計 sum を計算するプログラムは、

```

sum = 0
do i = 1, n
  sum = sum + a(i)
enddo

```

で良いのですが、 $a(i)$ が全て正数で、 n が非常に大きい場合には、合計 sum がだんだん大きくなっていくので、後の方で加えた $a(i)$ の情報が失われる可能性があります。倍精度実数を使うことを推奨している理由の一つがこれです。この桁落ちを防ぐには、例えば、2個ずつ加えてそれらを別の配列に入れ、その後それらを同様に2個ずつ加えていって、 \dots とするのが良いのですが、プログラムが複雑になってしまいます。

ひとつの比較的簡単な解決法は、補助変数を使って、誤差を別に評価しておく方法です [2]。

```

sum = 0
res = 0
do i = 1, n
  res = res + a(i)
  tmp = sum
  sum = sum + res
  tmp = sum - tmp
  res = res - tmp
enddo

```

この方法では、桁落ち分を補助変数 res に保存して別途加えるので、計算精度が上がります。倍精度計算をしていれば必ずしもこの方法にする必要はありませんが、精度の良い合計計算が必要になった時のために覚えておくと良い方法です。

2.4.2 多項式を計算する手法

多項式を計算するときは、掛け算の回数ができるだけ少なくなるように考えます。一般的に、一番良い計算手法は horner 法です。例えば、

$$y = a_0 + a_1x + a_2x^2 + a_3x^3 + a_4x^4$$

を、乗算を明示してプログラムにすると、

```
y = a0 + a1*x + a2*x*x + a3*x*x*x + a4*x*x*x*x
```

のようになり、10回の乗算が必要です。ところが、これを变形して、

```
y = a0 + (a1 + (a2 + (a3 + a4*x)*x)*x)*x
```

と書けば、4回の乗算で計算できます。これが horner 法です。一般的に、horner 法は0の係数が少ない多項式の計算で有効です。

一般的に、 n 次の多項式、

$$y = a_0 + a_1x + a_2x^2 + \dots + a_nx^n$$

における係数 $a_0, a_1, a_2, \dots, a_n$ が、下限が0の1次元配列、 $a(0), a(1), a(2), \dots, a(n)$ にそれぞれ代入されている場合には、次のような do 文を使って計算することができます。

```
y = a(n)
do i = n-1, 0, -1
  y = a(i) + x*y
enddo
```

2.4.3 do while 文と無条件 do 文

do 文には、2.1 節で述べた基本形の他に、条件でループの動作を続けるか終了するかを決める形式も用意されています。これが do while 文です。do while 文は以下のような書式です。

```
do while (条件)
  .....
  .....
enddo
```

この場合、do while 文の条件を満足しなくなるまで、その do ブロック内部を繰り返し実行します。もし条件を満足しなければ、do ブロックは終了して enddo 文の次に実行が移ります。例えば、

```
integer n
n = 100
do while (n > 0)
  n = n/2
  print *,n
enddo
```

と書けば、 n を2で割って行って、0になった時点でループが終了します。ここで、 n を整数型にしているところがポイントです。実数型だといつループが終了するかわかりません。

また、“do のみ”の do 文もあります。この場合、do ループを終了させる条件がないので、do 文と enddo 文で指定した範囲をいつまでも繰り返します。例えば、

```
do
  sum = sum + x**2
  if (sum > 100) exit
  x = 1.2*x + 0.5
enddo
```

と書くと、sumが100を超えるまで計算を続けます。この無条件do文を使うときは、ブロック内部に適切な条件分岐で外に出る記述を入れておかないと、無限ループになって永遠に終了しないので注意して下さい。

2.4.4 0.1を10回足しても1に等しくならない

実数型を使って条件分岐をするときに注意しなければならないことがあります。次のプログラムを考えてみましょう。

```
real x
integer m
x = 0.0
do m = 1, 20
  x = x + 0.1           ! 0.1を繰り返し加える
  if (x == 1.0) exit
enddo
print *, 'x = ', x
```

このdoループは、実数型変数xに0.1を繰り返し加えていきますが、ループの10回目でif文の条件「xが1.0に等しい」を満足し、exit文によりループの外に出て、print文の出力は1.0になると予想されます。しかし、実際にやってみるとわかりますが、print文の出力は2.0です。つまり、if文の条件は満足しない、すなわち、xが1.0に等しくなることはないのです。なぜでしょうか。

これは、コンピュータが2進数で計算しているためです。1.4.1節で述べたように、10進数で123と表現される数を2の多項式で表せば、

$$123 = 1 \times 2^6 + 1 \times 2^5 + 1 \times 2^4 + 1 \times 2^3 + 0 \times 2^2 + 1 \times 2^1 + 1 \times 2^0$$

となり、これが2進数で表したときに1111011となる理由でした。コンピュータの数値は全て2進数なので、小数点以下の数値も2進数で表さねばなりません。例えば、10進数の0.123は

$$0.123 = 1 \times 10^{-1} + 2 \times 10^{-2} + 3 \times 10^{-3}$$

のように、10の負べき乗を基本として表された数のことですが、これと同様に、2進数で小数点以下の数xを表すには、

$$x = b_1 \times 2^{-1} + b_2 \times 2^{-2} + b_3 \times 2^{-3} + \dots$$

となるような0または1の係数 b_1, b_2, b_3, \dots を選んで、

$$0.b_1b_2b_3\dots$$

と並べます。コンピュータはこの係数 b_1, b_2, b_3, \dots で小数点以下を表現しています。

さて、コンピュータのメモリで表現できる2進数の桁(bit数)には限りがあるので、小数はどこかで打ち切る必要があります。ところが、2進数の場合、有限小数で表せるのは、 2^{-n} の和だけです。例えば、0.5(= 2^{-1})や0.125(= 2^{-3})やその和である0.625は有限の2進小数で表すことができますが、 2^{-n} の和で表せない数値は無限小数になってしまいます。例えば、10進数の0.1は、

$$0.1 = 0 \times 2^{-1} + 0 \times 2^{-2} + 0 \times 2^{-3} + 1 \times 2^{-4} + 1 \times 2^{-5} + 0 \times 2^{-6} + 0 \times 2^{-7} + 1 \times 2^{-8} + 1 \times 2^{-9} \dots$$

となり、2進数で表せば、0.0001100110011... という循環小数になります。よって、コンピュータにおける0.1は近似値でしかなく、10回加えても1に等しくなりません。

このように、実数計算をするときは10進数と2進数の違いを考慮してプログラムを書かないと、予想外の結果になることがあります¹⁰。また、同じ少数点以下の数を繰り返し加えるときには、回数を掛け算した方が精度が上がります。例えば、

```
real x
integer m
x = 0.0
do m = 1, 100
  x = x + 0.1
  .....
enddo
```

と書くよりも、

```
real x
integer m
do m = 1, 100
  x = 0.1*m
  .....
enddo
```

と書く方が、xの精度は上がります。

2.4.5 ブロックを明確にするために字下げする

これまでのプログラム例でも示していますが、doやifなどのブロック中では字下げ(インデント)をしましょう。これは、ブロックの範囲が一目でわかるからです。

例えば、ifブロック、

```
if (n < 0) then
  x = 10.0
else if (n < 10) then
  x = 1.0
else
  x = 0.0
endif
```

やdoブロック、

```
do i = 1, n
  b(i) = a(i)
  c(i) = a(i)*sin(b(x))
enddo
```

のように、ブロック内部の文をスペースを入れてずらせておくとブロックの範囲が一目で判断できます。通常、2~4個のスペース程度が良いと思います。

最近のエディタには、オートインデントといって、改行するとその前の行と先頭がそろうようにカーソルが移動する機能があるので、インデントをするのはそれほどの手間ではありません。

2.4.6 文字間や行間は適当に空ける

文中の文字間は適当に空けたほうが読みやすくなります。筆者は“=”と“+”の両側に必ずスペースを入れることにしています。また、代入文が何行も続くときには“=”を上下にそろえると読みやすくなり

¹⁰ちなみに、この理屈で言えば、 0.1×10 という掛け算も1.0と等しくならないはずですが、手元のパソコンでやってみるとif文を満足しました。これは、四捨五入のように計算結果を丸めるためです。しかし、丸め方はハードウェアやコンパイラの仕様に依存するので、それを期待してプログラムを書くのは避けた方が無難だと思います。

ます。特に、同じパターンで少しずつ内容が違ふ文を並べるときには、パターンが並ぶようにスペースを入れることをお勧めします。以下は、筆者のシミュレーションプログラムの一部です。

```
wm( 1,m1) = (1-dx)*(1-dy)*(1-dz)
wm( 2,m1) =   dx *(1-dy)*(1-dz)
wm( 3,m1) = (1-dx)*  dy *(1-dz)
wm( 4,m1) =   dx *   dy *(1-dz)
wm(11,m1) = (1-dx)*(1-dy)*  dz
wm(12,m1) =   dx *(1-dy)*  dz
wm(13,m1) = (1-dx)*  dy *  dz
wm(14,m1) =   dx *   dy *  dz
```

これをスペース抜きで書くと、以下ようになります。

```
wm(1,m1)=(1-dx)*(1-dy)*(1-dz)
wm(2,m1)=dx*(1-dy)*(1-dz)
wm(3,m1)=(1-dx)*dy*(1-dz)
wm(4,m1)=dx*dy*(1-dz)
wm(11,m1)=(1-dx)*(1-dy)*dz
wm(12,m1)=dx*(1-dy)*dz
wm(13,m1)=(1-dx)*dy*dz
wm(14,m1)=dx*dy*dz
```

この二つは全く同じことが書いてあるのですから全く同じ結果を出します。しかし、パターンをそろえた前者は単なる美的趣味でスペースを入れたのではありません。この例では、どこか一箇所、例えば dx を dy に書き間違えただけでも正しい結果が得られません。しかも、 dx と dy を書き間違えても文法的なミスにはならず、出力結果を見ても間違いに気づかない可能性があります。よって、できるだけプログラムを書いている段階でミスを取り除いておかねばなりません。

この時、前者のようにパターンをそろえておけば、横方向に一行一行チェックするだけでなく縦方向にも比較しながらチェックすることができます。色々な角度からチェックすることで、ミスのないプログラムにすることができるのです。

この他、文と文の間に、適当なコメント文や何も書いていない行を挿入すれば、プログラムの流れに区切りができて読みやすくなります。これは文章を段落に分けるように、プログラムを区分けすると考えればいいでしょう。

2.4.7 定数は意味のある名前をつけた変数に代入して使う

数学的に決まっている単純な定数(2倍するとか、1を加えるとか、3乗するとか)の場合以外は、できるだけ定数の使用量を減らすほうがいいと思います。プログラムを書いている時には理解していても時間が経ってから読んだ時に意味がわからなくなる可能性があるからです。

例えば、電子の質量 m_e ($= 9.1093897 \times 10^{-31}$ kg) と光の速度 c ($= 2.99792458 \times 10^8$ m/s) を使って計算すれば $m_e c^2 = 8.187111168 \times 10^{-14}$ J ですが、これを単位とするエネルギー、 $energy/m_e c^2$ を計算するプログラムを、

```
ene = energy/8.187111168e-14
```

と書くと、後で $8.187111168 \times 10^{-14}$ という数字がどこから来たのかわからなくなる可能性があります。こういう時には多少面倒でも、

```
me = 9.1093897e-31      ! 電子の質量
cl = 2.99792458e8      ! 真空中の光速
mc2 = me*cl**2
ene = energy/mc2
```

としておけば、わかりやすいしプログラム内容の記録にもなります。

また、do ブロックを 100 回繰り返す、とか、10 回ごとに出力する、とかいう制御数も、変数にしておけば数値の意味が明示されるし、変更もしやすくなります。例えば、

```
do i = 1, 100                ! 100 が 1 個
  a(i) = b(i)**2
enddo
sum = 0
do i = 1, 100                ! 100 が 1 個
  sum = sum + a(i)
enddo
ave = sum/100                ! 100 が 1 個
```

というプログラムにおいて、100 という数は配列要素数に依存する共通の数値ですから、以下のように変数にします。

```
imax = 100                  ! 100 を変更するときはここだけで OK
do i = 1, imax
  a(i) = b(i)**2
enddo
sum = 0
do i = 1, imax
  sum = sum + a(i)
enddo
ave = sum/imax
```

こうしておけば意味がはっきりするし、100 を 200 に変更したい時には変更点が一箇所ですみます。このような変数化はプログラムが長くなるほど価値が増します。

演習問題 2

(2-1) 繰り返し出力

$i=1,2,3,\dots,n$ の整数に対し, $i, i^2, \frac{1}{i}, \sqrt{i}$ の値を並べて出力するプログラムを作成せよ. なお, 整数と実数の変換に注意すること.

(2-2) 統計計算

1次元配列, $a(n)$ に, $1,2,3,\dots$ のデータを順に n 個代入し,

$$\begin{aligned} \text{平均 } \bar{A} &= \frac{1}{n} \sum_{i=1}^n A_i \\ \text{標準偏差 } \sigma &= \sqrt{\frac{1}{n} \sum_{i=1}^n (A_i - \bar{A})^2} \end{aligned}$$

を計算するプログラムを作成せよ.

次に, $a(n)$ に $1,3,5,\dots$ と奇数を順に n 個入れて, 同様に平均と標準偏差を計算するプログラムを追加せよ.

(2-3) 定積分

等間隔 h ごとに並んだ座標点 $x_1(=a), x_2, \dots, x_n(=b)$ に対して, 関数値, $f(x_1), f(x_2), \dots, f(x_n)$ が与えられているとする. このとき $f(x)$ の定積分を

$$\int_a^b f(x) dx = h \left(\frac{1}{2} f(x_1) + f(x_2) + \dots + f(x_{n-1}) + \frac{1}{2} f(x_n) \right)$$

と近似的に計算する方法を台形公式という. 以下に与えた関数と積分範囲の定積分値を台形公式を使って計算せよ. その際, 分割数 n が大きくなるほど正確な値に近づくことも確かめよ.

(1) $f(x) = \sin(x), a = 0, b = \pi$

(2) $f(x) = (x-2)^3, a = 1, b = 5$

(2-4) 漸化式

次の漸化式で計算される数値を, それぞれ, 配列 $a(n), b(n)$ に代入し, それぞれの平均と標準偏差を計算せよ.

(1) $a_{i+1} = 3a_i + 2, \quad a_1 = -1$

(2) $b_{i+1} = \frac{2}{3}b_i - 4, \quad b_1 = 3$

(2-5) horner 法による多項式計算とテーラー展開

2.4.2 節で紹介した horner 法を使って、テーラー展開の公式から指数関数や三角関数を計算するプログラムを作成せよ。ただし、第 0 項から第 n 項までの指数関数と三角関数のテーラー展開の係数は以下の通りである。

$$\begin{aligned} e^x &= 1 + \frac{x}{1!} + \frac{x^2}{2!} + \frac{x^3}{3!} \cdots + \frac{x^n}{n!} &= \sum_{i=0}^n \frac{x^i}{i!} \\ \cos x &= 1 - \frac{x^2}{2!} + \frac{x^4}{4!} \cdots + (-1)^n \frac{x^{2n}}{(2n)!} &= \sum_{i=0}^n (-1)^i \frac{x^{2i}}{(2i)!} \\ \sin x &= x - \frac{x^3}{3!} + \frac{x^5}{5!} \cdots + (-1)^n \frac{x^{2n+1}}{(2n+1)!} &= \sum_{n=0}^n (-1)^i \frac{x^{2i+1}}{(2i+1)!} \end{aligned}$$

次数 n と変数値 x は複数個選んで計算し、組み込み関数で計算した値と比較して、 n が大きくなるほど正確な関数値に近づくこと、 x が大きくなるほど近似が悪くなることなどを確かめよ。

(2-6) 2 次方程式の根 (根の判別付き)

3 個の実変数 a, b, c に適当な値を代入し、それから、

$$ax^2 + bx + c = 0$$

の根を計算して出力するプログラムを作成せよ。まず、根の判別式、 $D = b^2 - 4ac$ を計算して、2 実根になった時、重根になった時、虚根になった時でそれぞれ正しい結果を出力するようにせよ。さらに、その根を $ax^2 + bx + c$ に代入して 0 に近い値になることを、判別式の値と数値型に応じて計算手順を変えて確かめよ。

(2-7) 非線形方程式の解法：逐次代入法

$x_0 = 0$ として、次のように次々に代入していくと、 x_n は徐々に $\cos x = x$ の解に近づくことがわかっていく。

$$\begin{aligned} x_1 &= \cos x_0 \\ x_2 &= \cos x_1 \\ x_3 &= \cos x_2 \\ &\vdots \\ x_{n+1} &= \cos x_n \end{aligned}$$

最初に、この代入を 100 回繰り返す、その後で 101 回目の値 x_{101} と 100 回目の値 x_{100} の差を計算するプログラムを作成せよ。

それができたら、収束条件を $|x_{n+1} - x_n| < \varepsilon$ とし、 ε を適当に小さい値に定めて (例えば、 10^{-7})、収束したら x_n を出力して終了するプログラムにせよ。このとき、収束するまでの回数も出力せよ。

なお、この問題では配列を使わないこと。

(2-8) 非線形方程式の解法：Newton 法

次の公式を繰り返し用いて方程式 $f(x) = 0$ の根を近似的に求めるアルゴリズムを Newton 法という。

$$x_{n+1} = x_n - \frac{f(x_n)}{f'(x_n)}$$

以下の関数 $f(x)$ に対し、適当な初期値 x_0 を与えて Newton 法で根を求めるプログラムを作成せよ。また収束条件を $|x_{n+1} - x_n| < \varepsilon$ とし、 ε は適当に定めよ。このとき、収束するまでの回数も出力せよ。

(1) $f(x) = x - \cos(x)$

(2) $f(x) = x^3 - 2$

それぞれの問題に対し、得られた結果を $f(x)$ に代入して、答えがどの程度 0 に近いかを調べよ。

(2-9) 常微分方程式 (初期値問題) の解法：サイクロトロン運動

一様な磁場中での荷電粒子の速度 $\mathbf{v}(t) = (u(t), v(t))$ は次の運動方程式を満足する。この運動方程式をオイラー法で解け。ただし、サイクロトロン周波数 ω_c は一定とする。

$$\begin{aligned}\frac{du}{dt} &= \omega_c v \\ \frac{dv}{dt} &= -\omega_c u\end{aligned}$$

ここで 1 回の時間ステップを Δt として $f_n = f(n\Delta t)$ とした時に、オイラー法とは n ステップ目の値 f_n と $n+1$ ステップ目の値を使って $\frac{df}{dt} \simeq \frac{f_{n+1} - f_n}{\Delta t}$ という近似で時間微分を評価する方法である。例えば、 x 方向は

$$\frac{u_{n+1} - u_n}{\Delta t} = \omega_c v_n$$

と近似できるから、これを变形して、 $u_{n+1} = u_n + \omega_c v_n \Delta t$ となる。 v_{n+1} も同様に变形すれば、オイラー法とは次の連立の漸化式を使って (u_n, v_n) から (u_{n+1}, v_{n+1}) を計算することである。

$$u_{n+1} = u_n + \omega_c v_n \Delta t$$

$$v_{n+1} = v_n - \omega_c u_n \Delta t$$

1 周期 $\frac{2\pi}{\omega_c}$ を適当な回数 N で分割して Δt を決定し、初期条件 $u_0 = 1, v_0 = 0$ から出発して、オイラー法で N 回計算するようなプログラムを作成せよ。そして、その結果得られる 1 周期後の速度、 u_N, v_N と u_0, v_0 を比較せよ。また、10 周期後の速度、 u_{10N}, v_{10N} とも比較せよ。

次に、オイラー法を少し修正したシンプレクティック法、(v_{n+1} の式に注意！)

$$u_{n+1} = u_n + \omega_c v_n \Delta t$$

$$v_{n+1} = v_n - \omega_c u_{n+1} \Delta t$$

のプログラムを作成し、オイラー法の結果と比較せよ。

なお、以上のプログラムは 配列を使わないで作成すること。

(2-10) 常微分方程式 (境界値問題) の解法: 1次元ポワソン方程式

電荷密度 $\rho(x)$ が与えられているときに電位が満足する1次元ポワソン方程式

$$\frac{d^2V(x)}{dx^2} = -\rho(x)$$

を解くには、以下のようにする。

等間隔 h ごとに並んだ x 座標点 $x_i = hi, (i = 0 \sim N)$ に対して, $V_i = V(x_i), \rho_i = \rho(x_i)$ として, 上式の左辺を差分で近似すれば,

$$\frac{V_{i+1} - 2V_i + V_{i-1}}{h^2} = -\rho_i$$

となる。これを座標点に関して書き下せば,

$$\begin{aligned} V_0 - 2V_1 + V_2 &= -\rho_1 h^2 \\ V_1 - 2V_2 + V_3 &= -\rho_2 h^2 \\ V_2 - 2V_3 + V_4 &= -\rho_3 h^2 \\ &\vdots \\ V_{N-2} - 2V_{N-1} + V_N &= -\rho_{N-1} h^2 \end{aligned}$$

になるが, 両端の電位 $V_0 = V(x_0), V_N = V(x_N)$ が与えられていれば, これらは V_1, V_2, \dots, V_{N-1} に対する連立1次方程式になる。

一般に, 連立1次方程式,

$$a_i V_{i-1} + b_i V_i + c_i V_{i+1} = d_i \quad (i = 1, 2, \dots, N-1)$$

を V_0 と V_N を与えて解くときには, まず, $G_0 = 0, H_0 = V_0$ から出発して,

$$G_i = -\frac{c_i}{b_i + a_i G_{i-1}}, \quad H_i = \frac{d_i - a_i H_{i-1}}{b_i + a_i G_{i-1}} \quad (i = 1, 2, \dots, N-1)$$

を計算し, 次に, V_i を以下の式で計算する (計算の方向に注意!).

$$V_i = G_i V_{i+1} + H_i \quad (i = N-1, N-2, \dots, 1)$$

以上の方法を用いて, 次の2種類の電荷密度と境界条件における電位を計算するプログラムを作成せよ。

- (1) $\rho(x) = 0$ 境界条件は, $V_0 = 0, V_N = 1$
- (2) $\rho(x) = \sin\left(\frac{2\pi x}{Nh}\right)$ 境界条件は, $V_0 = 0, V_N = 0$

第3章 サブルーチン

ここまで、Fortranの基本的な文法と高速化のテクニック、エラーの出にくいプログラムの書き方などを説明しました。ここまでの知識を使うだけで、原理的にはどんなプログラムを作ることでも可能です。プログラムとは計算機に仕事をさせるための手順ですから、それほどバラエティはありません。動作の繰り返しを書くためのdo文や、条件分岐のif文が使いこなせれば、計算機の能力のほとんどを引き出すことができます。

しかし長いプログラムを書くときは、メンテナンスのしやすさを考慮して書かなければなりません。プログラムが短ければ全体を見ながらチェックや修正ができますが、長くなるとチェックに時間がかかり、見落としによる修正ミスが起こる可能性も高くなります。見落としを防ぐには何度も見直す必要があり、チェック時間はさらに増大します。

このメンテナンスを容易にする手法の一つが“サブルーチン”です。サブルーチンとはメインプログラムと同じレベルの閉じたプログラムのことで、必要に応じて他のプログラムから呼び出してその機能を使います。長いプログラムをサブルーチンに分割し、それらをビルディングブロックとして全体を構成すれば、プログラム作成や修正の作業が楽になります。本章ではこのサブルーチンの作り方と使い方を説明します。

3.1 サブルーチンの利用目的

“ルーチン”とは、動作開始点と終了点が定義されている閉じたプログラム手順を示す用語です。プログラムを起動したときに最初に動作するのはメインプログラムですが、これを“メインルーチン”ともいいます。“サブルーチン”は、メインプログラムと同様に動作開始点と終了点を持っていますが、メインプログラムと異なり、単独で動作することはできません。必ず、他のルーチンに動作開始命令を記述して、必要に応じて実行させる必要があります。一つのプログラムにメインプログラムは一つですが、サブルーチンは名前が異なれば複数存在してもかまいません。また、いくつかのルーチンごとにファイルを作成して別々にコンパイルし、必要に応じて結合(リンク)させることも可能です。

サブルーチンを利用する目的は主として3つあります。

- (1) 複数の場所で同じ計算手順を使用するため
- (2) 方程式の解法や行列式の計算などのような定型処理をするため
- (3) 長いプログラムを分割してメンテナンスを容易にするため

(1)がサブルーチン本来の使い方です。ある一連の手順を複数の場所で使うとき、それぞれの場所に同じプログラムを書くと、プログラムが長くなるし、修正の際に何カ所も同じ修正をしなければなりません。このとき、その手順をサブルーチンに置き換えれば、プログラムは短くなるし、チェック作業も短縮されます。

(2)は(1)を発展させたものです。複雑な数学公式や汎用性のある数値計算アルゴリズムをプログラム中に直接記述するときは、プログラム中で宣言された変数や配列を使って書きます。このため、同じ計算手順を別のプログラムで使いたくてもそのまま使えるとは限りません。そこで計算手順をサブルーチンにして、計算に必要な数値を与える部分と計算結果の数値を受け取る部分だけが外部から見えるようにしておきます。そうすれば、受け渡し部を書き換えるだけで、複雑な計算手順を色々なプログラムから利用することができます。このようにブラックボックス化したサブルーチンは、他の人に提供したり、他の人からもらって利用することも可能で、そのような汎用性のあるサブルーチンを集めたものが、“ライブラリ”です。

さて、計算機シミュレーションのように多数の解析手順を複合したプログラムを書く場合には、(3)の目的が重要になります。様々な物理過程の計算や、入出力に関する作業など、ある程度まとまった内容ごとにサブルーチンにするのです。文章でいえば、章や節に分割するようなものです。このため、メイ

ンプログラムには、それぞれのサブルーチンを呼び出す文と、必要ならそれらを繰り返すための do 文だけを書いておきます。シミュレーションプログラムは、サブルーチンという部品を使って組み立てるものだと考えればいいでしょう。

サブルーチンに分割しておけば、プログラムのメンテナンスが楽になります。これは、それぞれのルーチンが独立しているので、プログラムを修正したときの影響や、エラーが発生する原因がルーチン内に限定されるからです。

3.2 サブルーチンの宣言と呼び出し

サブルーチンは subroutine 文で開始を宣言し、end subroutine 文で終了します。すなわち、次のような構造にします。

```
subroutine subr1
  implicit none
  real a,b
  integer i
  .....
  .....
end subroutine subr1
```

先頭の subroutine 文で、subroutine の後に指定した文字の並び (この例では subr1) を “サブルーチン名” といいます。また、最後の end subroutine 文には subroutine 文と同じサブルーチン名を指定します。見てわかるように、メインプログラムと同じ構造です。サブルーチン内部のプログラムの書き方も基本的にメインプログラムと同じで、subroutine 文の次に implicit none を書き、非実行文を上方に集約して、その後に実行文を書きます。サブルーチンはそれ自体で閉じているので、実行文中で使う変数や配列は、基本的にその内部で宣言しなければなりません。ただし、異なるルーチン間で共用可能な変数を別途用意することは可能です。これについては、3.7 節で説明します。

上の例では、subroutine 文にサブルーチン名しかありませんが、これを “引数なしサブルーチン” といいます。これに対して、サブルーチン名の後に、かっこで囲んだ変数リストを付加することができ、これを “引数ありサブルーチン” といいます。以下に引数ありサブルーチンの一例を示します。

```
subroutine subr2(x,m,y,n)
  implicit none
  real x,y,a,b,z(10)
  integer m,n,i,k
  .....
  .....
end subroutine subr2
```

リスト中の変数を “引数” といいます。この例では、x,m,y,n が引数です。引数は、サブルーチンとそのサブルーチンを使うルーチンの間で数値を受け渡すために使います。引数もサブルーチン内部の変数なので、必要な型に応じた宣言をしなければなりません。

サブルーチンを動作させてその機能を使うときは、call 文を使ってそのサブルーチンを指名します。このため、サブルーチンを指名することを、“サブルーチンを呼び出す” とか “コールする” といいます。call 文は以下のような形式です。

```
call サブルーチン名           ! 引数なしサブルーチン用
call サブルーチン名(数値または変数のリスト) ! 引数ありサブルーチン用
```

例えば、先ほど出てきた、引数なしと引数ありの二つのサブルーチンを使うときは、それぞれ、次の (1) と (2) のようにコールします。

```

real z
integer m
call subr1          !..... (1)
m = 21
call subr2(10.0,100,z,m*5+1)      !..... (2)

```

1.2.1節で説明したように、計算式は計算結果の数値を動作命令に与えるので、call文の引数には、(2)の一番右のように計算式を与えることもできます。

サブルーチンは単独では動作できないので、メインプログラムが別途必要です。例えば、次のようなセットを構成して初めて一つのプログラムが完成します。

```

program stest1
  implicit none
  real x,y
  x = 5.0
  y = 100.0
  call subr(x,y,10)      ! サブルーチンの呼び出し
  print *,x,y
end program stest1

subroutine subr(x,y,n)
  implicit none
  real x,y
  integer n
  x = n
  y = y*x
end subroutine subr

```

ここでは、メインプログラムを先に、サブルーチンを後に書きましたが、逆でもかまいません。サブルーチンが複数存在する場合も、ルーチンを記述する順番は実行結果とは無関係です。サブルーチンの中から別のサブルーチンをコールすることも可能です。

上記のプログラム実行の流れを図3.1に示します。




図 3.1 サブルーチンの呼び出しと実行の流れ

図のように、メインプログラムのcall文でサブルーチンを指名すると、サブルーチンの一番最初の実行文に動作が移り、サブルーチン内部の動作が完了すると、コールしたメインプログラムに戻って、そのcall文の次の文から実行を続けます。

条件に応じて、途中でサブルーチンの処理を打ち切って戻るときにはreturn文を用います。

```

subroutine subr(x,y,m,n)
  implicit none
  real x,y
  integer m,n
  .....
  if (m < 0) return
  .....
end subroutine subr

```

この例では、 $m < 0$ のときにサブルーチンの処理が終了し、コールしたルーチンに戻ります。ここで `return` 文の代わりに `stop` 文を用いれば、プログラム自体が終了します。

3.3 ローカル変数と引数

サブルーチン内部で宣言した変数や配列は、メインプログラムや他のサブルーチンから独立しています。すなわち、同じ名前を使っても全く別の変数です。例えば、

```

program stest1
  implicit none
  real x,y
  x = 10.0
  y = 30.0
  call subr1
end program stest1

subroutine subr1
  implicit none
  real x,y
  print *,x,y
end subroutine subr1

```

と書いても、サブルーチン `subr1` 中の `print` 文によって出力される `x` と `y` はメインプログラムで代入した 10.0 と 30.0 ではなく、全く無関係な数字です。逆に言えば、他のルーチンでの宣言を気にせずに変数や配列の名前を決めることができます。このルーチン内部でのみ有効な変数を“ローカル変数”といいます。

上記のプログラムを期待通り働かせるために、コール側ルーチンの数値をサブルーチンに伝えるのが引数です。例えば、上記のプログラムを次の `subr2` のように書き直せば、サブルーチン中の `print` 文で出力される `x` と `y` は、メインプログラムと同じ 10.0 と 30.0 になります。

```

program stest2
  implicit none
  real x,y
  x = 10.0
  y = 30.0
  call subr2(x,y)
end program stest2

subroutine subr2(x,y)
  implicit none
  real x,y
  print *,x,y
end subroutine subr2

```

引数は、数値の受け渡しをするための窓口に過ぎないので、コール側の引数とサブルーチン側の引数の変数名を同じにする必要はありません。次のサブルーチンに置きかえても全く同じ動作をします。

```

subroutine subr2(a,b)
  implicit none
  real a,b
  print *,a,b
end subroutine subr2

```

外部からの影響を受けたり、外部に影響を与える、という性質を持つことを除けば、引数もローカル変数です。すなわち、コールしたルーチンからその詳細は見えません。見えるのは、引数という窓口の並びだけです。このため、引数ありサブルーチンを使うときに重要なポイントは、

- (1) 引数の数
- (2) 対応する引数の型

が call 文と subroutine 文とで一致していなければならないことです。例えば、

```

program stest3
  implicit none
  real z
  integer n
  z = 200.0
  n = 21
  call subr3(10.0,z**2,100,n*5+1)
end program stest3

subroutine subr3(x,y,m,n)
  implicit none
  real x,y
  integer m,n
  print *,x,y,m,n
end subroutine subr3

```

というプログラムでは、表 3.1 のような対応になっています。

表 3.1 サブルーチンの引数対応

call 文	subroutine 文	数値型
10.0	x	実数
z**2	y	実数
100	m	整数
n*5+1	n	整数

ここで注意すべきなのは、第 1 引数の x は実数型なので実定数 10.0 を与え、第 3 引数の m は整数型なので整数 100 を与えていることです。もし、第 1 引数に同じ意味だろうと思って 10 という整数を与えると、実行時エラーになります。

サブルーチン内部の実行文で、引数に数値を代入すると、call 文の対応する引数に与えた変数にその数値が代入されます。例えば、

```

program stest4
  implicit none
  real x,y,p
  x = 10.0
  y = 30.0
  call subr4(x+y,20.0,p)
  print *,x,y,p
end program stest4

```

```

subroutine subr4(x,y,z)
  implicit none
  real x,y,z
  z = x*y
end subroutine subr4

```

と書くと、サブルーチン subr4 中の x はコール側の x+y, すなわち 40.0 であり、サブルーチン中の y はコール側の 20.0 なので、サブルーチン中の z には x*y の計算結果である 800.0 が代入されます。このとき、z が引数なので、call 文の対応する位置にある変数 p に 800.0 が代入され、call 文の次の print 文では x, y, p として、10.0, 30.0, 800.0 が出力されます。

このようにサブルーチンの引数に数値を代入することで、call 文の引数変数に代入される数値を“戻り値”といいます。戻り値を使えば、サブルーチンの動作で得られた結果をコール側で受け取ることができます。ただし、call 文の引数には定数を与えたり計算式を書いてもよい、と述べましたが、戻り値を指定する引数は別で、必ず変数か配列にしなければなりません。理由を次節で説明します。

3.4 間接アドレスを用いたルーチン間におけるデータの受け渡し

ルーチン間のつながりをもう少し詳しく考えてみましょう。各ルーチンは基本的にプログラム領域とデータ領域から構成されています。プログラム領域とは文字通りプログラム命令が記録されている領域のことであり、データ領域とは変数や配列のために用意されたメモリ領域のことです。プログラムの際には、変数や配列の宣言文がデータ領域の指定を意味しています。




図 3.2 プログラム領域とデータ領域

変数や配列が各ルーチンごとに宣言されなければならないのは、データ領域が各ルーチンそれぞれに付属しているからです。このため、図 3.2 のように異なるルーチンで同じ名前の変数を宣言しても、それらは別のデータ領域に所属しています。これがローカル変数です。ローカル変数は、そのルーチン内部からしか参照することができません。

このため、あるルーチンで計算した数値を別のルーチンで使うには特別な手続きが必要になります。これが引数です。引数を使えば、コール側のルーチンとサブルーチン間でデータをやりとることができます。では、具体的にどうやってデータの受け渡しをしているのでしょうか。

データをサブルーチンに渡す手段として、もっとも単純に考えられるのは、引数の数値を直接サブルーチンの変数に代入することです。図 3.3 を見て下さい。図のように、コール側の引数 x と y の内容 (こ

の例では 5.0 と 100.0) が、サブルーチンの引数として宣言された変数、a と b に代入されています。この方式は、引数の値を直接渡して呼び出す、という意味で “call by value” と呼ばれています。“call by value” は、コール側からサブルーチン側への値の引渡しについては問題ありません。C 言語はこの方式を採用しています。




図 3.3 “call by value”を使ったデータ渡しでは値を送り返さない

しかし、この方式では、サブルーチン側で作成したデータをコール側に戻すことはできません。サブルーチンは、コール側から送られてきた“値”はわかるのですが、それ以上の情報がないので、コール側のどこにデータを代入すればいいかわからないからです。図 3.3 のように、サブルーチン中の代入文、 $b = a + 25.0$ 、でサブルーチンの変数 b に 30.0 という値を代入しても、サブルーチンのデータ領域の b に代入されるだけで、メインプログラムのデータ領域には影響がありません。

この問題を解決するために、Fortran は “call by value” ではなく、“call by reference” という方式を採用しています。“call by reference” とは、変数の内容ではなく、変数の存在場所 (メモリアドレス) をサブルーチンに伝えて呼び出す方式です。

コンピュータのメモリには、“番地”と呼ばれる通し番号がついています。メモリアドレス (あるいは単にアドレス) とはこの番地のことです。プログラムの実行中に、あるメモリのデータを読み書きするときは、そのメモリのアドレスを指定して行います。例えば、簡単な代入文、

```
a = 1500.0
b = a
```

を考えてみましょう。この結果、b のメモリに 1500.0 という値が書き込まれますが、代入式 $b = a$ において、a (アドレスを [103] とする) という変数からデータを取って来て b に代入するという流れは、

```
a のアドレス指定 → [103] 番地のメモリ内のデータ (1500.0) の呼び出し
                  → b に代入
```

となります。流れを図に描けば図 3.4 のようになります。このような指定したアドレスのメモリに入っているデータを読み書きする方式を “直接アドレス” といいます。

これに対し、あるアドレス (AD_1) のメモリに入っているのが別のメモリのアドレス (AD_2) で、アドレス AD_1 を指定して、アドレス AD_2 のメモリに入っているデータを読み書きする方式を “間接アドレス” といいます。Fortran では、“call by reference” でサブルーチンを呼び出すので、引数にはコール側ルーチンのメモリアドレスが入っています。このため、サブルーチン内で引数の値を読み書きするとき



図 3.4 直接アドレスによるメモリ参照

は間接アドレスを使います¹¹。例えば、

```

program stest3
  implicit none
  real a
  a = 1500.0
  call subr(a)
end program stest3

subroutine subr(s)
  implicit none
  real b,s
  b = s
end subroutine subr

```

というプログラムを考えてみましょう。サブルーチン `subr` 中の代入文 `b=s` によって、変数 `b` のメモリの値は 1500.0 になりますが、引数変数 `s` には `a` の内容である 1500.0 ではなく、メインプログラムの変数 `a` のアドレス (103) が入っています。このため、`b=s` という代入文で、変数 `s` (メモリアドレスを [106] とする) からデータを取ってきて変数 `b` に代入する流れは、

```

s のアドレス指定 → [106] 番地にあるメモリアドレスデータ (103) の呼び出し
                  → [103] 番地にあるデータ (1500.0) の呼び出し
                  → b に代入

```

となります。図示すれば図 3.5 のようになります。




図 3.5 間接アドレスによるメモリ参照

間接アドレスによる読み書きは直接アドレスよりも時間がかかります。よって、`b=s` の代入文において、代入する値を保持している右辺の変数 `s` が間接アドレス指定の場合にはメリットがありません。しかし、代入される左辺の変数 `b` が間接アドレス指定の場合には、この仕組みを利用することで、コール側ルーチンの変数にサブルーチン側のデータを代入することができます。

¹¹ 配列を実現するのもにも間接アドレスが使われています。例えば、`a(10)` は `a(1)` から数えて 10 番目のメモリです。よって、配列の先頭要素 `a(1)` のアドレスをメモリに記録し、それに 9 (=10 - 1) を加えて間接アドレスを用いれば、`a(10)` のメモリを読み書きすることになります。




図 3.6 Fortran では “call by reference” によりデータを送り返すことができる

例えば、図 3.6 のようなプログラムを考えます。サブルーチンの引数変数 b に $a+25.0$ の結果である 30.0 を代入すると、 b にはメインプログラムの変数 y のアドレスが入っているので、間接アドレス指定により y に 30.0 が代入されます。

戻り値を利用するときは、対応する引数にコール側変数のアドレスを与えなければなりません。これが変数または配列を指定しなければならない理由です。戻り値指定の引数に定数や計算式を与えると実行時エラーになります。

3.5 配列を引数にする場合

配列を `call` 文の引数にするときは、配列名を引数にすることも、配列要素を引数にすることも可能です。配列名は配列の先頭要素アドレスを示すので、配列名を引数にすることと、その配列の第 1 要素を引数にすることは同じ意味になります。例えば、`real a(10)` と宣言した 1 次元配列に対して、

```
call sub(a) と call sub(a(1))
```

は同じ動作をします。

これを受けるサブルーチンでの宣言は、サブルーチン内部でそのデータをどう使うかで決めます。一例を示します。

```
subroutine sub(x)
  implicit none
  real x           ! 単一変数宣言
  x = 10.0
end subroutine sub
```

この例では引数 x が単一変数として扱われています。このため、`call` 文で指定した `a(1)` だけがサブルーチン内部で利用でき、他の要素は使えません。これは、`call sub(a)` のように、配列名を与えた場合でも同じです。

これに対し、

```

subroutine sub(x)
  implicit none
  real x(10)           ! 配列宣言
  integer i
  do i = 1, 10
    x(i) = i
  enddo
end subroutine sub

```

のように引数 x を配列宣言すれば、あたかもコール側ルーチンの a という配列がサブルーチン内部の配列 x になったかのように取り扱うことができます。ここで“あたかも”という言葉を使ったのは、サブルーチンでの宣言文の書き方によってはそうならないことがあるからです。例えば、サブルーチンでの配列宣言 $x(10)$ の要素数 10 はコール側と同じ数にする必要はありません。 $x(100)$ のように大きくても、 $x(1)$ のように小さくてもかまわないのです。これは、引数 x が配列の先頭要素アドレスを保持している 1 個の変数にすぎないからです。サブルーチンでの引数配列の宣言は、配列の形状 (次元や下限値) を明示するためのダミーにすぎません。

このため、コール側ルーチンで宣言した要素数とサブルーチンで宣言した要素数を一致させる必要はありません¹²。与えられた配列をどう宣言するかは、サブルーチン内部の計算の都合に合わせて、サブルーチン側で決めることができます。よって、汎用性のあるサブルーチンにするには、配列のアドレス以外に、配列の要素数も引数にしてサブルーチンに伝える必要があります。

配列の形状さえわかればいいのですから、サブルーチンで引数配列を宣言するときは、数字の代わりに“*”を書くことができます。例として、要素数 n の 1 次元配列 a のデータを、同じ長さの 1 次元配列 b にコピーするプログラムを考えてみましょう。最も単純な答えは、以下のようなものです。

```

subroutine copy(a,b,n)
  implicit none
  real a(*),b(*)
  integer n,i
  do i = 1, n
    b(i) = a(i)
  enddo
end subroutine copy

```

このように、サブルーチンの引数配列 a や b の宣言を“*”にすることで、要素数が不定の 1 次元配列であることを明示しています¹³。

“*”による宣言は 2 次元以上の配列でも使うことができますが、制限があります。例えば、 $a(*,*)$ という形の宣言はできません。なぜなら、2 番目の要素を進めるのは 1 番目の要素が最大値、すなわち宣言した上限数に達したときなので、1 番目の要素が不定では情報不足で進められないからです。このため、“*”が使えるのは、一番右側の要素数のみです¹⁴。例えば、 $\text{real } a(3,*)$ と宣言すると、第 1 要素数が 3 の 2 次元配列として計算されます。

しかし、これではどんな 2 次元配列でも使えるサブルーチンを作ることはできません。そこで、“整合配列”と呼ばれる機能が用意されています。整合配列とは subroutine 文の引数の中にある整数型変数を利用して宣言した形の引数配列のことです。例えば、以下のように宣言することができます。

¹²要素数だけでなく、次元さえ一致させる必要はありません。コール側が 2 次元配列で、サブルーチンでの宣言が 1 次元配列でも動作は可能です。例えば、上記のサブルーチン `copy` は、2 次元配列でも使用できます。これは、1.3.2 節で述べたように 2 次元配列もメモリ上では 1 次元的に並んでいるからです。ただし、要素数を与える引数 n には全要素数を指定する必要があります。例えば、 $a(10,10)$ と宣言された配列ならば、 n に 100 ($=10 \times 10$) を与えます。

¹³なお、“*”だけ書くと配列の下限は 1 です。もし下限を変更したいときには下限を別途指定します。例えば、配列 a の下限を 0 にする場合は、 $\text{real } a(0:*)$ と宣言します。

¹⁴1.3.2 節の配列の順番を与える公式が一番右側の要素数に依存しないことを思い出して下さい。

```

subroutine copy2d(a,b,m,n)
  implicit none
  real a(m,n),b(m,n)
  integer m,n,i,j
  do j = 1, n
    do i = 1, m
      b(i,j) = a(i,j)
    enddo
  enddo
end subroutine copy2d

```

この例では、 m と n が引数の中にあるので、これらを使って引数の配列 a と b を宣言することができるのです。 $a(m,n)$ のような単純な宣言だけではなく、 $a(0:2*m,n-1)$ のような、下限指定や計算式を使った宣言も可能です。

このサブルーチンの使用例を次に示します。 サブルーチンの引数 a , b にどんな要素数の2次元配列を与えても、コール側ルーチンにおける配列宣言と同じ要素数を m , n に与えれば、正しく動作します。

```

program stest1
  implicit none
  real a(10,20),b(10,20),c(100,200),d(100,200)
  .....
  call copy2d(a,b,10,20)
  call copy2d(c,d,100,200)
end program stest1

```

なお、配列宣言の際には、コロンを付加して下限を指定することができますが、これをサブルーチンに引き継ぐにはサブルーチン側でも同じ下限指定をする必要があります。例えば、

```

program stest2
  implicit none
  real a(-1:100)
  call sub(a,a,100)
end program stest2

subroutine sub(x,y,n)
  implicit none
  real x(-1:n),y(*)
  integer n
  x(10) = 10.0
  y(10) = 10.0
end subroutine sub

```

と書いた場合、メインプログラムと同じ下限指定をした配列 x では $x(10)$ がメインプログラムの $a(10)$ に対応するので、 $a(10)$ に 10.0 が代入されます。しかし、下限指定をしていない配列 y では下限が 1 ですから、 $a(-1)$ から数えて 10 番目の $a(8)$ に 10.0 が代入されます。任意の下限を持つ配列に対して動作するプログラムにするには、下限の数値も引数にして整合配列を使います。

3.6 関数副プログラム

$\sin(x)$ のように、引数を使って内部で計算した結果を計算式中で使うことができる“関数”を自作することもできます。これを“関数副プログラム”といいます。関数副プログラムは、サブルーチンとほとんど同じ構造をしていますが、以下のような違いがあります。

- (1) subroutine の代わりに function を書く
- (2) 関数名を変数として宣言し、それに計算結果 (関数値) を代入しなければならない

これ以外は、引数ありサブルーチンと同じです。引数に関する注意もサブルーチンと同じで、戻り値を与える引数を含めることもできます。関数副プログラムとは、引数以外に戻り値を1個持つサブルーチンの変種だといえます。この戻り値が関数値になります。

関数副プログラムの一例を示します。

```
function square(x)
  implicit none
  real square,x           ! 関数名の型宣言
  square = x*x           ! 関数名に値を代入する
end function square
```

このように、function文で開始し、end function文で終了します。また、関数名 square を実数型宣言し、引数 x の2乗を代入して終了しています。すなわち、この例は引数 x に実数型の数値を与えると、 x^2 を関数値として与える関数になります。

関数副プログラムを使うことは、“関数名”という変数を使うことであると考えます。このため、作成した関数を使用するルーチンでも関数名を型宣言する必要があります。例えば、上例の square を使うには、

```
program ftest1
  implicit none
  real x,y,square        ! 使用する関数名を型宣言する
  x = 5.2
  y = 3.0*square(x+1.0) + 50.5
  print *,x,y
end program ftest1
```

のように、square という関数名を関数副プログラムでの宣言と同じ型で宣言しておかなければなりません。関数副プログラム中で宣言した関数名の型と、その関数を使用するルーチンで宣言した関数名の型が異なる場合には実行時エラーになります。

3.7 モジュールを使ったグローバル変数の利用

ローカル変数のおかげで各ルーチンの独立性は保たれますが、引数でしかルーチン間のデータ受け渡しができないのは不便です。引数は、かっこを使った並びで指定するので、受け渡す変数が多いと書くのが大変です。しかも並びが重要なので、一カ所順番を間違えただけでもエラーが発生します。そもそも引数による受け渡しはアドレス情報を経由したもので、コール側とサブルーチン側で完全に同じ変数であるという保証はありません。

3.1節で、計算機シミュレーションでは計算内容に応じてサブルーチンを作成すると述べましたが、この用途においては、ルーチン間で共用する変数が多数必要です。しかし、共用変数を全て引数にするのは効率が悪く、エラーも発生しやすくなります。

そこで、ルーチン内部でのみ意味を持つ“ローカル変数”に対して、“グローバル変数”が用意されています。グローバル変数とは、どのルーチンから参照しても共通した値を保持している変数のことで、Fortranでは、モジュールと use 文の組み合わせで利用可能です¹⁵。

モジュールとは、変数やサブルーチンなどを集めて一つのパッケージにしたもので、module 文で開始して end module 文で終了します。サブルーチンと同じ構造をしていることからわかるように、モジュールの記述は、メインプログラムやサブルーチンと同レベルです。例えば、

¹⁵モジュールや use 文は Fortran90 から採用された比較的新しい仕様です。それ以前では common 文でグローバル変数を実現していました。しかし、common 文には不便な点が多いので本書では説明を省略します。

```

module data1
  integer nmin,nmax
  real tinitial,amatrix(20,30)
end module data1

```

のように書きます。先頭の module 文で、module の後に指定した文字の並び (この例では data1) を “モジュール名” といいます。また、最後の end module 文には module 文と同じモジュール名を指定します。サブルーチンや関数副プログラムと異なり、モジュールはプログラムに記述しただけでは利用することができません。利用するルーチンの先頭に、use 文を使ってモジュール名を指定し、その利用を宣言する必要があります。use 文は以下のような形式です。

```
use モジュール名
```

use 文は implicit 文よりも前に書かなければなりません。モジュールを使ったプログラムの一例を以下に示します。

```

module global
  real xaxis,yaxis
end module global

program stest4
  use global
  implicit none
  xaxis = 5.0
  yaxis = 100.0
  call subr4
  print *,xaxis,yaxis
end program stest4

subroutine subr4
  use global
  implicit none
  print *, xaxis,yaxis
  yaxis = 25.0
end subroutine subr4

```

モジュールの中で宣言された変数や配列は、メインプログラムやサブルーチンとは独立して存在したデータ領域にあり、use 文で利用を宣言すれば、どのルーチンからでも参照することができます。このプログラムのメモリイメージを図示すれば、図 3.7 のようになります。

モジュールは、名前が異なれば、一つのプログラムに複数作成することも可能であり、一つのルーチンが複数のモジュールを利用することも可能です。ただし、モジュールは use 文で利用を宣言するより前で (プログラムの上方で) 定義されている必要があります。このため、通常はこの例のように全てのルーチンより前に記述しておきます。

モジュールで宣言されている変数は、use 文を書くだけで利用できるという利便性がありますが、ルーチン内で明示的に宣言されていないので、不用意に使う可能性があります。そこで、次のような only 句を使って、ルーチン内で必要な変数だけに宣言を限定することができます。

```
use モジュール名, only : 変数 1, 変数 2, ...
```

例えば、前の例で、

```
use global, only : yaxsis
```

のように書くと、yaxsis 以外の変数 (xaxis) は、このルーチンでは宣言されていないのと同様です。宣言されていない変数は、ローカル変数として別途宣言することも可能です。




図 3.7 モジュールで宣言されているグローバル変数の参照

グローバル変数は、多用するとルーチンの独立性が失われてしまいます。基本的にはローカル変数でプログラムを作り、ルーチン間で共有する必要がある変数のみグローバルにしてください。また、グローバル変数は広い範囲で存在する可能性があるため、意味のある長めの名前を付けます。これは、1.2.3 節で述べた“大域的に有効な変数名の付け方”です。

3.8 サブルーチンや関数副プログラムを引数にする

ここまで説明したサブルーチンや関数副プログラムの引数は変数や配列でしたが、サブルーチンや関数副プログラムの名前を引数に持つサブルーチンや関数副プログラムを作ること可能です¹⁶。例えば、方程式 $f(x) = 0$ を解くサブルーチンにおいて、方程式を与える関数副プログラムの名前も引数に加えておけば、コールするときにユーザーが関数を指定することができる、より汎用的なサブルーチンになります。

サブルーチン名を引数にしたサブルーチンを作るのは簡単で、引数並びの中にサブルーチン名を入れるだけです。例えば、

```
subroutine subrout(subr,xmin,xmax,n)
  implicit none
  real xmin,xmax,dx,y
  integer n,i
  dx = (xmax-xmin)/n
  do i = 0, n
    call subr(dx*i+xmin,y)      ! 引数 subr を使った call 文
    print *,i,y
  enddo
end subroutine subrout
```

のように書くことができます。この例では、subr が引数としてのサブルーチン名で、これをコールする文を書くことができます。関数副プログラムを引数にする場合も同様ですが、関数名の型宣言は必要です。例えば、次のように書きます。

¹⁶ここではサブルーチンと関数副プログラムを合わせて「外部副プログラム」と呼びます。


```

subroutine funcout(func,xmin,xmax,n)
  implicit none
  real func,xmin,xmax,x,dx,y      ! 引数 func が関数なので型宣言をする
  integer n,i
  dx = (xmax-xmin)/n
  do i = 0, n
    x = dx*i + xmin
    y = func(x)**3                ! 引数 func を関数として使う
    print *,x,y
  enddo
end subroutine funcout

```

このように、外部副プログラムを引数に持つサブルーチンを作るのは簡単ですが、このサブルーチンを使用するときは、「引数に変数ではなく、外部副プログラムである」という宣言が必要です。この宣言は external 文で行います。例えば、

```

subroutine sub(x,y)
  implicit none
  real y,x
  y = 2*sin(x) + cos(x**2)
end subroutine sub

function fun(x)
  implicit none
  real fun,x
  fun = sin(x)**3
end function fun

```

というサブルーチンや関数副プログラムを引数にして、上記のサブルーチン (subrout や funcout) をコールするときは、

```

program test_func
  implicit none
  external fun,sub                ! external 文を使って宣言する
  call subrout(sub,0.0,3.0,10)
  call funcout(fun,0.0,3.0,10)
end program test_func

```

のように書きます。external 文による宣言がないと、sub や fun という単語が「変数」と見なされて、型宣言されていないというエラーになります。external 文は非実行文なので、変数の型宣言と同様に、全ての実行文より前に書く必要があります。

さて、なぜこの程度で外部副プログラムを引数にできるのか、しくみを簡単に説明しましょう。3.4 節で述べたように、Fortran では引数に変数を与えるとき、変数の内容を与えるのではなく、そのアドレスを与えます。変数にアドレスがあるのと同様に、プログラムにもアドレスが付いています。これは、プログラムが保存されているメモリ番地のことです。goto 文での無条件ジャンプや if 文での分岐、サブルーチンコールなど、プログラムを 1 行ずつ実行するのではなく、必要に応じて特定の場所に実行を移すときには、プログラムのアドレスを指定してジャンプします。プログラムも変数も場所は違いますが、コンピュータのメモリ上にあるという意味では同じです。よって、引数に変数を与えるのも、サブルーチン名を与えるのも、メモリアドレスを与えるという意味では同じであり、特に書式を変更する必要はありません。

しかし、プログラム名が指定するアドレスと変数や配列名が指定するアドレスは指定方法が異なるので区別しなければなりません。外部副プログラムを引数として持つサブルーチンの内部では、その副プログラムを使用している記述があるはずなので、それを判定材料として引数がプログラムアドレスか変数アドレスかを区別しています。一方、このサブルーチンを使う call 文側では、外部副プログラムの

名前しか引数に書けないので、それだけで変数か外部副プログラム名かを判定することはできません。この区別のために用意されているのが `external` 文です。 `external` 文を使って宣言された名前は、サブルーチン名または関数副プログラム名と判定されてコンパイルされます。なお、 `external` 文による宣言を忘れてコンパイルエラーになったとき、エラーを防ぐために `sub` や `fun` を `real` や `integer` などで型宣言すると、エラーは出ず、コンパイルもリンクも成功します。しかし、実行しても正常な動作はしないので注意が必要です。

なお、関数を引数に持つサブルーチンに `sin` や `exp` のような組み込み関数名を与えてコールするときは、 `external` 文の代わりに `intrinsic` 文で宣言します。例えば、

```
program test_sin
  implicit none
  intrinsic sin
  call funcout(sin,0.0,3.0,10)
end program test_sin
```

のように書きます。これは、Fortranにおける組み込み関数が宣言をしなくても使える特別なものだからです。しかし、それなら宣言せずとも使えるようにしてくれればいいと思うのですが、 `intrinsic` 宣言をしておかないと「`sin` という変数が宣言されていない」というエラーになります。このあたりはFortranコンパイラの仕様の問題だと思います。

3.9 プログラミング スマートテクニック (その3)

サブルーチンを利用するときは、変数の使い分けが一つのポイントです。これまでローカル変数とグローバル変数の使い分けについて説明しましたが、ローカル変数にも2種類あり、これを使い分ければサブルーチンの動作をより細かく制御することができます。ここでは、ローカル変数の使い分け方や、プログラムのメンテナンスをさらに容易にするための文法などを紹介します。

3.9.1 拡張宣言文とそれを用いたローカル変数の使い分け

これまで変数や配列の宣言に使っていた宣言文は、

```
型指定 変数 1, 変数 2, ...
```

という形式でした。ここで、“型指定”には“`integer`”とか、“`real`”のような数値型を記述し、“変数”には変数名か、配列名とその要素数を記述します。この宣言文は、データ領域におけるメモリの確保という意味もあります。

さて、Fortran90から宣言文の記述形式が拡張されました¹⁷。拡張宣言文を使うと、メモリを確保すると同時に初期値を代入したり、メモリの特性を指定することができます。拡張宣言文は以下の形式です。

```
型指定 [, 属性 1, 属性 2, ...] :: 変数 1 [=数値 1] , 変数 2 [=数値 2], ...
```

ここで、角かっこ、“`[`”と“`]`”は、この中が省略可能であるという意味で使っているだけなので、角かっこ自体は書かないで下さい。拡張宣言文を使用するときは、型指定部と宣言する変数や配列の間にコロンの2個“`::`”を書く必要があります。また、“属性”には宣言する変数の特性を指定するための予約語を記述します。属性も数値も省略して、単に型指定と変数の間に“`::`”を書いただけの宣言文は、書かずに宣言した旧来の書式と同じ意味になります。

属性を書かずに、単に数値を代入した形で宣言すると、その変数に指定した数値を代入して、プログラムの動作が開始することを意味します。例えば、

¹⁷本書では、拡張された宣言文のことを“拡張宣言文”と呼んで、旧来の型指定だけの宣言文と区別します。

```
integer :: imax=10, jmax=100
real :: xx=1.0, yy=2.0
```

のように宣言すると、それぞれの変数に指定された値を代入した状態で動作が開始します。ただし、この宣言文中での数値代入は一度だけです。サブルーチン中で数値代入した変数を宣言しても、コールするたびに数値が代入されるのではありません。このため、その変数を実行文で変更すると、その変更した結果が残ります。例えば、

```
program stest1
  implicit none
  call subr1
  call subr1
  call subr1
end program stest1

subroutine subr1
  implicit none
  integer :: n=1
  print *,n           ! コールするたびに n は増加する
  n = n + 1
end subroutine subr1
```

というプログラムでは、メインプログラムでサブルーチン `subr1` が3回コールされていますが、サブルーチン中の `print` 文の出力は、1回目が1、2回目が2、3回目が3、となります。宣言文における代入は、プログラム開始時の初期値だと考えて下さい。

さて、上記のサブルーチンの動作を考えると、一つ注意しなければならないことがあります。3.4節で、ローカル変数は各ルーチンのデータ領域に所属しているという説明をしましたが、もう少し詳しく言うと、サブルーチンのデータ領域には2種類あります。一つは、サブルーチンがコールされた時点で一時的に生成されるメモリ領域で、もう一つは、サブルーチンの呼び出しに関係なく常に確保されたメモリ領域です。本書では、前者を“一時メモリ領域”と呼び、後者を“固定メモリ領域”と呼ぶことにします。旧来の宣言文で宣言したローカル変数は、一時メモリ領域に所属します¹⁸。一時メモリ領域は、サブルーチンを呼び出すたびに場所や内容が変わる可能性があるため、同じサブルーチンを2回コールした場合、1回目のコールでローカル変数に代入した値が、2回目にコールした時点でそのまま残っているとは限りません。

例えば、

```
program stest2
  implicit none
  call subr2(1)
  call subr2(2)
end program stest2

subroutine subr2(n)
  implicit none
  integer n
  real x,y
  if (n == 2) print *,x,y           ! この出力値は不定
  x = 10.0
  y = 100.0
end subroutine subr2
```

のようなプログラムを書いたとします。1回目の `call` 文、`call subr2(1)` で、ローカル変数 `x` と `y` に、それぞれ10.0と100.0という値が代入されますが、2回目の `call` 文、`call subr2(2)` を実行したとき

¹⁸これに対し、メインプログラムの変数とモジュール内のグローバル変数は、宣言文の記述にかかわらず、全て固定メモリ領域に所属します。

に、`print` 文で出力した `x` と `y` が 10.0 と 100.0 になる保証はないのです¹⁹。ローカル変数に代入した値を保証するには、固定メモリ領域に所属させなければなりません。拡張宣言文を使って初期値を代入したローカル変数は、一時メモリ領域ではなく、固定メモリ領域に所属します。最初のサブルーチン例、`subr1` の変数 `n` が、コールするごとに 1 ずつ増加した値を出力するのは、初期値 1 を代入して宣言することで固定メモリ領域に所属させたためです。

数値を代入せずに、ローカル変数を固定メモリ領域に所属させるには、変数に `save` 属性を指定します。例えば、上記のサブルーチン `subr2` を以下のように書き換えれば、2 回目のコールで 10.0 と 100.0 が出力されます。

```
subroutine subr2(n)
  implicit none
  integer n
  real, save :: x,y           ! save 属性を指定
  if (n == 2) print *,x,y
  x = 10.0
  y = 100.0
end subroutine subr2
```

変数の初期値が未定のときや、ローカル配列を固定メモリ領域に所属させたいときは、`save` 属性を使って下さい。また、初期値を代入する場合でも、固定メモリ領域に所属させることを積極的に利用するのであれば、次のように `save` 属性を付加して、固定メモリ領域にあることを明示した方が良いでしょう。

```
integer, save :: n=1
```

拡張宣言文の属性には `save` の他にも色々ありますが、ここでは配列を宣言するときに便利な `dimension` 属性を紹介しましょう。例えば、10×10 の 2 次元実数型配列を 5 個用意するとき、通常の宣言では、

```
real a(10,10), b(10,10), c(10,10), d(10,10), e(10,10)
```

のように書きますが、`dimension` 属性を使えば、これを次のように書くことができます。

```
real, dimension(10,10) :: a, b, c, d, e
```

すなわち、`dimension` 属性に配列の形状 (次元や要素数) を書いておけば、宣言する変数の位置には、配列名を記述するだけになります。属性は複数付加することができるので、`dimension` 属性に `save` 属性を加えて宣言することも可能です。

3.9.2 parameter 変数の利用

配列を宣言するときに便利なのが、`parameter` 属性を指定した変数、`parameter` 変数です。`parameter` 変数には必ず値を代入しなければなりません。例えば、

```
integer, parameter :: imax=10, jmax=200
```

のように書くと、`imax` と `jmax` が `parameter` 変数になります。

さて、配列を宣言するとき、その要素数は整数定数で指定しなければなりません。このため、配列の長さを変更するときには、宣言した要素数に関連した数値を全て変更する必要があります。手間がかかるだけでなく、見落とししたり、書き間違える可能性があります。例えば、

¹⁹ コンパイラによっては、全てのローカル変数を固定メモリ領域に所属させるものがあり、このプログラムでもうまくいく場合があります。しかし、文法的には保証されていないのですから、当てにするべきではありません。

```

real a(100),b(100)           ! 100 が 2 カ所
integer i
do i = 1, 100               ! 100 が 1 カ所
  a(i) = i*i
  b(i) = a(i)**2
enddo

```

というプログラムにおいて、配列要素数の 100 は、宣言文だけでなく、do 文の終了値にも入っています。このため、配列要素数を変更するときは 3 カ所変更しなければなりません。このとき、配列宣言の要素数を parameter 変数に代入する形で用意しておく、配列要素数をその parameter 変数で置き換えることができます。例えば、上のプログラムは次のように書き換えることができます。

```

integer, parameter :: imax=100
real a(imax),b(imax)
integer i
do i = 1, imax
  a(i) = i*i
  b(i) = a(i)**2
enddo

```

このプログラムならば、imax に代入する数値を変更するだけで、配列要素数の変更は完了です。

parameter 変数を含んだ計算式を別の parameter 変数への代入値にしたり、配列宣言の要素数に使うこともできます。これらは、その計算式の結果で宣言したことに相当します。例えば、

```

integer, parameter :: imax=100, imax2=imax**2
real a(imax-1),b(0:imax*2),c(-imax2:imax2)

```

という宣言は、以下の宣言文と同等です。

```

integer, parameter :: imax=100, imax2=10000
real a(99),b(0:200),c(-10000:10000)

```

ただし、他の parameter 変数に代入するときは、代入される変数 (この例では imax2) が、代入する計算式に使う変数 (この例では imax) より後で宣言されなければなりません。

実は、parameter 属性を指定した変数は、“変数=数値”の形式で変数名と数値の対応関係を示しているだけで、メモリとしての実体はありません。この対応関係を使った“変数”から“数値”への置き換えは、コンパイルの段階で行われるので、非実行文である宣言文でも使えるのです。逆に言えば、parameter 変数に実行文を使って代入することはできません。例えば、

```

integer, parameter :: imax=100
real a(imax),b(imax)
imax = 5           ! これはエラー

```

と書いた場合、“imax=5”はコンパイルエラーです。これは、プログラムに“100=5”と書いたことに等しいのですから当然だと言えます。

このように、parameter 変数を使えば配列要素数の変更が楽になりますが、それでもルーチンごとのローカルな宣言では、変更が必要になったときに使用している全てのルーチンで parameter 変数の設定値を変更しなければなりません。そこで、

```

module global_param
  integer, parameter :: imax=300, jmax=200
end module global_param

```

のように、parameter 変数を記述したモジュールを用意して、共有するルーチンで use 宣言をするのが良いと思います。use 宣言はモジュールの中でも使うことができるので、以下のように記述することが

可能です。

```
module global_param
  integer, parameter :: imax=300, jmax=200
end module global_param

module mod_array
  use global_param           ! モジュール内でも use 宣言可能
  real abc(imax,jmax),cd(jmax*2-1) ! グローバル配列宣言
end module mod_array

program mtest1
  use mod_array              ! use global_param は不要
  implicit none
  integer km(imax)          ! ローカル配列宣言
  .....
```

ここで、あるモジュールの中で別のモジュールを use 宣言した場合、前者のモジュールを use 宣言したルーチンでは、後者のモジュールも合わせて use 宣言したことになります。上記のメインプログラム mtest1 ではモジュール mod_array しか use 宣言をしていませんが、mod_array の中で use 宣言をしているモジュール global_param も同時に宣言したことになり、その中にある変数 imax もメインプログラムで利用可能になります。

ただし、次のように、メインプログラムに global_param の use 宣言を加えてその使用を明示してもエラーではありません。

```
program mtest1
  use global_param           ! 書いてもエラーではない
  use mod_array
  implicit none
  integer km(imax)
  .....
```

これは、同じモジュールの use 宣言を重複して書いてもエラーにはならないからです。

parameter 変数は、数学定数や物理定数を用意するのにも便利です。筆者は次のような定数の入ったモジュールを作成して、必要なサブルーチンから利用するようにしています。

```
module constants
  real, parameter :: pi=3.141592653589793, pi2=pi*2
  real, parameter :: clight=2.99792458e8, hplanck=6.62606957e-34
  real, parameter :: emass=9.10938291e-31, pmass=1.672621777e-27
  real, parameter :: echarge=1.602176565e-19, emc2ev=emass*clight**2/echarge
end module constants
```

parameter 変数は不用意に書き換えられることがないので、このような定数の定義にはうってつけです。また、コンパイル時に数値の置き換えをしますので、代入という実行動作が不要になり、わずかながら計算時間の短縮になります。

3.9.3 include 文

さて、プログラムが完成して、後は問題に応じて parameter 変数を変更するだけになると、parameter 変数の変更のためだけにプログラムファイルを修正するのは煩わしいし、変更するときに誤ってプログラムの内容を書き換えてしまうリスクもあります。プログラムを変更すると、プログラムファイルの日付が変更されるので、いつの時点で完成したプログラムかもわからなくなります。これらは、特に他人にプログラムを提供するときに問題になります。

そこで、parameter 変数の部分だけをプログラム本体から分離して保存し、必要に応じて結合すると便利です。結合する一つ的手段に include 文の利用があります。include 文とは、以下のように include の後に、ファイル名を文字列で記述した文です。

```
include 'ファイル名'
```

include 文を書くと、コンパイラはその位置に“ファイル名”で指定したファイルの内容を挿入したプログラムを生成して、それをコンパイルします。例えば、3.9.2 節のモジュール global_param を“global.inc”というファイルに書き込んで保存しておけば、次のように記述することができます。

```
include 'global.inc'           ! ここに global.inc の内容が挿入される

module mod_array
  use global_param
  real abc(imax,jmax),cd(jmax*2-1)
end module mod_array

program mtest1
  use mod_array
  implicit none
  integer km(imax)
  .....
```

include 文が指定するファイルの内容は、それを include 文の位置に挿入したときに、プログラム全体が正しく動作すればいいので、モジュールやサブルーチンのような完結したプログラムである必要はありません。また、同じファイルを一つのプログラムの複数の場所で include 指定することもできます。例えば、

```
integer, parameter :: imax=300, jmax=200
```

だけをファイルに書いて、サブルーチンごとに include 指定することも可能です。

3.9.4 interface 文

Fortran は古くからある言語なので、文法的にゆるい記述がいくつか残っています²⁰。implicit none を付けなければ暗黙の変数宣言と見なされ、宣言しなくても変数が自由に使えるというのはその一つです。これ以外に、サブルーチン呼び出しのための call 文の記述において引数のチェックをしない、という問題があります。このため、call 文で引数の数を間違えたり引数の数値型を間違えたりしてもコンパイラエラーにはならず、実行時に正常な動作をしないという形で不具合が現れます。

この欠点を補うため、最近の Fortran では、interface 文を使ってプログラム中に外部副プログラムの引数の形状を宣言しておき、副プログラムを使用する際の引数チェックをコンパイラにさせることができます。interface 文は、interface と end interface ではさんだブロックで、その中に subroutine 文または function 文、その引数宣言 (関数副プログラムの場合は関数値も)、および最後の end 文だけを取り出して入れておきます。例えば、3.8 節のサブルーチン subrout や関数 fun を使用する場合、以下のように書くことができます。

```
program test_interface
  implicit none
  external sub
  interface
    subroutine subrout(subr,xmin,xmax,n)
      external subr           ! 外部副プログラムは external 文で宣言する
      real xmin,xmax
      integer n
    end subroutine subrout
```

²⁰ キーボードやモニターがなかった時代、プログラムを書くのは時間がかかる作業でした。このため、記述量を減らすために暗黙の宣言などがあったのだと思います。C 言語は比較的新しい言語なので、変数宣言は必ず行わなければならないし、関数の引数チェック機能も含まれています。

```

        function fun(x)
            real fun,x
            end function fun
        end interface
        call subrout(sub,0.0,3.0,10)
!       call subrout(0.0,3.0,10)
        print *,fun(10.0)
!       print *,fun(10)
        end program test_interface

```

! 関数の場合は関数値も宣言する

! これを入れるとエラーになる

! これを入れるとエラーになる

この例のように、interface 文中には複数のサブルーチンや関数副プログラムの宣言を書くことができます。また、interface 文中に関数副プログラムの宣言を入れておくと、関数名の型宣言が入っているので、その関数名の型宣言を別途する必要はなくなります。なお、外部副プログラムを引数にする場合には、この例のように external 文で宣言しておきます。

このプログラム例で、二つ目の subrout の call 文を含めてコンパイルするとエラーになります。なぜなら、二つ目の call 文は引数の数が足らず、一番左の引数が外部副プログラムではなく実数になっているからです。また、二つ目の関数 fun を使用した print 文を含めてもエラーです。これは、x が実数指定なのに整数 10 を与えているからです。

interface 文で引数の型を宣言しておくと、引数変数に値を代入した形でサブルーチンをコールすることができます。この場合、引数を記述する順番は自由です。例えば、

```

program test_interface
    implicit none
    external sub
    integer m
    interface
        subroutine subrout(subr,xmin,xmax,n)
            external subr
            real xmin,xmax
            integer n
        end subroutine subrout
    end interface
    m = 5
    call subrout(n=m**2,subr=sub,xmax=3.0,xmin=0.0)
end program test_interface

```

と書くことができます。

interface 文を使うと、より安全なプログラムになります。大きなプログラムを作る時や、引数が多いサブルーチンを利用するときは、interface 文の利用をお勧めします。必要な部分だけをコピー・ペーストするだけなので、それほど手間はかかりません。

演習問題 3

(3-1) 2次方程式の根 (サブルーチン利用)

3個の実変数 a, b, c を引数とし、それから、

$$ax^2 + bx + c = 0$$

の2根 x_1, x_2 、および判別式 $D = \sqrt{b^2 - 4ac}$ を計算して戻り値にするサブルーチンを作成せよ。ただし、2実根になる場合は、 x_1, x_2 をそれぞれ戻り値変数 $x1$ と $x2$ に代入し、2虚根になる場合は、実部を $x1$ に、虚部を $x2$ に代入するようにせよ。

次にそれを使って問題 (2-6) の解答をサブルーチンを使う形に修正せよ。

(3-2) ヘロンの公式 (サブルーチン利用)

三角形の三辺の長さを a, b, c とすると、その三角形の面積 S は次式 (ヘロンの公式) で表される。

$$S = \sqrt{p(p-a)(p-b)(p-c)}$$

ただし、 $p = \frac{a+b+c}{2}$ である。

a, b, c の数値を引数とし、この公式を使って三角形の面積を計算して戻り値にするサブルーチンを作成せよ。

次にそれを使って問題 (1-2) の解答をサブルーチンを使う形に修正せよ。

(3-3) 回路計算 (サブルーチン利用)

N 個の抵抗 $R_1, R_2, R_3, \dots, R_N$ に対し、この N 個を直列にした時の合成抵抗 R_s と N 個を並列にした時の合成抵抗 R_p を同時に計算して戻り値として返すサブルーチンを作成せよ。ただし、 N 個の抵抗値は、要素数 n の1次元配列 $R(n)$ で表すとする。

次に、それを使って問題 (1-7) の解答をサブルーチンを使う形に修正せよ。

(3-4) 3次元ベクトルと電磁気学 (サブルーチン利用)

3次元ベクトル $\mathbf{A} = (A_1, A_2, A_3)$ を要素数3の1次元配列 $a(3)$ で表すとする。電場ベクトル \mathbf{E} 、磁場ベクトル \mathbf{B} および荷電粒子の速度ベクトル \mathbf{v} を配列で表して、これらの配列を引数として与えると内積 $\mathbf{v} \cdot \mathbf{E}$ 、および外積ベクトル $\mathbf{v} \times \mathbf{B}$ を計算して戻り値とするサブルーチンを作成せよ。なお戻り値の外積ベクトルも配列とする。これを使って問題 (1-8) の解答をサブルーチンを使う形に修正せよ。

(3-5) 統計計算 (サブルーチン利用)

要素数 n の1次元配列、 $a(n)$ を引数とし、

$$\begin{aligned} \text{平均 } \bar{A} &= \frac{1}{n} \sum_{i=1}^n A_i \\ \text{標準偏差 } \sigma &= \sqrt{\frac{1}{n} \sum_{i=1}^n (A_i - \bar{A})^2} \end{aligned}$$

を計算して戻り値とするサブルーチンを作成し、問題 (2-2) の解答をサブルーチンを使う形に修正せよ。

(3-6) 非線形方程式の解法：Newton法 (サブルーチン利用)

問題(2-8)のNewton法を用いて方程式の根を計算するプログラムをサブルーチンを使う形に修正せよ。まず、変数値 x を与えると関数値 $f(x)$ とその微分値 $f'(x)$ を返すようなサブルーチンを作成する。それを使ってNewton法で根を求めるプログラムにすればよい。

(3-7) 行列計算

n 行 n 列の行列 a_{ij} を2次元配列 $a(n,n)$ の要素 $a(i,j)$ で表すとする。2個の行列 a_{ij} と b_{ij} から行列の和, c_{ij} , および行列の積, d_{ij} を計算するサブルーチンを作成せよ。この時、整合配列を用いて与える配列の次元が自由に変えられるようにすること。

ここで、行列の和と積の定義は以下の通りである。

$$c_{ij} = a_{ij} + b_{ij}$$
$$d_{ij} = \sum_{k=1}^n a_{ik} b_{kj}$$

(3-8) 行列式

$u(3)$, $v(3)$, $w(3)$ という3つの1次元配列データから行列式

$$Det = \begin{vmatrix} u(1) & v(1) & w(1) \\ u(2) & v(2) & w(2) \\ u(3) & v(3) & w(3) \end{vmatrix}$$

を計算するサブルーチンを作成せよ。

(3-9) 3元連立方程式の解法

3元1次の連立方程式,

$$a_{11}x_1 + a_{12}x_2 + a_{13}x_3 = b_1$$

$$a_{21}x_1 + a_{22}x_2 + a_{23}x_3 = b_2$$

$$a_{31}x_1 + a_{32}x_2 + a_{33}x_3 = b_3$$

を解くプログラムのサブルーチンを作成せよ。

クラメールの公式によれば、答えは以下のようになる。

$$x_1 = \frac{\begin{vmatrix} b_1 & a_{12} & a_{13} \\ b_2 & a_{22} & a_{23} \\ b_3 & a_{32} & a_{33} \end{vmatrix}}{D}, \quad x_2 = \frac{\begin{vmatrix} a_{11} & b_1 & a_{13} \\ a_{21} & b_2 & a_{23} \\ a_{31} & b_3 & a_{33} \end{vmatrix}}{D}, \quad x_3 = \frac{\begin{vmatrix} a_{11} & a_{12} & b_1 \\ a_{21} & a_{22} & b_2 \\ a_{31} & a_{32} & b_3 \end{vmatrix}}{D}$$

ただし,

$$D = \begin{vmatrix} a_{11} & a_{12} & a_{13} \\ a_{21} & a_{22} & a_{23} \\ a_{31} & a_{32} & a_{33} \end{vmatrix}$$

である。

プログラムは、3行3列の行列 a_{ij} を2次元配列 $a(3,3)$ の要素 $a(i,j)$ で表し、 b_i を $b(i)$ という1次元配列で表し、 x_i を $x(i)$ という1次元配列で表して、サブルーチンの引数は、 a と b と x とする。行列式の計算には、問題(3-8)で作成したサブルーチンを利用すればよい。

(3-10) モンテカルロ法

世の中の動きは、運動方程式のような微分方程式を用いて確実に予測できるとは限らない。例えば、さいころを投げて次にどの数字が出るかで進み方を決めるすごろくのように、次の瞬間に何が起こるかわからない事象を元にして未来の様子を計算する方法をモンテカルロ法という。

計算機でさいころの代わりにするものは“乱数”と呼ばれている。Fortran90では、サブルーチン `random_number` が用意されているので、これを用いれば乱数を使った計算ができる。例えば、1個の乱数を使うときには

```
call random_number(x)
```

と書けば、変数 `x` に $0 \leq x \leq 1$ の範囲の実数が代入される。このサブルーチンの戻り値 `x` は予測がつかないので、同じ呼び出しを繰り返しても、それ以前に代入された値と等しい数値が代入される確率は非常に小さい。

乱数を1回で複数用意したいときには配列を使う。例えば、配列 `a(10)` に10個の乱数を代入するときは次のように書けばよい。

```
call random_number(a(1:10))
```

モンテカルロ法を使って円周率 π を計算してみよう。2個の乱数 x と y を発生させてこれから座標 (x, y) を作る。この乱数の座標点を N 個作り、その中で原点から半径1の円より内側にある点の数 N_1 を数えて $p = \frac{4N_1}{N}$ を計算する。座標点数 N が増えると、この p が π に近づくことを確かめよ。

第4章 データ出力の詳細とデータ入力

プログラムは計算をさせるだけでは意味がありません。結果を出力して初めて完結します。これまで、もっとも単純な `print` 文でとりあえず画面に表示する方法を使ってきましたが、本章では好みに応じて数値の出力形式を変える方法や、ファイルに保存する方法について説明します。さらにデータを入力することでプログラムを変更せずに動作を変える方法についても説明します。

4.1 データ出力先の指定

これまでもたびたび登場しましたが、もっとも単純な出力命令は、`print` 文です。

```
print form, データ 1, データ 2 ...
```

`print` 文で出力すると、画面 (正確には標準出力) にデータが表示されます。`form` は出力形式を指定するためのものですが、とりあえず “*” を書いておけば、データの数値型に応じた標準形式で表示します。例えば、

```
print *, x, y(i), x**2+5, ' abc = ', 10
```

のように、データの位置には、変数や配列を与えても良いし、計算式や定数を与えることも可能です。また、文字列を適当に入れて、データの意味を並記することもできます。

画面ではなく、ファイルに出力するときは `write` 文を用います。`print` 文との違いは、出力先を指定する整数値 `nd` と出力形式指定の `form` をかっこで記述することです。

```
write(nd, form) データ 1, データ 2 ...
```

`nd` を装置番号といいます。装置番号は、出力ファイルを識別する数字で、任意に選ぶことができます。また、変数や整数値を結果とする計算式を与えることもできるので、条件に応じて出力先を変更することも可能です。`nd` に適当な整数を与えて `write` 文を実行すると、“`fort.nd`” という名のファイルに出力されます²¹。例えば、

```
write(20,*) x,y,z
```

と書けば、`x,y,z` の値が “`fort.20`” という名前のファイルに出力されます。この例のように、`form` に “*” を与えれば、`print` 文と同じく、データの数値型に応じた標準形式で出力します。

なお、原則として $nd \geq 10$ にして下さい。これは、Fortran コンパイラの仕様によって、1桁の数値は予約されている可能性があるからです。例えば、 $nd = 6$ にすれば `print` 文と同じ標準出力の指定になるので、`write(6,*)` と書けば、ファイルへの出力はなく、画面に表示されます。

4.2 配列の出力、do 型出力

出力文において、データの位置に配列名を書けば、全要素が並んで出力されます。例えば、

```
real a(4)
do i = 1, 4
  a(i) = i
enddo
print *,a
```

²¹ “`fort`” の部分はコンパイラに依存しますが、多くは `fort` のようです。ファイル名は 4.6 節で説明する `open` 文を使えば変更することができます。また、コンパイラによっては、`open` 文を使わなくても装置番号に対応する環境変数の指定で、任意の名前を持つファイルに変更することが可能です。

というプログラムを実行すると、

```
1.0000000000000000 2.0000000000000000 3.0000000000000000 4.0000000000000000
```

のように、配列要素が先頭から順に並んで出力されます。このとき、配列要素が多いと適当に改行を入れて出力されます。よって、この単純な出力は配列要素が少ないとき以外はあまりお勧めできません。例えば、2次元配列を、

```
real a(3,3)
do j = 1, 3
  do i = 1, 3
    a(i,j) = i*j
  enddo
enddo
print *,a
```

のように出力すると、 $a(1,1), a(2,1), a(3,1), a(1,2), \dots, a(3,3)$ という並びで9個の要素が出力されます。そこで、次のように do 文を使って出力する方がいいでしょう。

```
do j = 1, 3
  do i = 1, 3
    print *,a(i,j)
  enddo
enddo
```

しかし、print 文や write 文は、1文ごとに改行をするので、複数の print 文を使って、横に続けて書くことはできません。よって、この例の出力では全部で9行必要です。行数を減らすために、1行に行列の1行を出力するには、次のように書かなければなりません。

```
do i = 1, 3
  print *,a(i,1),a(i,2),a(i,3)
enddo
```

しかし、この例のように列の数がわかっているときは良いのですが、列数が多いときや変数で指定したいときには不便です。そこで do 型出力が用意されています。do 型出力とは、

```
(データ 1, データ 2, ..., 整数型変数 = 初期値, 終了値)
```

のような、かっこで囲んだ形式です。これをデータの位置に記述すれば、do 文のように、整数型変数が初期値から終了値まで1つつ増えていき、その変数で指定されるデータが横に並んで出力されます。do 型出力を複数並べたり、通常のデータと並べて書くこともできます。例えば、

```
print *,(a(i,j),j=1,3)
```

は、さきほどの “print *,a(i,1),a(i,2),a(i,3)” と書くことと同等です。また、

```
print *,5*x,(i,a(i),b(i),i=1,3)
```

は、 “print *,5*x,1,a(1),b(1),2,a(2),b(2),3,a(3),b(3)” と書くことと同等です。

do 文と同様に、3番目の数として増分値を付加することもできます。

```
(データ 1, データ 2, ..., 整数型変数 = 初期値, 終了値, 増分値)
```

例えば、

```
print *,(a(i),i=10,1,-2)
```

は、 “print *,a(10),a(8),a(6),a(4),a(2)” と書くことと同等です。

do 型出力は多重にすることもできます。例えば、

```
print *,((a(i,j),j=1,3),i=1,3)
```

は、“print *,a(1,1),a(1,2),a(1,3),a(2,1),a(2,2),a(2,3),a(3,1),a(3,2),a(3,3)”と書くことと同等です。この例のように、多重のときには内側の整数型変数(この例では j)が先に進みます。

なお、do 型出力の整数型変数は、do 文のカウンタ変数に相当します。このため、do 型出力の入っている出力文を do ブロックの中に入れるときには、do 文のカウンタ変数とも重複しないようにしなければなりません。

4.3 format による出力形式の指定

標準出力形式 (“*”指定)で実数を画面に出力すると、有効数字 15 桁の数字を使って表示されます。このため、あまり多くのデータを横に並べることができないし、結果の確認だけなら、それほど有効数字は必要ありません。また、標準形式の出力には 1 行の出力文字数に制限があるため、出力数が制限を超えると自動的に改行されてしまいます。このため、同じ形式の出力を繰り返しても、行ごとに小数点の位置が違うこともあり、大量に出力するのには向きません。

これらの問題は、出力形式 (format) を指定することで解決することができます。これまで “*”を書いていた *form* の位置に format を指定すれば、小数点以下の桁数を小さくしたり、必要に応じて数字と数字の間にスペースを空けたり、改行を入れたりすることができます。また、自動改行されることがないので、幅広く出力することも可能です。

format の指定方法は 2 通りあります。まず、format 文による指定です。一例を示します。

```
real x,y
integer n
x = 1.5
y = 0.03
n = 100
print 600,x,y,n          ! 600 は format 文の文番号
600 format(' x = ',f10.5,' y = ',es12.5,' n = ',i10)
```

最後の文が format 文です。format 文は、サブルーチンの引数のように、出力形式の指定をリストにしてかっこで囲み、先頭に文番号を付けます。この文番号を print 文や write 文の *form* に指定すれば、その format にしたがってデータが整形されて出力されます。この例では、600 が *form* 指定です。文番号は重複できないので、ルーチン内では format 文ごとに異なる数字をつけなければなりません²²。

複数の print 文や write 文が同じ format 文を指定するのは可能です。例えば、次のように共用することができます。

```
real x,y,u,v
integer n,k
.....
print 600,x,y,n
write(20,600) u,v,k
600 format(' x = ',f10.5,' y = ',es12.5,' n = ',i10)
```

format を指定するもう一つの方法は、文字列を使って *form* の位置に直接 format の内容を記述することです。例えば、上記の format を print 文や write 文に埋め込んで、

```
print "(' x = ',f10.5,' y = ',es12.5,' n = ',i10)",x,y,n
write(20,"(' x = ',f10.5,' y = ',es12.5,' n = ',i10)") u,v,k
```

²²文番号の付け方に決まったルールはありませんが、“format 用である”という意味をどこかに入れておいた方が良いでしょう。この例で 600 を使っているのは、“6”が昔の出力装置番号だったころの慣習です。

のように書くことができます。このとき、“format”の文字は不要ですが、両端のかっこは必要です。なお、Fortranでの文字列は「'」で囲むのが基本ですが、「"」も使えるので、format内部に「'」が入っているときには「"」で囲みます。

formatを出力文中に書き込む方法は、文番号を必要としない点は良いのですが、出力文が長くなるし、同じformatを何度も使うときには不便です。そこで、文字列変数を利用して書く方法があります。文字列変数の利用方法については5.3節で説明します。

出力形式の指定方法を上記のformatを例にして説明しましょう。まず、format中の文字列はそのまま出力されるので、必要に応じて適宜挿入します。この例では、' x = 'や、' y = 'は、スペースも含めてそのまま出力されます。

次に、出力文中のデータ値の並びに対し、それぞれの出力形式を指定する“編集記述子”を選んで、前から順に記述します。この例では、f10.5, es12.5, i10, が編集記述子で、print文の並びに対し、

```
print 600,          x          ,          y          ,          n
600 format(' x = ',f10.5,' y = ',es12.5,' n = ',i10)
```

という対応で出力形式を指定しています。文字列と、編集記述子で指定したデータ値は、その並び順に出力されます。よって、このprint文を実行したときの出力は以下のようになります。

```
x =      1.50000   y =  3.00000e-02   n =          100
```

出力形式を指定する編集記述子の主要なものを表4.1に示します。データの数値型に応じた編集記述子を使用しないと正しい値が出力されないので注意して下さい。なお、この表で斜体文字(*w, m, d*)は整数で指定します。

表 4.1 主要な編集記述子

編集指定	数値の型	編集の意味
<i>Iw</i>	整数	幅 <i>w</i> で整数を出力する
<i>Iw.m</i>	整数	幅 <i>w</i> で整数を出力する 出力整数の桁が <i>m</i> より小さい時には、先頭に0を補う ($w \geq m$)
<i>Fw.d</i>	実数	幅 <i>w</i> で実数を固定小数点形式で出力する <i>d</i> は小数点以下の桁数 ($w \geq d + 3$)
<i>Ew.d</i>	実数	幅 <i>w</i> で実数を浮動小数点形式で出力する <i>d</i> は小数点以下の桁数 ($w \geq d + 8$) 仮数部の1桁目は0になる
<i>Gw.d</i>	実数	幅 <i>w</i> で実数を固定小数点形式または浮動小数点形式で出力する どちらになるかは、実数の指数部の大きさで決まる <i>d</i> は小数点以下の桁数
<i>ESw.d</i>	実数	幅 <i>w</i> で実数を浮動小数点形式で出力する (<i>d</i> は E 編集と同じ) 0以外の数値を出力すると、仮数部の1桁目は1から9になる
<i>ENw.d</i>	実数	幅 <i>w</i> で実数を浮動小数点形式で出力する (<i>d</i> は E 編集と同じ) 0以外の数値を出力すると、仮数部の整数は1以上1000未満となり、指数部は3で割り切れる数になる
<i>A</i>	文字列	文字列をそれ自身の長さの幅で出力する
<i>Aw</i>	文字列	幅 <i>w</i> で文字列を出力する

ここでは編集指定の文字(FやESなど)を指定数(*w*など)と区別するために大文字で書きましたが、小

文字でも同じ意味です。例えば、`i10` は整数型値を幅 10 文字で出力することを意味し、`f10.5` は実数型値を幅 10 文字、小数点以下 5 桁で出力することを意味しています。このため、

```
real x,y
integer m,n
x = 1.5
y = 0.03
m = 100
n = 10
print "(f10.5,f10.5,i10,i10.5)",x,y,m,n
```

というプログラムの出力は、

```
1.50000 0.03000 100 00010
+-----+-----+-----+-----+-----+-----+-----+-----+
```

となります。2 行目の目盛りは位置を確認するために書いたものですが、10 文字の中に右寄りで出力されているのがわかります。なお、出力文字数が指定の幅 w を越えると、“*****”のように“*”が w 個出力されます。

E 編集を使って実数を浮動小数点形式で出力すると、小数点の前が 0 になります。例えば、

```
real x,y
x = 1.5
y = 3.14e10
print "(e15.5,e15.5)",x,y
```

の出力結果は、

```
0.15000e+01 0.31400e+11
```

となります。これでは感覚的にわかりにくいし、表示文字数が 1 個無駄になります。そこで、ES 編集や EN 編集を使う方が良いでしょう。例えば、

```
real x
x = 3.14e10
print "(es15.5,en15.5)",x,x
```

の出力結果は、

```
3.14000e+10 31.40000e+09
```

となります。

各編集記述子と出力の数値は 1 対 1 対応にしなければならないので、配列を出力するときには出力要素数と同数の編集記述子を書かなければなりません。このとき、同じ編集記述子を繰り返すならば、編集記述子の前に整数 r を付加して、“ r 回反復する”という指定ができます。例えば、“`3f10.5`”は“`f10.5,f10.5,f10.5`”と書くことと同等です。

さらに、実数、整数、実数、整数のような繰り返しのときには、かっこで囲んで反復指定をすることができます。例えば、次のように書くことができます。

```
real x,y
integer m,n
x = 1.5
y = 0.03
m = 5
n = 100
print "(2(f10.5,' ',i7))",x,m,y,n
```

この `print` 文の `format` は、“`f10.5,' ',i7,f10.5,' ',i7`”と書くことと同等です。

複素数は、“実部、虚部”という実数のペアであり、計算機内部的には要素数2の1次元配列と同型です。このため、複素数を出力するときは、複素数1個あたり、実数の編集記述子を2個並べる必要があります。例えば、以下のように書きます。

```
complex x
c = (1.0,-2.0)
print "(4f8.3)",c,c**2
```

この出力結果は、

```
3.000 -2.000 5.000 -12.000
```

となります。しかし、これでは複素数という感じが出ないので、少し工夫して、

```
complex x
c = (1.0,-2.0)
print "(2(f8.3,' + (' ,f8.3,' )i  '))",c,c**2
```

などとしてみれば良いでしょう。この出力結果は、

```
3.000 + ( -2.000)i 5.000 + ( -12.000)i
```

となります。

なお、format中の編集記述子の数よりも出力文のデータ値の方が多い場合には、編集記述子の数だけ出力した後で改行し、同じformatを再度使って残りのデータ値を出力します。文字列が入っていれば、文字列も再度出力されます。

逆に、format中の編集記述子の数よりも出力文のデータ値の方が少ない場合には、指定したデータ値を出力した段階で終了し、残りの編集記述子は無視されます。無視された記述子以降は文字列等が入っていても全て無視されます。そこで、配列を出力するときなどは、反復指定に大きめの数値を与えておくことができます。例えば、

```
real a(4)
integer m
a(1) = 2.25
a(2) = 30.2
a(3) = 400.7
a(4) = 5000.6
print "(10(f10.2,'cm  '))", (a(m),m=1,4) ! 反復指定は10
```

のように、反復指定を10回にしておいても、出力結果は、

```
2.25cm 30.20cm 400.70cm 5000.60cm
```

となります。

文字列のように、出力文中のデータ値との対応がない編集記述子もあります。

表 4.2 出力文中のデータ値との対応がない編集記述子

編集指定	編集の意味
/	改行する
r/	r回改行する
rX	r個スペースを挿入する
:	出力文中の数値の出力が終わった時点でformat中の以後の出力を打ち切る

ここで、 r は整数で指定します。例えば、2次元配列 $a(3,3)$ の配列要素を3行3列の行列のように1行あたり3個ずつ出力するときは、スラッシュ(/)編集を使って、

```
real a(3,3)
.....
print "(3(3f12.5/))",((a(i,j),j=1,3),i=1,3)
```

と書くことができます。

X編集は出力中に適当な数のスペースを入れるときに使います。これは「' '」のように、スペースの文字列を与えるのと同様です。

最後のコロン(:)編集はわかりにくいので、例を使って説明します。先ほど反復指定に大きめの数値を与えておくことができる例を示しましたが、このとき数字の前に文字列を付けようとするとう問題が起ります。例えば数値の前に\$記号を付けたくて、

```
real a(4)
integer m
a(1) = 2.25
a(2) = 30.2
a(3) = 400.7
a(4) = 5000.6
print "(10('   $',f10.2))", (a(m),m=1,4)      ! 反復指定は10
```

と書くと、出力結果は、

```
$      2.25   $      30.20   $      400.70   $      5000.60   $
```

となります。すなわち、最後に余分な5個目の“\$”が出力されるのです。これは、数値と編集記述子が対応しなくなった時点で出力が終了するのですが、4個目の“\$”の時点ではまだ終了するかどうか未定だからです。これを防ぐために使うのがコロン(:)編集です。上記のformatを修正して、

```
print "(10('   $',f10.2:))", (a(m),m=1,4)      ! f10.2の後にコロンを挿入
```

のようにf10.2の後に“:”を入れておけば、対応しなくなった時点からさかのぼって、“:”の場所で出力が打ち切られます。これならば、

```
$      2.25   $      30.20   $      400.70   $      5000.60
```

のように、“\$”は4個しか出力されません。

4.4 データの入力方法

プログラムが実行しているとき、そのプログラム中の指定した変数に外部からデータを代入することを“入力する”といいます。計算条件を設定するための変数にデータを入力できるようにしておけば、プログラムの実行を開始してから条件を設定して、それに応じた計算をさせることができます。これにより、プログラムはその動作だけを利用する“ブラックボックス”になり、コンパイルしたプログラムを“アプリケーション”として他の人に提供することも可能です。

4.4.1 入力文の一般型

データ入力にはread文を用います。

```
read(nd,*) 変数1, 変数2, ...
```

*nd*は装置番号で、*nd*に適当な整数を与えると“fort.*nd*”という名のファイルから入力します²³。装置番号に関する条件や注意事項は出力の場合と同じで、原則として $nd \geq 10$ にしてください。出力ファイルと違うのはfort.*nd*という名のファイルが存在していなければエラーになるので、あらかじめ用意しておかなければならないことです。なお、“*”の位置には、write文のようにformatによる書式指定ができますが、あまり使うことがないので説明は省略します。

例えば、fort.30 という名前のファイルに、

```
5.2    1.5    3
```

と書いて保存しておき、プログラム中に、

```
read(30,*) x,y,z
```

と書けば、このread文実行後、 $x = 5.2$, $y = 1.5$, $z = 3.0$ となって実行が継続します。入力ファイルに改行が入っていても、read文の変数入力が完了するまで読み込みを続けるので、fort.30の入力数値は次のように3行に分けて書くこともできます。

```
5.2
1.5
3
```

read文実行時に、ファイルが存在しなかったり、データが足りない場合には、エラーになってプログラムが強制終了します。これに対し、read文が要求する数値よりもファイルに書かれている数値の方が多い場合は、read文に記述されている全ての変数に数値が代入された時点で入力が終了します。この後、別のread文で再び入力を実行すると、最後に読み込んだ行の次の行から入力を再開します。例えば、fort.20 という名前のファイルに、

```
5.2    1.5
10     20
```

と書いて保存しておき、プログラム中に、

```
read(20,*) x,y
read(20,*) m,n
```

と書けば、この2回のread文実行後、 $x = 5.2$, $y = 1.5$, $m = 10$, $n = 20$ となります。read文の処理は行単位で行われるので、最後に読み込んだ行に余分な数値が書かれている場合は無視されます。例えば、上記のfort.20を書き換えて、

```
5.2    1.5    30    40
10     20
```

のように1行目に数字を余分に書いても、2回目のread文の結果は変わりません。

read文の変数の位置には、配列名や、do型出力と同型のdo型入力を書くこともできます。これらは、出力と入力という方向が異なりますが、入力要素数や繰り返しの意味は同じです。

ファイルではなく、キーボード(正確には標準入力)を使って入力したいときには、以下のように書きます²⁴。

```
read *, 変数 1, 変数 2, ...
```

²³出力ファイルと同様、“fort”の部分はコンパイラに依存します。ファイル名は4.6節で説明するopen文を使えば変更することができます。また、コンパイラによっては、open文を使わなくても装置番号に対応する環境変数の指定で、任意の名前を持つファイルに変更することが可能です。

²⁴標準入力の装置番号は5です。よって、“read *,”と書く代わりに、“read(5,*)”と書くこともできます。

この文を実行すると、プログラムの実行が一時停止し、キーボードからの数値入力を待つ状態になります。そこで、適切な数値をキーボードから入力すると、その数値を所定の変数に代入した後、実行が再開します。このため、`read` 文のタイミングを考慮して数値を入力しなければ、いつまでたっても停止したままです。そこで、`read` 文の前に入力を促すような文字を出力する `print` 文を入れることをお勧めします。例えば、

```
print *, 'Input X and Y :'  
read *, x, y
```

と書いておけば、入力のタイミングもわかるし、どの変数へ入力するための数値を要求されているかもわかります。

4.4.2 入力時のエラー処理

ファイルからデータを入力するとき、要求したファイルが存在しなかったり、書き込まれたデータ数より多くのデータを入力しようとするれば、実行時エラーになってプログラムは強制終了します。これを防ぐため、`read` 文中にエラー処理指定を入れることができます。

```
read(nd, *, err=num) 変数 1, 変数 2, ...
```

ここで、`num` には文番号を与えます。この `read` 文を実行したとき、入力エラーが起こると `num` で指定した文番号の行へジャンプします。例えば、

```
do k = 1, 100  
  read(10, *, err=999) x, y, z  
  .....  
enddo  
999 x = 100
```

と書けば、エラーが起こると文番号 999 の行にジャンプして、その行から実行を続けます。

もし“ファイルの終了”，すなわち、データを入力するときに、それ以上入っていないかった、という場合を検知するだけなら、`err=num` の代わりに `end=num` と書くこともできます。入力データ数が不明のときには、`err` か `end` 指定を入れておき、データ終了時点で次の処理に進むようなプログラムにしておくとい良いでしょう。

4.4.3 ネームリストを用いた入力

便利なデータ入力手段として、“ネームリスト”を用いる方法があります。例えば、

```
read(10, *) x, y, n
```

という入力文では、入力ファイル `fort.10` を、

```
10.0  1.e10  100
```

のように作成しますが、作成するためには入力変数の対応を常に覚えておかなければなりません。必要なデータを全部書き込まなければならないし、順番を間違えることもできません。

これに対し、ネームリスト入力では、入力データを“変数=データ”という代入形で記述するので、どの変数に代入するかを入力ファイルの中で明示することができます。

ネームリスト入力を使うときは、まず入力する可能性のある変数や配列名を `namelist` 文で登録します。`namelist` 文は次のような形式です。

```
namelist /ネームリスト名/ 変数 1, 変数 2, ...
```

`namelist` 文は非実行文なので、全ての実行文より前に書かなければなりません。また、変数や配列名の登録だけなので、型宣言は別途必要です。例えば、

```
real x,y,a(10)
integer n
namelist /option/ x,y,n,a
```

と書きます。ローカル変数だけではなく、`use` 文で指定されたモジュール中で宣言されているグローバル変数も登録可能です。

`namelist` 文を使ってネームリストに登録された変数に対し、入力文は、

```
read(nd, ネームリスト名)
```

だけです。これをネームリスト入力文といいます。ネームリスト入力文は変数を指定しません。必要ならば、文番号 *num* を使って、

```
read(nd, ネームリスト名, err=num)
```

のように、`err=num` や `end=num` を追加することで、エラー発生や終了時の処理をすることも可能です。例えば、ネームリスト名が `option` ならば、

```
read(15, option, err=999)
```

などのように書きます。

ネームリスト入力文に対する入力ファイルは次の形式で用意します。変数の順番は `namelist` 文の登録順とは無関係なので、自由に並べることができます。

```
&ネームリスト名
 変数 1 = データ 1, 変数 2 = データ 2, ...
/
```

“&ネームリスト名”から“/”までがネームリスト入力文1回で入力されるデータです。例えば上例のようにネームリスト名が `option` のときには、

```
&option
  x=10.0, y=1.e10, n=100
/
```

のようにファイルに書いておきます。入力を開始すると、ネームリスト入力終了の記号“/”を読み込むまで入力を続けるので、次のように1行ずつ書くこともできます。

```
&option
  x=10.0
  y=1.e10
  n=100
/
```

ネームリスト入力にはもう一つ利点があります。それは、必ずしも登録された変数全部を入力ファイルに記述する必要がないことです²⁵。記述しなかった変数には、ネームリスト入力文の実行前までに代入されていた値がそのまま残ります。このため、あらかじめ全ての登録変数にデフォルト値を代入しておけば、変更したい変数だけ入力ファイルに記述することができます。例えば、

²⁵逆に、入力ファイルに同じ変数を複数回記述することもできます。この場合は最後に代入した値が有効になります。

```

real x,y,a(10)
integer n,i
namelist /option/ x,y,n,a
x = 100.0
y = 100.e10
n = 0
do i = 1, 10
  a(i) = i
enddo
read(10,option)

```

のようにプログラムを書いたとします。入力ファイルとして fort.10 という名のファイルに、

```
&option x=10.0, a(3)=5.0 /
```

と書き込んでおけば、read 文実行後、x は変更されますが、y や n はそのままです。配列 a の場合には、代入された要素 a(3) のみが変わります。

なお、ネームリストに登録された変数の内容は、次のネームリスト出力文で出力することもできます。

```
write(nd, ネームリスト名)
```

この場合、全登録変数が“変数=データ値”という形で出力されます。もっとも、ネームリスト出力文を実行すると、登録された全変数のデータが標準形式で出力されるので、変数が多いと煩雑です。取りあえず値を確認したいとき以外は、あまり使わない方が良いでしょう。

4.5 書式なし入出力文によるバイナリ形式の利用

これまで、print 文・write 文による出力や read 文による入力には、文字を使って表現した数値を用いていました。これは我々人間の都合によるものです。計算機内部の数値は2進数で表現されていて、“10”とか、“1.000e20”とかの文字ではありません。しかし、2進数の並びでは我々が理解できないため、出力するときは“2進数→10進数文字表現”というデータ変換を行って画面に表示したりファイルに保存し、入力するときは、“10進数文字表現→2進数”というデータ変換を行って所定の変数に数値を代入するのです。この文字で表現した形式を“テキスト形式”といいます。

しかし、2進→10進変換には時間がかかる上に、文字に変換することでデータ量が増えます。これは、文字データが1文字あたり1byte 必要だからです。例えば、倍精度実数の2進数表現は8byte ですが、実数を有効数字15桁で出力するには15文字(15byte) 以上が必要です。そこで、大量のデータを精度を落とさずに保存するときは、内部表現のままに保存するのが有効です。この内部表現を“バイナリ形式”といいます。Fortran でバイナリ形式の入出力を行うときは、write 文や read 文において form を省略した書式を用います。これを“書式なし入出力文”といいます。これに対し、これまで説明してきた form を指定する write 文や read 文は“書式付き入出力文”です。

書式なし write 文は、次のように装置番号だけをカッコ内に指定した write 文です。

```
write(nd) データ 1, データ 2, ...
```

また、書式なし read 文は、装置番号だけをカッコ内に指定した read 文です。

```
read(nd) 変数 1, 変数 2, ...
```

書式なし read 文は、書式付き read 文と同様に、文番号 num を使って、

```
read(nd,err=num) 変数 1, 変数 2, ...
```

のように、err=num や end=num を追加したエラー発生や終了時の処理をすることもできます。

書式なし write 文で作成したバイナリ形式ファイルから、書式なし read 文を使ってデータを読み込むときにはいくつか注意が必要です。まず、write 文で出力したときのデータと同じ数値型の変数を同じ順番で read 文に並べる必要があります。例えば、

```
real x,y
integer n
x = 10.0
y = 100.0
n = 10
write(20) x,n,y
```

というプログラムで作成された fort.20 というファイルから数値を入力するには、

```
real x,y
integer n
read(20) x,n,y
```

のように書かなければなりません。この場合、x も n も同じ 10 だからとって、

```
read(20) n,x,y
```

と入力すると、出力と数値型の異なる n と x には、正しい値が代入されません。

また、テキスト出力のような“改行”はありませんが、write 文 1 回ごとに印(ヘッダ)が付くので、write 文 1 回の出力に対し、read 文 1 回で入力しなければなりません²⁶。ただし、write 文 1 回で書き込まれたデータ数よりも read 文 1 回で入力するデータが少ないのは問題ありません。このとき入力しなかったデータは読み飛ばしたことに相当します。例えば、上記の fort.20 のデータを、

```
read(20) x,n
read(20) y
```

のように 2 行の read 文に分けて入力しようとする、1 行目の read 文実行時における x と n には正常な値が代入されますが、2 行目の read 文実行時に「入力データがない」というエラーで強制終了します。逆に、上記の write 文を修正して、

```
write(20) x,n
write(20) y
```

という 2 行で出力したファイルから、

```
read(20) x,n,y
```

のように、read 文 1 回で読み込むこともできません。

書式なし write 文は、1 回で出力できるデータ数に特に制限はありません。このため、大きな配列を 1 回で出力することも可能です。例えば、

```
real p(10000),q(10000),r(10000)
write(30) p,q,r
```

のように、1 回の write 文で複数の配列の全要素を出力することができます。

もっとも、出力用のプログラムと入力用のプログラムのメンテナンスを考えれば、次のように出力する配列要素数をあらかじめ出力しておいた方が良いでしょう。

²⁶このように、書式なし write 文を使って保存したバイナリ形式ファイルには、データだけでなく Fortran 独自のヘッダが付加されています。このため、C 言語など、他のプログラミング言語で作成したプログラムから読み込む場合や、データ解析ソフトを使って解析するときにはうまく入力できない場合があります。そのような利用が目的でデータを保存するときは、テキスト形式で保存した方が良いでしょう。

```
real p(10000),q(10000),r(10000)
write(30) 10000
write(30) p,q,r
```

これを入力するときは、出力数を考慮して

```
real p(10000),q(10000),r(10000)
integer i,imlx
read(30) imax
read(30) (p(i),i=1,imax),(q(i),i=1,imax),(r(i),i=1,imax)
```

のように書きます。こうしておけば、出力用のプログラムを修正して要素数を減らしても、入力用のプログラムの変更は不要です。

4.6 ファイルのオープンとクローズ

`write` 文や `read` 文を使ってファイルから入出力をする場合、何も指定がなければ、装置番号 *nd* を付加した “`fort.nd`” という名のファイルを使用します。これに対し、任意の名前を持つファイルを使いたいときには、`open` 文を使って入出力文の実行前にファイル名を指定しておきます。これを “ファイルを開く” といいます。`open` 文は以下の形式です。

```
open(nd,file=name [, form=format] [, status=stat] [, err=num] )
```

[] は、その内容が省略可能という意味です。それぞれの記述 (制御指定子) の意味を表 4.3 に示します。

表 4.3 `open` 文の制御指定子の意味

指定子	指定情報	指定子の意味と注意
<i>nd</i>	装置番号	整数を与える (整数型変数や整数式を与えることも可能) オープンした後、 <code>read</code> 文や <code>write</code> 文の装置番号として使う
<i>name</i>	ファイル名	文字列で指定する (文字変数も可能) ファイル名は大文字・小文字を正しく指定する必要がある
<i>format</i>	ファイル形式	文字列で指定する 省略するとテキスト形式の入出力が仮定される バイナリ形式の入出力を使うときは「 <code>unformatted</code> 」を指定する
<i>stat</i>	ファイル情報	文字列で指定する 既存のファイルを使うときは「 <code>old</code> 」を、存在しないファイルを使うときは「 <code>new</code> 」を指定する 条件に合わないとエラーが発生する
<i>num</i>	文番号	エラーのときにジャンプする行の文番号を指定する 省略すると、エラーが起きたときにはプログラムが強制終了する

例えば、装置番号 10 のテキスト形式ファイルを “`text.out`” という名にするときは、

```
open(10,file='text.out')
```

と書きます。また、装置番号 30 のバイナリ形式ファイルを “`binary.dat`” という名にするときは、

```
open(30,file='binary.dat',form='unformatted')
```

と書きます。ただし、既存のファイルを指定して `write` 文で書き込むと、それまで書き込まれていたデー

タが上書きされて消えるので注意が必要です。上書きを防ぎたいときには、

```
open(30,file='binary.dat',form='unformatted',status='new',err=999)
```

のように、`status='new'` と `err` を指定します。 `status` に `'new'` を指定すると、ファイルが存在しなければ新しく作成し、存在すればエラーになります。そこで、エラーの処理を `err` で指定した文番号 999 の行に用意しておけば上書きを防ぐことができます。

`status` の指定を省略して `open` 文を実行したとき、指定したファイルが存在しなければ、その名前のファイルが新たに作成されます。このとき、そのファイルが `read` 文の入力用であれば、データが入っていないので `read` 文実行時にエラーになりますが、プログラム終了後も作成された空のファイルが残ってしまいます。そこで、ファイルが入力用のときは、`status='old'` を指定して、その存在をチェックした方が良いでしょう。例えば、バイナリ形式ファイル “binary.inp” を入力ファイルとして装置番号 20 に指定するには、

```
open(20,file='binary.inp',form='unformatted',status='old',err=999)
```

のように書きます。この `open` 文では、オープンした時点でファイルが存在しなければ、`err=999` で指定した文番号 999 の行へジャンプします。

ファイルへ出力する場合、`write` 文実行時の出力命令のタイミングと実際にディスクに書き込まれるタイミングは必ずしも一致していません。これは入出力ハードウェアを効率よく運用するために、一時記憶領域への読み書きが介在するためです。このため、確実に書き込みを完了させたいときにはファイルをクローズします。クローズは、次の `close` 文で行います。

```
close(nd)
```

`nd` はクローズするファイルの装置番号です。例えば、装置番号 30 のファイルをクローズするときは、

```
close(30)
```

のように書きます。 `close` 文を実行すると、その時点までに装置番号 `nd` に出力した全てのデータがディスクに書き込まれます。もっとも、プログラムが正常に終了すれば、全てのファイルが自動的にクローズされるので、通常は `close` 文を書かなくても問題ありません。

ファイルをクローズすると、装置番号 `nd` と `open` 文で指定したファイル名の関係は途切れます。このため、同じ装置番号に新たに別のファイルを指定してオープンすることも可能です。また、同じ名前のファイルを再度オープンすることもできます。ただし、再オープンしてもそれ以前の読み書きの継続にはならず、ファイルの先頭に戻って読み書きをします。すなわち、“巻き戻し”をすることになります。ファイルを巻き戻すと、`read` 文による入力は一からやり直すことになり、`write` 文による出力はファイルの先頭から書き直します。このため、巻き戻してから `write` 文で出力すると、それ以前に書き込んだデータは消去されます。

なお、巻き戻すのが目的であれば、`rewind` 文を使うことで、クローズせずともファイルを巻き戻すことができます。 `rewind` 文とは、次のような文です。

```
rewind(nd)
```

`nd` は再度最初から読み書きするファイルの装置番号です。巻き戻しを利用すれば、まず `write` 文を使ってファイルにデータを出力しておき、次にそれを巻き戻して、`read` 文を使ってそのデータを入力して使うことも可能です。

演習問題 4

(4-1) 連立常微分方程式：2 次の Runge-Kutta 法

$N+1$ 個の質量 m のおもりが一直線に並んでいて、隣り合ったおもりはバネ定数 k のバネでつながれているとする。このおもりの位置を左側から、 z_0, z_1, \dots, z_N とすると、 i 番目のおもりの運動方程式はその速度を v_i として、

$$\begin{aligned}\frac{dz_i}{dt} &= v_i \\ m \frac{dv_i}{dt} &= -k(z_i - z_{i-1}) - k(z_i - z_{i+1})\end{aligned}$$

である。このおもりの運動を 2 次の Runge-Kutta 法で解析せよ。

ここで、2 次の Runge-Kutta 法とは微分方程式 $\frac{dx}{dt} = f(t, x)$ に対し、時刻 t_n の値 x_n から出発して時刻 $t_{n+1} = t_n + \Delta t$ の値 x_{n+1} を

$$\begin{aligned}k_1 &= f(t_n, x_n) \\ k_2 &= f(t_n + \Delta t, x_n + \Delta t k_1) \\ x_{n+1} &= x_n + \frac{\Delta t}{2}(k_1 + k_2)\end{aligned}$$

で計算する方法である。 Δt は適当に決める。プログラムは

- (1) 時間を進めるメインプログラム
- (2) Runge-Kutta 法の 1 ステップを計算するサブルーチン
- (3) 運動方程式の右辺を計算するサブルーチン

の 3 個から構成するものとする。さらに、 z_i や v_i は引数で与え、 m や k はグローバル変数としてモジュールと use 文を使って共有するようにせよ。なお、初期条件は以下の通りとする。

$$\begin{aligned}z_i &= hi + g \sin(k_p i) & (i = 0 \dots N) \\ v_i &= 0\end{aligned}$$

ここで、 k_p は適当な整数 p を使って、 $k_p = \frac{2\pi p}{N}$ で与える。また、端にあるおもり (z_0 と z_N) は動かないとする。(つまり、計算しないで初期条件のままとする)

(4-2) 連立常微分方程式：4 次の Runge-Kutta 法

問題(4-1)を修正して 4 次の Runge-Kutta 法で解析せよ。ここで、4 次の Runge-Kutta 法とは微分方程式 $\frac{dx}{dt} = f(t, x)$ に対し、時刻 t_n の値 x_n から出発して時刻 $t_{n+1} = t_n + \Delta t$ の値 x_{n+1} を

$$\begin{aligned}k_1 &= f(t_n, x_n) \\ k_2 &= f(t_n + \frac{\Delta t}{2}, x_n + \frac{\Delta t}{2} k_1) \\ k_3 &= f(t_n + \frac{\Delta t}{2}, x_n + \frac{\Delta t}{2} k_2) \\ k_4 &= f(t_n + \Delta t, x_n + \Delta t k_3) \\ x_{n+1} &= x_n + \frac{\Delta t}{6}(k_1 + 2(k_2 + k_3) + k_4)\end{aligned}$$

で計算する方法である。初期条件や、境界条件は問題(4-1)と同じにし、問題(4-1)の結果と比較せよ。

(4-3) 楕円型偏微分方程式

電荷密度 $\rho(x, y)$ が与えられているときに電位の満足する 2 次元ポワソン方程式

$$\frac{\partial^2 V(x, y)}{\partial x^2} + \frac{\partial^2 V(x, y)}{\partial y^2} = -\rho(x, y)$$

を解くには次のようにする。

等間隔 h ごとに並んだ x 座標点 $x_i = hi, (i = 0 \sim M)$ と y 座標点 $y_j = hj, (j = 0 \sim N)$ に対して, $V_{i,j} = V(x_i, y_j), \rho_{i,j} = \rho(x_i, y_j)$ として, 上の偏微分方程式を差分化すれば次式になる。

$$\frac{V_{i-1,j} - 2V_{i,j} + V_{i+1,j}}{h^2} + \frac{V_{i,j-1} - 2V_{i,j} + V_{i,j+1}}{h^2} = -\rho_{i,j}$$

これは $V_{i,j}$ に関する連立 1 次方程式である。これを解くには, 以下のような漸化式に変形して反復法を用いる。

$$\begin{aligned} V_{i,j}^{(k+1)'} &= \frac{1}{4} \left(V_{i-1,j}^{(k+1)} + V_{i,j-1}^{(k+1)} + V_{i+1,j}^{(k)} + V_{i,j+1}^{(k)} + \rho_{i,j} h^2 \right) \\ V_{i,j}^{(k+1)} &= V_{i,j}^{(k)} + \omega (V_{i,j}^{(k+1)'} - V_{i,j}^{(k)}) \end{aligned}$$

これを適当な初期条件 $V_{i,j}^{(0)}$ から出発して, k 回目の反復値 $V_{i,j}^{(k)}$ と $k+1$ 回目の反復値 $V_{i,j}^{(k+1)}$ の差が十分小さくなれば, 元の連立方程式の近似解になると考えられる。

なお, 第 1 式の $V_{i,j}^{(k+1)'}$ をそのまま次の反復値 $V_{i,j}^{(k+1)}$ として用いる手法を, ガウス・ザイデル法という。しかし, ガウス・ザイデル法は収束が遅いので, 第 2 式を使って解を改良する。これを, SOR (Successive Over Relaxation, 逐次過緩和) という。 ω は加速係数であり, $\omega = 1$ のときがガウス・ザイデル法, $\omega > 1$ が SOR である。 ω は 2 を越えると発散するので, $1 < \omega < 2$ の範囲で最も収束が速くなる数値を選ぶ。

この反復法を使って, 次の電荷密度における電位を計算するプログラムを作成せよ。

$$\rho(x, y) = \sin \left(\frac{2\pi x}{Mh} \right) \sin \left(\frac{4\pi y}{Nh} \right)$$

ここで, 初期条件は $V_{i,j}^{(0)} = 0$, 境界条件は $V_{0,j} = V_{i,0} = V_{M,j} = V_{i,N} = 0$ ($i = 0 \sim M, j = 0 \sim N$) である。また, 収束の判定は

$$\max_{i,j} |V_{i,j}^{(k+1)} - V_{i,j}^{(k)}| < \varepsilon$$

で行うこと。ただし, ε は適当な小さい値とする (例えば, 10^{-12})。

プログラムは,

- (1) 反復を行い, 収束条件を確認するメインプログラム
- (2) 初期値を代入するサブルーチン
- (3) ガウス・ザイデル法 + SOR を 1 ステップ行うサブルーチン

の 3 個から構成するものとする。

(4-4) 放物型偏微分方程式：陽解差分法

次の温度 $T(x, t)$ に関する熱伝導方程式を、陽解差分法 (Explicit 法) を用いて解析せよ。

$$\frac{\partial T}{\partial t} = \kappa \frac{\partial^2 T}{\partial x^2}$$

このような時間発展の偏微分方程式を解くには、まず等間隔 h ごとに並んだ x 座標点 $x_i = hi, (i = 0 \sim L)$ に対して初期の温度分布を与え、時間間隔 Δt ごとに時間を進めた温度分布を計算する。陽解差分法とは、 n ステップ目の温度分布を $T_i^n = T(x_i, \Delta t n)$ と表したときに、熱伝導方程式を、

$$\frac{T_i^{n+1} - T_i^n}{\Delta t} = \kappa \frac{T_{i+1}^n - 2T_i^n + T_{i-1}^n}{h^2}$$

と近似する方法である。この方程式を T_i^{n+1} に関して解けば、

$$T_i^{n+1} = T_i^n + \frac{\kappa \Delta t}{h^2} (T_{i+1}^n - 2T_i^n + T_{i-1}^n) \quad (i = 1, 2, \dots, L-1)$$

となるので、これを使って T_i^n から T_i^{n+1} を繰り返し計算していけばよい。

ただし、境界条件として、 $T_0^n = 0, T_L^n = 0$ とする。また、初期条件は、

$$\begin{cases} T_i^0 = i & (i < L/2) \\ T_i^0 = L - i & (i > L/2) \end{cases}$$

とする。プログラムは

- (1) 時間を進めるメインプログラム
- (2) 初期条件を計算するサブルーチン
- (3) 陽解法で1ステップ進めるサブルーチン

の3個から構成するものとする。温度 T_i^n と、 $h, \Delta t, \kappa$ などのパラメータはグローバル変数にしてモジュールと use 文で共有する。また、メモリの節約のため、 T_i^n を配列 T1(i) で表し、 T_i^{n+1} を配列 T2(i) で表して、T1 と T2 のデータをうまく交換するように考える。

完成したプログラムを使って、 $\frac{\kappa \Delta t}{h^2} = 0.2$ にすると安定に計算ができるのに、 $\frac{\kappa \Delta t}{h^2} > 0.5$ にすると、不安定になることなども確かめよ。

(4-5) 放物型偏微分方程式：陰解差分法

温度 $T(x, t)$ に関する熱伝導方程式を、陰解差分法 (Implicit 法) を用いて解析せよ。

ただし、陰解差分法とは、問題 (4-4) の偏微分方程式を

$$\frac{T_i^{n+1} - T_i^n}{\Delta t} = \frac{1}{2} \left[\kappa \frac{T_{i+1}^{n+1} - 2T_i^{n+1} + T_{i-1}^{n+1}}{h^2} + \kappa \frac{T_{i+1}^n - 2T_i^n + T_{i-1}^n}{h^2} \right]$$

と近似する方法である。この方程式を変形すれば、

$$a_i T_{i-1}^{n+1} + b_i T_i^{n+1} + c_i T_{i+1}^{n+1} = d_i \quad (i = 1, 2, \dots, L-1)$$

の形になるので、問題 (2-10) で示した連立1次方程式の解法を使って T_i^{n+1} が計算できる。

ここで、境界条件や初期条件は陽解法と同じでよい。プログラムは

- (1) 時間を進めるメインプログラム
- (2) 初期条件を計算するサブルーチン
- (3) 陰解法で1ステップ進めるサブルーチン

の3個から構成するものとする。温度 T_i^n と、 h , Δt , κ などのパラメータはグローバル変数にしてモジュールと use 文で共有する。また、メモリの節約のため、 T_i^n を配列 T1(i) で表し、 T_i^{n+1} を配列 T2(i) で表して、T1 と T2 のデータをうまく交換するように考える。

完成したプログラムでは、 $\frac{\kappa\Delta t}{h^2} > 0.5$ にしても不安定にならないことなどを確かめよ。

(4-6) 分子動力学

2個の距離 r 離れた分子間の位置エネルギーを表す近似式の一つに次式のようなレナード・ジョーンズポテンシャルがある [3].

$$U(r) = 4\epsilon \left\{ \left(\frac{\sigma}{r} \right)^{12} - \left(\frac{\sigma}{r} \right)^6 \right\}$$

この式は、分子が近づくと斥力（反発力）が働き、遠ざかると引力が働くことを示している。レナード・ジョーンズポテンシャルの勾配を計算することで分子間に働く力が計算できる。すなわち、2個の分子 i と j の位置ベクトルを \mathbf{r}_i , \mathbf{r}_j とすると、分子 j が分子 i に及ぼす力は、

$$\mathbf{F}_{ij} = -\nabla_i U(|\mathbf{r}_i - \mathbf{r}_j|) = \frac{4\epsilon}{\sigma} \left\{ 12 \left(\frac{\sigma}{r_{ij}} \right)^{13} - 6 \left(\frac{\sigma}{r_{ij}} \right)^7 \right\} \frac{\mathbf{r}_i - \mathbf{r}_j}{r_{ij}}$$

となる。ここで、 $r_{ij} = |\mathbf{r}_i - \mathbf{r}_j|$ である。

今、分子が N 個あるとすると、分子 i に加わる力は i 以外の分子から受ける力の合力なので、

$$\mathbf{F}_i = \sum_{\substack{j=1 \\ i \neq j}}^N \mathbf{F}_{ij} = \sum_{\substack{j=1 \\ i \neq j}}^N \frac{4\epsilon}{\sigma} \left\{ 12 \left(\frac{\sigma}{r_{ij}} \right)^{13} - 6 \left(\frac{\sigma}{r_{ij}} \right)^7 \right\} \frac{\mathbf{r}_i - \mathbf{r}_j}{r_{ij}}$$

となる。これを用いて N 個の分子の運動方程式

$$m_i \frac{d^2 \mathbf{r}_i}{dt^2} = \mathbf{F}_i = \sum_{\substack{j=1 \\ i \neq j}}^N \frac{4\epsilon}{\sigma} \left\{ 12 \left(\frac{\sigma}{r_{ij}} \right)^{13} - 6 \left(\frac{\sigma}{r_{ij}} \right)^7 \right\} \frac{\mathbf{r}_i - \mathbf{r}_j}{r_{ij}} \quad (i = 1 \dots N)$$

を解くプログラムを作成せよ。ここで、分子数が増えると時間がかかるので、効率よく時間積分をするためにベルレ法を用いる [3]。ベルレ法とは次式のように2段階で位置ベクトルを更新する方法である。

まず第1段階で速度ベクトルを更新する。

$$\mathbf{v}_i^n = \mathbf{v}_i^{n-1} + \frac{\Delta t}{m_i} \frac{\mathbf{F}_i^n + \mathbf{F}_i^{n-1}}{2}$$

この更新された速度ベクトルを用いて、第2段階で位置ベクトルを更新する。

$$\mathbf{r}_i^{n+1} = \mathbf{r}_i^n + \Delta t \mathbf{v}_i^n + \frac{\Delta t^2}{2m_i} \mathbf{F}_i^n$$

ここで、 \mathbf{F}_i^n とは、 n ステップ目の位置ベクトル \mathbf{r}_i^n を用いて計算した分子 i に加わる力 \mathbf{F}_i である。

なお、力の計算に r_{ij}^{-2k} が入っているため、2分子の距離がかなり接近すると、力が大きくなりすぎて、安定に計算することができなくなる。そこで、 r_{ij}^{-2} を次式のように修正するとよい。

$$\frac{1}{r_{ij}^2} \rightarrow \frac{1}{(x_i - x_j)^2 + (y_i - y_j)^2 + (z_i - z_j)^2 + \epsilon^2}$$

ここで、 ϵ はもっとも接近する可能性のある距離で、 σ に比べて十分小さい数にする。

プログラムは、

- (1) 時間を進めるメインプログラム
- (2) 初期条件を計算するサブルーチン
- (3) ベルレ法第1段階で速度ベクトルを更新するサブルーチン
- (4) ベルレ法第2段階で位置ベクトルを更新するサブルーチン

の4個から構成するものとする。

初期条件としては、まず適当な間隔離れた2個の分子でテストする。これにより正常に動作することが確認されたら、正三角形の頂点に配置した3個の分子や正四面体の頂点に配置した4個の分子などの動きを計算せよ。

(4-7) 1次元移流方程式

次の関数 $f(x, t)$ に関する1次元移流方程式を、CIP法を用いて解析せよ [4]。

$$\frac{\partial f}{\partial t} + u \frac{\partial f}{\partial x} = 0$$

ただし、CIP法とは、間隔 h の x 座標点 $x_i = hi, (i = 0 \sim L)$ で定義された、 n ステップ目の関数値 $f_i^n = f(x_i, n\Delta t)$ と、その x 方向微分 $g(x, t) = \frac{\partial f}{\partial x}$ の値 $g_i^n = g(x_i, n\Delta t)$ が与えられているとき、時間間隔 Δt 後の関数値と微分値を、

$$\begin{aligned} f_i^{n+1} &= a_i X^3 + b_i X^2 + g_i^n X + f_i^n \\ g_i^{n+1} &= 3a_i X^2 + 2b_i X + g_i^n \end{aligned}$$

で近似する方法である。ここで、 $X = -u\Delta t$ である。また、 a_i, b_i は次式で与えられる。

$$\begin{aligned} a_i &= \frac{g_i^n + g_{i+1}^n}{h^2} + \frac{2(f_i^n - f_{i+1}^n)}{h^3} \\ b_i &= \frac{3(f_{i+1}^n - f_i^n)}{h^2} - \frac{2g_i^n + g_{i+1}^n}{h} \end{aligned}$$

初期条件は、グリッド数 L と、その中間の値、 i_1, i_2 を適当に決めて、

$$\begin{cases} f_i^0 = 1 & (i_1 < i < i_2) \\ f_i^0 = 0 & (\text{それ以外}) \end{cases}$$

とせよ。なお、 $h = 1, u = -1$ とする。

まず、 n ステップ目の座標点データ f_i^n と g_i^n を1次元配列 $f1(i)$ と $g1(i)$ で表し、これらを用いて、 $n+1$ ステップ目の座標点データ f_i^{n+1} と g_i^{n+1} を表す1次元配列 $f2(i)$ と $g2(i)$ を計算するサブルーチンを作成する。そして、これらを適宜交換しながら繰り返すことで、 f_i^n や g_i^n が時間発展するプログラムを作成すればよい。なお、 i の最大値 L では a_i, b_i を計算することができないので、常に $f_L^n = 0, g_L^n = 0$ で良い。

プログラムは、

- (1) 時間を進めるメインプログラム
- (2) 初期条件を計算するサブルーチン
- (3) CIP法で関数値と微分値を更新するサブルーチン

の3個から構成するものとする。

(4-8) 粒子密度 – 分布関数

2次元領域, $0 \leq x \leq Lh$, $0 \leq y \leq Mh$, の内部に N 個の粒子があるとき, 粒子の密度分布は以下のような手順で計算することができる.

まず, 2次元領域に等間隔 h ごとに並んだ x 座標点 x_0, x_1, \dots, x_L と y 座標点 y_0, y_1, \dots, y_M で指定したグリッドを導入する. 次に密度配列 $\text{Rho}(0:L, 0:M)$ に初期値として0を代入しておいて, k 番目の粒子座標 $(x(k), y(k))$ から以下のようにして各グリッドに所属する粒子数をカウントする.

```
i = x(k)/h
j = y(k)/h
Rho(i,j) = Rho(i,j) + 1
```

ただし, i と j は整数型変数である. すなわち, $x(k)/h$ と $y(k)/h$ を切り捨てることで格子座標を計算する.

問題 (3-10) で紹介した, 乱数発生用のサブルーチンを使って, ランダムな位置にある粒子を N 個作成し, 密度の平均と標準偏差を計算せよ. そして, N が大きくなると, 標準偏差が小さくなることを確かめよ.

(4-9) ブラウン運動 – 拡散

ある点から粒子を速度 v で出発させたところ, 時刻 Δt ごとに周りの粒子に衝突して方向を変えらるとする. この時の粒子の進む様子を計算してみよう.

粒子の初期位置は, 全て $(\frac{Lh}{2}, \frac{Mh}{2})$, すなわち, 2次元領域の中央とする. 計算は

- (1) 1ステップごとに乱数 R を発生させ, $\theta = 2\pi R$ で, 方向角 θ を計算する
- (2) 1ステップ後の位置を $(x_{s+1}, y_{s+1}) = (x_s + v\Delta t \cos \theta, y_s + v\Delta t \sin \theta)$ で計算する
- (3) 粒子が $0 \leq x \leq Lh$, $0 \leq y \leq Mh$ の領域から外に出たら計算をストップする

のように進めればよい.

この過程によって N 個の粒子の N_T ステップ後の位置を計算し, その密度を計算せよ. また, ステップ N_T を増やすと, 密度がどの様に変化するかを調べよ.

プログラムは,

- (1) 時間を進めるメインプログラム
- (2) 初期条件を計算するサブルーチン
- (3) 粒子の座標を更新するサブルーチン
- (4) 粒子の位置から密度を計算するサブルーチン

の4個から構成するものとする.

第5章 文字列の活用

これまで `print` 文や `format` 文に記述して、数値の意味などを表示するのに使ってきた文字列ですが、この他にも様々な用途があります。また、固定した文字の並びだけではなく、文字列変数を使って条件に応じてその内容を変更したり、文字列と文字列を連結したり、文字列の一部を取り出したりするなどの“文字列演算”をすることもできます。本章では文字列をより積極的に活用する手法を紹介します。

5.1 文字列定数と文字列変数

Fortran における文字列とは2個の「`'`」または「`"`」で囲んだ文字の並びのことです。例えば、

```
'abc'  'Taguchi_T.'  "(123.5678+X^2)"  "漢字も書けます"
```

等が文字列です。文字列にはスペースや記号を入れることもできます²⁷。「`'`」と「`"`」は同等なので、どちらを使うかは自由です。例えば、通常は「`'`」で囲み、文字列の中に「`'`」を使いたいときは、「`"Teacher's"`」のように全体を「`"`」で囲む、というように決めておけば良いでしょう。

コンピュータの“文字”は文字コードという整数値と対応していて、文字列は文字コードを表す整数の配列で実現されています。このため、整数定数の「`1`」と文字列の「`'1'`」は全く異なるものです。文字コードは半角英数字なら1文字あたり1byteの整数、全角の漢字やひらがななら1文字あたり2byteの整数です。半角英数字のコードは、現在ほとんど全てのコンピュータでASCIIコードが使われています。このため、半角英数字だけでプログラムを書けば移植性の問題はありません。しかし、全角文字はOSや利用環境によって文字コードが違う可能性があるため、プログラム中に日本語を記述すると、別のコンピュータに移したときに文字化けすることがあります²⁸。Fortranにおいて日本語が使えるのはコメント文とテキスト表示くらいなので、それほど重要ではありません。以下、文字列中の“文字”は1byteのASCIIコードであると仮定して説明をします。

「`'`」または「`"`」で囲んだ文字列は、文字列“定数”ですが、文字列を代入して保存する文字列“変数”を作ることができます。文字列変数を作るには `character` 宣言を使います。 `character` 宣言は、以下のような書式です。

```
character 文字列変数 1*文字数 1, 文字列変数 2*文字数 2, ...
```

文字数とは、文字列変数に代入可能な最大の文字数です。指定できる最小の文字数は1です。また、拡張宣言文にして属性などを付加することも可能です。例えば、

```
character c1*10,c2*20,cc*1
character chr(20)*30,chs(100,200)*50
```

と宣言すると、文字列変数 `c1` には10個まで、`c2` には20個までの文字を代入することができます。これに対し `cc` には1文字しか入りません²⁹。また `chr` や `chs` のように文字列の配列を宣言することも可能です。 `chr` は30文字まで入る1次元配列で、`chs` は50文字まで入る2次元配列です。

同じ文字数の文字列変数を複数個宣言するときは、以下の2種類の書式で宣言することもできます。

```
character(文字数) 文字列変数 1, 文字列変数 2, ...
character(len=文字数) 文字列変数 1, 文字列変数 2, ...
```

²⁷本章では半角スペース記号を明記するときに“`_`”を使います。

²⁸パソコンで使われているのはShift JISコードが多く、LinuxなどのUNIX系OSで使われているのはEUCコードが多いようです。最近では、コード体系をより統一化したUnicodeを使うOSが増えてきました。

²⁹文法的には“*文字数”を省略して変数名だけで宣言すると、文字数が1の文字列になります。しかし、プログラムがわかりにくくなるので、常に“*1”を付加して文字数1を陽に指定した方が良いでしょう。

ここで、“文字列変数”には、変数名か、配列名とその要素数を記述します。例えば、文字数 10 の文字列変数や文字列配列を宣言するときは、

```
character(10) cs1,ds1,cas1(100)
character(len=10) cs2,ds2,cas2(100)
```

などと書きます。2行目の書式は、入力に手間がかかりますが「文字数」という意味を明示しているところが利点です。文字数は、次のように parameter 変数 (3.9.2 節) で指定することもできます。

```
integer, parameter :: kp = 10
character(kp) ch2
character(len=kp) ch3
```

文字列変数に文字列を代入するには、数値の代入と同じで、イコール「=」を使います。例えば、上記の 10 文字の文字列変数 `c1` に文字列定数を代入するには、

```
c1 = 'abc'
```

のように書きます。このとき、代入される文字 'abc' は 3 文字なので、10 文字の変数 `c1` の先頭から順に 1 文字ずつ代入され、残りの領域は半角スペースが代入されます。スペースも文字ですから、`c1` の文字数は、代入した文字列の文字数にかかわらず、10 のままです。図に描けば、以下のようなイメージです。

c1	a	b	c						
----	---	---	---	--	--	--	--	--	--

逆に、代入する文字列の文字数の方が多い場合には、その文字列の先頭から代入できる最大文字数までが代入され、残りは切り捨てられます。拡張宣言文を使えば、次のようにあらかじめ文字列定数を代入した文字列変数を用意することも可能です。

```
character :: chr*10='abcde'
```

この場合、文字列変数の文字数が 10 文字なのに代入しているのは 5 文字ですから、後の 5 文字はスペースが代入されています。

5.2 部分文字列と文字列演算

文字列変数に代入された文字列は部分的に取り出すことができます。これを“部分文字列”といいます。部分文字列は、文字列変数の先頭から数えて何番目から何番目という範囲を「:」を使って指定します。例えば、`c1` という文字列変数の `n1` 番目から `n2` 番目の文字を取り出すには、

```
c1(n1:n2)
```

と指定します。この文字列は、`n2-n1+1` 文字の文字列として扱われます。例えば、`c1='abcdefg'` ならば、`c1(3:5)` は、'cde' です。`n1` と `n2` を等しくすることで 1 文字を取り出すこともできます。例えば、

```
c1 = 'abcdefg'
do i = 1, 7
  print *,c1(i:i)
enddo
```

とすれば、1 文字ずつ縦に出力されます。

文字列配列の部分文字列を取り出す場合には、要素指定を先に、部分文字列指定を後に書きます。例えば、1次元の文字列配列 `chr` に対し、`k` 番目の要素の部分文字列は、

```
chr(k)(n1:n2)
```

のように指定します。かっこが連続するので、順番に気を付けて下さい。

文字列は連結することもできます。文字列の連結には演算子「//」を用います。例えば、

```
c1 = 'abc'//'xyz'
```

と書くと、c1には'abcxyz'という文字列が代入されます。文字列の連結は、文字列定数と文字列変数、文字列変数と文字列変数という組み合わせでも可能です。ただし、文字列変数に代入された文字列を連結するときには注意が必要です。例えば、

```
character c1*10,c2*20
c1 = 'abc'
c2 = c1//'xyz'
```

と書いた場合、c2に'abcxyz'という文字列が代入されると思ったら間違いです。正しくは

```
'abc          xyz'
```

が代入されます。これは、上記のように10文字の文字変数c1に3文字の'abc'を代入しても、c1の文字数は変わらないからです。末尾の不要なスペースを削除するには、部分文字列を使う必要があります。

```
character c1*10,c2*20
c1 = 'abc'
c2 = c1(1:3)//'xyz'
```

このプログラムならば、c2に'abcxyz'が代入されます。

しかし、この方法は変数に代入されている文字数が不明のときには使えません。そこで、関数trimが用意されています。trimは末尾のスペースを除去した文字列を返す組み込み関数です。例えば、上記の例は、

```
character c1*10,c2*20
c1 = 'abc'
c2 = trim(c1)//'xyz'
```

と書くことができます。この結果もc2には'abcxyz'が代入されます。

文字列をサブルーチンの引数にするときは、「(*)」を指定して宣言します。例えば、

```
subroutine csubr1(chr)
  implicit none
  character(*) chr      ! 文字数は不要
  .....
```

のchrのように宣言します。Fortranの文字列には、文字コードの並びだけではなく、文字数の情報も含まれています。このため、(*)指定のように文字数が明記されていなくても、コール側で指定した文字数の文字列として使用することができます。例えば、

```
program ctest1
  implicit none
  call csubr1('abcde')
end program ctest1

subroutine csubr1(chr)
  implicit none
  character(*) chr
  print *,chr,len(chr)  ! 文字列 'abcd' と文字数 5 が出力される
end subroutine csubr1
```

のようなプログラムにおいて、サブルーチンcsubr1中のprint文の出力は、文字列'abcde'と文字数5になります。ここで、関数lenは文字列の文字数を取得する関数です。文字列に関する組み込み関数

は、5.5 節で説明します。

文字列には大小関係があり、これを利用して if 文で条件分岐をすることもできます。2 個の文字列を比較するときは、以下の手順で行います。

- (1) 文字数の長さが異なるときは、短い方の文字列の末尾にスペースを追加して文字数を等しくする
- (2) 2 個の文字列を先頭から 1 文字ずつ比較して行って、全て同じならば、“等しい(==)”
- (3) 異なる文字があれば、最初に異なる文字の ASCII コードを比較して、コード値の大小が文字列の“大(>)”または“小(<)”

ASCII コードは、“スペース”<“数字”<“英大文字”<“英小文字”の順で大きくなり、個々の文字は次のような順序になっています。

```
□ < '0' < '1' < ... < '9' < 'A' < 'B' < ... < 'Z' < 'a' < 'b' < ... < 'z'
```

例えば、文字列の比較を使って次のようなプログラムを書くことができます。

```
character c1*10,c2*20
c1 = 'abcde'           ! 3 番目が 'c'
c2 = 'abdce'           ! 3 番目が 'd'
if (c1 < c2) print *, 'c1 < c2'
if (c1 == c2) print *, 'c1 == c2'
if (c1 > c2) print *, 'c1 > c2'
```

この結果は、“c1 < c2”です。なぜなら、3 番目の文字が初めて異なり、c1 は 'c'、c2 は 'd' ですが、'c' < 'd' だからです。なお、c1 は 10 文字、c2 は 20 文字ですが、後ろにスペースを補うので、この例の結果には無関係です。

5.3 出力における文字列の利用

最もよく文字列を使う場面は、print 文や write 文の中に入れて表示の補助に使うことでしょう。例えば、

```
real spd,pres
spd = 10.0
pres = 960.0
print *, ' Wind Speed = ',spd,' Pressure = ',pres
```

と書けば、出力が

```
Wind Speed =      10.0000000000000      Pressure =      960.000000000000
```

のようになって、どの数値がどの変数の値なのかが一目でわかります。

ここで、文字列はその長さに合わせて出力されています。これは、print 文が文字列に入っている文字数の情報に合わせて出力するからです。この機能は、format で文字列の出力指定をするときにも使うことができます。文字列の出力指定には、A 編集(表 4.1)を用いますが、A 編集は、「a」だけ書くと、文字列の文字数が出力幅になります。例えば上記の print 文を、

```
print 600,' Wind Speed = ',spd,' m/s'
print 600,' Pressure = ',pres,' hPa'
600 format(a,f8.2,a)
```

と書けば、次のような文字幅に合わせた出力結果が得られます。

```
Wind Speed =      10.00 m/s
Pressure =      960.00 hPa
```

4.3 節の `format` の説明で紹介しましたが、出力形式指定を `print` 文や `write` 文の中に埋め込む書式も、文字列の利用方法の一つです。例えば、

```
print 600,x
write(10,600) x
600 format(' x = ',es12.5)
```

というプログラムは次のように書き換えることができます。

```
print "(' x = ',es12.5)",x
write(10,"(' x = ',es12.5)") x
```

すなわち、`format` 文のかっこ以下を、両端のかっこを含めて文字列にして `print` 文や `write` 文の書式指定の位置 (*form*) に書き込みます。このとき、文字列変数を使えば、複数の場所で同じ `format` を使うときに便利です。

```
character :: form*20="(' x = ',es12.5)"
print form,x
write(10,form) x
```

文字列変数を利用すれば、出力内容に応じて実行時に `format` を変更することも可能です。例えば、

```
real x
character form*20
if (abs(x) >= 1.e5) then
  form = "('x = ',es12.5)"
else
  form = "('x = ',f10.5)"
endif
print form,x
```

とすれば、`x` の絶対値が 10^5 以上のときには `es12.5` 編集で、さもなければ `f10.5` 編集で出力されます。また、部分文字列を利用して変更することも考えられます。例えば、

```
real x
character form*20
form = "('x = ',f10.5)"
if (x >= 100.0) form(10:13)='12.3'      ! 10.5 から 12.3 に変更
print form,x
```

のように書けば、`x` が 100.0 以上のときは `f12.3` 編集で、さもなければ `f10.5` 編集で出力されます。

5.4 数値・文字列変換

ここまでは `write` 文や `read` 文の書式指定に文字列を使用していましたが、装置番号の位置に文字列を記述することもできます。これは“文字列”から“数値”への変換、またはその逆変換を行うときに使用します。5.1 節で述べたように、「123」という数値と「'123'」という文字列は計算機内部の表現が異なりますが、処理の過程で 123 という数値からそれに相当する文字を作ったり、逆に '123' という文字列を数値として計算に利用したい場合があります。そもそも、書式付き `write` 文は“計算機内部の 2 進数”を“文字で表現された 10 進数”に変換して出力する動作であり、書式付き `read` 文はファイルなどから“文字で表現された 10 進数”を入力して“計算機内部の 2 進数”に変換する動作です。装置番号の位置に文字列を利用すると、その変換機能だけを使うことができます。

“数値”→“文字列”変換をするには `write` 文を用います。例えば、

```

real x
character ch*20
x = 123.5
write(ch,"(f10.5)") x

```

と書けば、文字列変数 `ch` に ' 123.50000 ' という文字列が代入されます。ただし、文字列に変換するときは、出力文字数以上の長さを持つ文字列変数を用意しておく必要があります。

`open` 文 (4.6 節) を使ってファイルをオープンするとき、ファイル名は文字列で指定する必要があります。そこで、`file01.dat`, `file02.dat`, ..., `file10.dat` という通し番号の付いた 10 個のファイルが存在して、それぞれを装置番号 11~20 としてオープンするときは、次のように書きます。

```

character name*20
integer m
do m = 1, 10
  write(name,"('file',i2.2,'.dat')") m
  open(10+i,file=name)
enddo

```

ここで、整数 `m` の出力編集を `i2.2` にしているのは、スペースを '0' で埋めて、整数の出力を 01, 02, ... のようにするためです。

逆に、“文字”→“数値”変換をするには `read` 文を用います。例えば、

```

real x
character ch*20
ch = '12345.0'
read(ch,*) x

```

と書けば、実数型変数 `x` に 12345.0 という“数値”が代入されます。

5.5 文字列に関する組み込み関数

文字列に関する組み込み関数は、`trim` や `len` の他にも色々あります。代表的なものを表 5.1 に示します。本節ではこれらの関数の中で比較的良く使うものの使用方法について説明します。

表 5.1 文字列に関する組み込み関数

文字列関数	引数の型	関数値の型	関数の意味
<code>trim(c)</code>	文字列	文字列	末尾の空白を削除
<code>len(c)</code>	文字列	整数	文字列の文字数
<code>len_trim(c)</code>	文字列	整数	末尾の空白を削除した文字列の文字数
<code>adjustl(c)</code>	文字列	文字列	文字列を左にそろえた文字列
<code>adjustr(c)</code>	文字列	文字列	文字列を右にそろえた文字列
<code>index(c,cp)</code>	2 個の文字列	整数	文字列 <code>c</code> 中で文字列 <code>cp</code> が含まれていれば、その開始位置を返す 無ければ 0 を返す
<code>repeat(c,n)</code>	文字列と整数	文字列	文字列 <code>c</code> を <code>n</code> 個連結した文字列
<code>char(n)</code>	整数	1 文字の文字列	ASCII コード値を与えると、それに対応する半角英数文字を返す
<code>ichar(c)</code>	1 文字の文字列	整数	半角英数 1 文字を与えると、それに対応する ASCII コード値を返す

まず、5.2 節にも出てきた関数 `len` は、文字列の文字数を取得するための関数ですが、使用の際は注

意が必要です。例えば、

```
character c1*10
integer m,n
m = len('abc')           ! m = 3
c1 = 'abc'
n = len(c1)              ! n = 10
```

とすると、mの値は3ですが、nの値は10になります。これは、文字列変数の文字数が宣言時の文字数だからです。末尾のスペースを除去した文字列の長さを取得するには、関数trimを使って除去してから関数lenで調べるか、関数len_trimを使います。例えば、上のプログラムで、最後のnへの代入文を

```
n=len(trim(c1))   または   n=len_trim(c1)
```

で置き換えれば、n=3になります。

最後のcharとicharの使用例を示します。この二つの関数を組み合わせれば、read文やwrite文を使わなくても、文字と数値の変換をすることができます。例えば、2桁の整数を3桁の8進数で表示するプログラムは以下ようになります。

```
integer zero,num
character c1*10
num = 12
c1 = '000'
zero = ichar('0')           ! '0'の文字コードを整数値に変換
c1(3:3) = char(zero+mod(num,8)) ! numの1の位の文字
c1(2:2) = char(zero+mod(num/8,8)) ! numの8の位の文字
c1(1:1) = char(zero+mod(num/64,8)) ! numの64の位の文字
print *,'decimal(',num,') = octal( ',trim(c1),' )'
```

ASCIIコードでは、'0','1','2',..., '8','9'の順で値が1ずつ増加しています。そこで、先頭の'0'のコード値に1桁の数字を加えれば、その数字のコード値が得られます。同様に、大文字の英数字や、小文字の英数字も1ずつ増加して並んでいるので、先頭の文字である'A'や'a'のコード値に、その文字の順番の差の数字を加えて逆変換すれば、対応する文字を得ることができます。整数の16進数表示をするプログラムを考えてみてはいかがでしょうか。

第6章 配列計算式

第5章で説明した文字列の処理は、それ以前に説明した数値の計算と少し毛色が違います。数値計算が1個の数値と1個の数値の間の演算を基本としているのに対し、文字列の処理では、“文字列”という文字コードの集合をひとまとめにして取り扱い、文字列と文字列の連結や文字列変数への代入を一つの式で実行することができます。文字列とは文字コードの配列なのですから、これを数値計算的に処理するならば、本来は1文字ずつ処理しなければなりません。文字列の演算が可能なのは、Fortran コンパイラが、“文字列”という数値集合の処理を、内部で1文字ずつ処理するプログラムに変換してくれるからです。

Fortran では、この概念を一般の配列に拡張した“配列演算”が可能です。配列演算とは、数値の集合を配列名で代表させ、配列要素と配列要素の演算を、配列名と配列名の演算の形で記述するものです。配列演算を使えば、do 文を使わずに配列を使った計算を記述することができます。これを“配列計算式”と呼びます。配列計算式を使う利点は、単にプログラムが短くなるだけではありません。計算が最適になるように、繰り返しの順序やメモリ配置をコンパイラが決めてくれるので、場合によっては、do 文を使って書くより高速に計算させることができます。

本章では、配列計算式の使い方について説明します。また、プログラム実行中に任意の要素数を持つ配列を用意する“動的割り付け”についても説明します。

6.1 基本的な配列計算式

配列演算は、配列の全要素に対して、全て同じパターンの演算をするのが基本です。このため、1行の配列計算式で用いる配列は、次元と各次元の要素数が全て等しい“同型の配列”でなければなりません。その点に注意すれば、配列の全ての要素に同じ定数を代入するときや、2個の配列の対応する要素間で全て同じ四則演算をするときに、あたかも配列名が一つの変数であるかのような記述をすることができます。例えば、次のような記述が可能です。

```
real x(10,10),y(10,10),z(10,10)
x = 1.0
y = 2.0
z = x + 3.5*x*y**2
```

このプログラムを実行すると、配列 x の全ての要素は 1.0 になり、配列 y の全ての要素は 2.0 になり、配列 z の全ての要素は 15.0 ($=1.0 + 3.5 \times 1.0 \times 2.0^2$) になります。このプログラムは do 文を使った以下のプログラムと同じ結果になります³⁰。

```
real x(10,10),y(10,10),z(10,10)
integer i,j
do j = 1, 10
  do i = 1, 10
    x(i,j) = 1.0
    y(i,j) = 2.0
    z(i,j) = x(i,j) + 3.5*x(i,j)*y(i,j)**2
  enddo
enddo
```

配列計算式中の定数は、全要素に対して共通の値として計算します。

次のような、関数を使った配列計算式も可能です。

³⁰ただし、ループの処理はコンパイラが自動的に生成するので、実際にこの通りの順序で計算するかどうかはわかりません。あくまでも、このプログラムと同じ結果になるという意味だと考えてください。

```

real a(10,10),b(10,10),c(10,10)
.....
a = sqrt(b*sin(c)/(3.2*c + 1.5e-3))

```

この配列計算式を do 文で表すと、次のようになります。

```

do j = 1, 10
  do i = 1, 10
    a(i,j) = sqrt(b(i,j)*sin(c(i,j))/(3.2*c(i,j) + 1.5e-3))
  enddo
enddo

```

配列計算式においては、式中に含まれる全ての配列が同型でなければならないので、異なる次元や要素数の配列が混じっているとエラーになります。ただし、下限は異なってもかまいません。例えば、

```

real a(3,3),r(-1:1,0:2)
.....
a = 3.14*r**2

```

のように計算することができます。ただし、この配列計算式を do 文で表せば、

```

do j = 1, 3
  do i = 1, 3
    a(i,j) = 3.14*r(i-2,j-1)**2
  enddo
enddo

```

のように、対応する要素の位置がずれるので注意して下さい。

このように、配列計算式を使うとプログラムがシンプルになります。逆に、単一変数の計算式と配列計算式との区別がなくなるので、両者が入り交じるとプログラムがわかりにくくなり、エラーが見つげにくくなるという欠点もあります。

なお、サブルーチンの引数配列を使って配列計算をするときには注意が必要です。例えば、

```

subroutine subr(a,b)
  implicit none
  real a(*),b(*)
  a = b**2           ! これはエラーになる
  .....

```

のようなプログラムでは、配列計算式はエラーになります。なぜなら、“*”を入れて配列宣言をした場合は、要素数が不定だからです。これに対し、

```

subroutine subr(a,b,n)
  implicit none
  real a(n),b(n)
  integer n
  a = b**2           ! この場合は OK
  .....

```

のように、整合配列を使うと、配列計算が可能になります。

6.2 部分配列

配列名だけを使った配列計算式では全ての要素について計算をしますが、常に全要素の計算が必要とは限りません。そこで、配列の一部を取り出した配列，“部分配列”を指定することができます。部分配列は“:”（コロン）を使って要素番号の範囲を指定します。例えば、1次元配列 a に対して、

```
a(n1:n2)
```

と書けば、 $a(n1) \sim a(n2)$ という $n2-n1+1$ 個の要素から構成された1次元配列として配列計算式に使うことができます。また、2次元配列 b に対して、

```
b(n11:n12,n21:n22)
```

と書けば、 $b(n11,n21) \sim b(n12,n21) \sim b(n11,n22) \sim b(n12,n22)$ という $(n12-n11+1) \times (n22-n21+1)$ の2次元配列として配列計算式に使うことができます。また、ある次元の要素番号を固定して、

```
b(n,n21:n22)
```

と書けば、 $b(n,n21) \sim b(n,n22)$ という $n22-n21+1$ 個の要素から構成された1次元配列として配列計算式に使うことができます。

逆に、要素番号に“:”だけを記述すると、「その次元の全要素」という意味になります。例えば、

```
b(:,n21:n22)
```

と書けば、第1次元は全要素で、第2次元が $n21 \sim n22$ の範囲の要素から構成された2次元配列として配列計算式に使うことができます。このため、全ての要素番号を“:”にすると、配列全体を代表することになります。そこで、全配列要素を使った配列計算式を書くときにも“:”を使って配列であることを明示することができます。例えば、6.1節の2次元配列計算式、

```
a = sqrt(b*sin(c)/(3.2*c + 1.5e-3))
```

は、次のように書くことができます。

```
a(:, :) = sqrt(b(:, :)*sin(c(:, :))/(3.2*c(:, :) + 1.5e-3))
```

この方が、単一変数の計算式と区別できるので良いと思います。本書でもこれ以降はこの書式を使って全要素の配列計算式を記述します。

要素範囲の指定に“:整数値”を追加して、do文のような増分値を指定することもできます。例えば、1次元配列 a に対して、

```
a(n1:n2:n3)
```

と書けば、 $a(n1)$, $a(n1+n3)$, $a(n1+2*n3)$, ... という要素からなる1次元配列として配列計算式に使うことができます。この場合、終了要素は $n2$ とは限らないので注意して下さい。例えば、次のように書くことができます。

```
real a(10),b(5)
.....
b(1:5) = a(1:10:2)
```

この結果は、 $b(1)=a(1)$, $b(2)=a(3)$, $b(3)=a(5)$, $b(4)=a(7)$, $b(5)=a(9)$ という5個の代入文と等価になります。増分値は負数を与えることも可能なので、次のように逆向きに代入させることも可能です。

```
real a(10),c(10)
.....
c(1:10) = a(10:1:-1)
```

この結果は、 $c(1)=a(10)$, $c(2)=a(9)$, ..., $c(10)=a(1)$ という 10 個の代入文と等価になります。

配列計算式で部分配列を使うときは、その部分配列がその他の配列と同型であれば良いので、宣言文での次元や要素数が一致している必要はありません。例えば 1 次元配列なら、

```
real a(10),b(5)
.....
a(3:5) = b(1:3)**2
```

のように書くことができます。また、2 次元配列の部分配列を使って、

```
real a(10,10),b(5,4),c(10)
.....
a(3:5,6:8) = b(1:3,1:3)**2
```

などと書くことができます。この配列計算式を do 文で表せば、

```
do j = 6, 8
  do i = 3, 5
    a(i,j) = b(i-2,j-5)**2
  enddo
enddo
```

となります。ある次元の要素番号を固定した 2 次元配列は 1 次元配列として扱えるので、2 次元配列と 1 次元配列が混在した計算も可能です。例えば、

```
real a(10,10),c(10)
.....
c(3:8) = a(3,3:8)**2
```

と書くことができます。この配列計算式を do 文で表せば、

```
do j = 3, 8
  c(j) = a(3,j)**2
enddo
```

となります。

なお、部分配列を使った配列計算式で、左辺と右辺に同じ配列を使う場合には、単純に do 文で置き換えた動作と結果が異なる場合があるので注意が必要です。例えば、

```
real a(10)
do i = 1, 10
  a(i) = i
enddo
a(2:10) = a(1:9)      ! この配列計算式は単純な do 文で置き換えられない
print *,(a(i),i=2,10)
```

というプログラムを考えます。この中の配列計算式 $a(2:10)=a(1:9)$ を do 文にすると、

```
do i = 1, 9
  a(i+1) = a(i)
enddo
```

と置き換えられそうですが、実際にやってみると print 文の出力結果は異なります。配列計算式の場合は 1 から 9 までの異なる数字が出力されるのに対し、この do ループで置き換えると全て 1 が出力され

ます。

これは、配列計算式がdoループのように1要素ずつ計算と代入を繰り返すのではなく、まず右辺の配列要素計算を全て行って結果を補助配列に保存しておき、その後で保存した補助配列から左辺の配列への代入を行うからです。このため、配列代入機能を利用すると、次のように配列要素の順番を逆転させるのも簡単です。

```
a(1:10) = a(10:1:-1)
```

これと同じ動作をするプログラムを1回のdoループで書くのは難しいです。この代入機能は、配列計算式を使う利点の一つです³¹。

6.3 where 文による条件分岐

配列演算は便利ですが、全ての要素について同じ形の計算をするので、要素の条件に応じて異なる処理をさせることができません。そこで、配列要素の条件に応じて動作を分岐させるためのwhere文が用意されています。where文は、以下のような形式です。

```
where (配列条件) 配列計算式
```

これは、“配列条件”に合った要素に対してのみ、“配列計算式”を実行するという意味です。例えば、

```
real a(10,10),b(10,10)
.....
where (a(:, :) > 0) b(:, :) = a(:, :)**2
```

のように書きます。このwhere文の部分をdo文で表せば、次のようになります。

```
do j = 1, 10
  do i = 1, 10
    if (a(i,j) > 0) b(i,j) = a(i,j)**2
  enddo
enddo
```

where文はブロック構文にすることもできます。ブロックwhere文では、次のようにelse where文を付加して、条件に一致しない場合の記述をすることもできます。

```
where (配列条件 1)
  配列計算式 1
.....
else where (配列条件 2)
  配列計算式 2
.....
else where
  配列計算式 0
.....
endwhere
```

この場合、配列条件1を満足する要素は配列計算式1のブロックを実行し、配列条件1を満足せず、配列条件2を満足する要素は配列計算式2のブロックを実行し,... という具合に続いて、全ての条件を満足しない要素は配列計算式0のブロックを実行するという動作になります。中間のelse whereに関するブロックは省略可能です。if文と違って、“then”が不要なこと、“全ての条件以外”を表す文が“else where”であること、ブロックの最後に“endwhere”文を付加すること、などに注意して下さい。例えば、

³¹ただし、配列要素数が非常に大きくなると、コンピュータのメモリ利用可能限界に近くなると、補助配列の生成のおかげでメモリオーバーになる可能性もあります。

```

real a(10,10),b(10,10)
.....
where (3.0*b(:,.) > 0)
  a(:,.) = b(:,.)**2
else where
  a(:,.) = b(:,.)**3
endwhere

```

のように書くことができます。このブロック where 文を do 文で表せば、次のようになります。

```

do j = 1, 10
  do i = 1, 10
    if (3.0*b(i,j) > 0) then
      a(i,j) = b(i,j)**2
    else
      a(i,j) = b(i,j)**3
    endif
  enddo
enddo

```

where ブロック中で使用する配列は、全て同型でなければなりません。このため、where ブロックの中に、単一変数の計算式や、次元や要素数の異なる配列計算式を入れることはできません。

6.4 配列定数

配列演算を使えば、do 文を使わなくても配列要素の計算ができますが、これまでの書式では配列要素ごとに異なる定数を代入することはできません。そこで、1次元配列だけですが、配列定数が用意されています。n 個の要素からなる1次元の配列定数は以下の形式です。

```
(/定数 1, 定数 2, ..., 定数 n/)
```

このとき、n 個の定数は全て同じ型の数値型でなければなりません。例えば、

```

real a(5)
a(:) = (/1.0,2.0,3.0,4.0,5.0/)

```

と書けば、右辺は実数型の1次元配列定数であり、この配列計算式は、 $a(1) = 1.0$ 、 $a(2) = 2.0$ 、 $a(3) = 3.0$ 、 $a(4) = 4.0$ 、 $a(5) = 5.0$ という5個の代入文を実行したことに相当します。この配列定数は、次のように拡張宣言文の初期値代入に使うこともできます。

```
real :: a(5)=(/1.0,2.0,3.0,4.0,5.0/)
```

しかし、この形式では数値を全て書かねばならず、配列要素が多いと大変です。そこで、4.2節で出てきた、“do 型形式”を使って定数設定をすることができます。これを do 型定数といいます。do 型定数とは、以下のような形式です。

```
(データ 1, データ 2, ..., 整数型変数=初期値, 終了値)
```

この形式では、まず整数型変数(カウンタ変数)を初期値にして、データ 1, データ 2, ... と並べ、次に、カウンタ変数を1増加して、再度、データ 1, データ 2, ... と並べ、カウンタ変数が終了値になるまで繰り返して得られる配列定数になります。このため、配列要素数は“データ並びの数×カウンタ変数のカウント回数”になります。例えば、

```
(/n, n=1,5/)
```

という do 型定数を使った配列定数は、整数型の配列定数、

```
(/1,2,3,4,5/)
```

と同じです。また、

```
(/(n, n**2, n=1,5)/)
```

という do 型定数を使った配列定数は、整数型の配列定数、

```
(/1,1,2,4,3,9,4,16,5,25/)
```

と同じです。

do 型出力と同様に増分値を指定することもできます。

```
(データ 1, データ 2, ..., 整数型変数=初期値, 終了値, 増分値)
```

例えば、

```
(/(n, n**3, n=1,8,2)/)
```

という do 型定数を使った配列定数は、整数型の配列定数、

```
(/1,1,3,27,5,125,7,343/)
```

と同じです。

複数の do 型定数を並べたり、通常の定数と混在させることも可能です。例えば、

```
(/(real(n), n=0,2),1.5,1.8,(real(n), n=3,5)/)
```

という実数型の配列定数は、

```
(/0.0,1.0,2.0,1.5,1.8,3.0,4.0,5.0/)
```

と同じです。このとき、型変換関数の `real(n)` を単に `n` と書くとエラーになるので注意して下さい。これは、`n` が整数型なので、`n` だけを書くとその配列定数の中に実数型と整数型が混在するからです。

do 型定数は多重にすることも可能です。例えば、

```
(/((i+j,i=1,3),j=1,4)/)
```

という do 型定数を使った整数型の配列定数は、

```
(/2,3,4,3,4,5,4,5,6,5,6,7/)
```

と同じです。多重にした場合、カウンタ変数は内側が先に進みます。また、do 型出力と同様に、do 型定数を do ブロックの中に入れるときには、カウンタ変数が do 文のカウンタ変数とも重複しないようにしなければなりません。

本節最初の例は、次のように do 型定数を使って書くことができます。

```
real a(5)
a(:) = (/n, n=1,5/)
```

ここで、右辺は整数型の配列で、左辺は実数型の配列ですが、この場合は全要素に対して整数型から実数型への変換が行われて代入されるのでエラーにはなりません。なお、do 型定数は拡張された宣言文での初期値代入に使うこともできますが、カウンタ変数はその宣言文より前に宣言されている必要があります。

配列定数は1次元配列しか用意されていません。このため、2次元以上の配列計算で使うときは、1次元ごとに代入する必要があります。例えば、

```
real a(3,4)
integer i,j
do j = 1, 4
  a(:,j) = (/sin(0.5*(i+j)), i=1,3/)
enddo
```

のように書きます。これは以下のプログラムと同等です。

```
real a(3,4)
integer i,j
do j = 1, 4
  do i = 1, 3
    a(i,j) = sin(0.5*(i+j))
  enddo
enddo
```

ただ、配列計算でdo文を使うのはあまりスマートではありません。そこで、配列の形状を変換する組み込み関数 `reshape` が用意されています。`reshape` を使って1次元の配列定数を多次元配列に変換すれば、そのまま多次元の配列計算式に記述できます。この方法は6.5節で説明します。

6.5 配列に関する組み込み関数

配列を引数にする組み込み関数も色々用意されています。まず、配列要素を利用した計算に関する関数を表6.1に示します。

表 6.1 配列要素の計算に関する組み込み関数

配列関数	引数の型	関数値の型	関数の意味
<code>sum(a)</code>	配列	配列要素の型	全ての配列要素の和
<code>product(a)</code>	配列	配列要素の型	全ての配列要素の積
<code>minval(a)</code>	配列	配列要素の型	全ての配列要素の最小値
<code>maxval(a)</code>	配列	配列要素の型	全ての配列要素の最大値

例えば、

```
real a(5)
a(:) = (/i, i=1,5/)
print *,sum(a)           ! 配列 a の全要素の合計
```

と書けば、配列 `a` の全要素の合計 (15.0) が出力されます。部分配列を利用すれば、指定した範囲の配列要素だけを使った計算も可能です。例えば、

```
real a(10)
integer n
a(:) = (/i, i=1,10/)
n = 5
print *,product(a(1:n)) ! 部分配列の要素 a(1)~a(n) の積
```

と書けば、`a(1)` から `a(n)` までの積 (120.0) が出力されます。

表6.1の関数を使う場合、引数配列の後に整数型の引数を追加すると、その整数値が指定する次元に関してのみ計算をすることができます。例えば、

```

real b(3,4),x,y(4),z(3)
integer i,j
do j = 1, 4
  b(:,j) = (/sin(0.5*(i+j)), i=1,3)/)
enddo
x = maxval(b)           ! 配列 b の全要素の最大値
y(:) = maxval(b,1)     ! 配列 b の第 1 次元方向の要素の最大値
z(:) = maxval(b,2)     ! 配列 b の第 2 次元方向の要素の最大値

```

と書けば、x には 2 次元配列 b の全要素の最大値が代入されますが、1 次元配列 y には、b の第 2 次元を固定して第 1 次元方向に要素を比較したときの最大値がそれぞれ代入されます。また、1 次元配列 z には、b の第 1 次元を固定して第 2 次元方向に要素を比較したときの最大値がそれぞれ代入されます。すなわち、次元を指定すると、その次元を抜いた要素数の配列が結果になります。

次に、配列情報に関する関数を表 6.2 に示します。

表 6.2 配列情報に関する組み込み関数

配列関数	引数の型	関数値の型	関数の意味
minloc(a)	配列	整数型配列	全ての配列要素の最小値の位置
maxloc(a)	配列	整数型配列	全ての配列要素の最大値の位置
lbound(a)	配列*	整数型配列	配列の各次元の最小要素番号リスト
ubound(a)	配列*	整数型配列	配列の各次元の最大要素番号リスト
shape(a)	配列	整数型配列	配列の各次元の要素数リスト
size(a)	配列*	整数	配列の全要素数
reshape(a,s)	配列と整数型配列	a の型の配列	配列 a を配列 s で指定された型の配列に変換する
allocated(a)	配列	論理	配列 a が割り付けられていれば真、さもなければ偽

表 6.2 で、引数の型に “*” の付いた関数は、引数配列の後に次元を示す整数型の引数を追加して、その次元に関する値を取り出すこともできます。例えば、

```

real array(3,4)           ! 3 × 4 の配列
integer m,n1,n2
m = size(array)          ! 全要素数 3 × 4 = 12
n1 = size(array,1)       ! 第 1 次元要素数 3
n2 = size(array,2)       ! 第 2 次元要素数 4

```

と書けば、m には、array の全要素数である 12 が代入され、n1 には array の第 1 次元の要素数である 3 が、n2 には array の第 2 次元の要素数である 4 が代入されます。

また、表 6.2 で関数値の型が “整数型配列” になっている関数は、引数に与えた配列の次元を要素数に持つ 1 次元整数型配列が戻り値です。その際、その 1 次元配列の各要素には、引数配列の各次元の情報が代入されています。例えば、下限と上限を取得するための関数、lbound と ubound は以下のように使います。

```

real bb(-10:10,4)        ! 2 次元配列
integer blow(2),bupp(2)  ! 2 次元なので要素数 2 の配列
blow = lbound(bb)
bupp = ubound(bb)
print *,blow,bupp

```

ここで、2次元配列 `bb` は、第1次元の下限が -10 、上限が 10 、第2次元の下限が 1 、上限が 4 ですから、定数で書けば、`blow` は $(/-10,1/)$ 、`bupp` は $(/10,4/)$ になります。このとき、`ubound(bb,2)` のように、次元に関する引数を加えると、その結果は1個の整数(この例では 4) になります。

`reshape` は、配列の形状を変換する関数です。そもそも、どんな次元の配列もメモリ上では1次元的に並んでいるので、2次元配列 `a(3,4)` と1次元配列 `b(12)` のように、全要素数が等しい配列は同じように取り扱えるはずですが、配列演算は次元および各次元の要素数が等しい同型の配列間でしか許可されていません。そこで、`a(3,4)` と `b(12)` の間で演算をするには、形式上、同型になるような変換が必要です。これを実行するのが関数 `reshape` です。`reshape` には、最初の引数に変換したい配列を与え、2番目の引数に変換後の形状を表す1次元配列を与えます。ここで形状を表す1次元配列とは、変換後の配列が、 $k_1 \times k_2 \times \dots \times k_n$ の n 次元配列であれば、

```
(/k1,k2,...,kn/)
```

で与えられる要素数 n の1次元整数型配列のことです。例えば、6.4節最後の例文は、

```
real a(3,4)
integer i,j
a(:, :) = reshape((/(sin(0.5*(i+j)), i=1,3), j=1,4)/), (/3,4/))
```

と書くことができます。`reshape` の第1引数は `do` 型定数なので1次元配列ですが、これを第2引数の $(/3,4/)$ で 3×4 の2次元配列に変換するよう指定しているわけです。

この配列の形状を表す1次元配列は、関数 `shape` を使って取得することができます。例えば、上記の `a(3,4)` という宣言をした配列に対し、`shape(a)` は、1次元整数型配列、 $(/3,4/)$ になります。そこで、上記の配列計算式は次のように書くことができます。

```
a(:, :) = reshape((/(sin(0.5*(i+j)), i=1,3), j=1,4)/), shape(a))
```

しかし、これではまだ `do` 型変数における `i` や `j` の範囲指定が定数のままです。そこで、ここは先ほど説明した関数 `size` で書き直すと良いでしょう。

```
a(:, :) = reshape((/(sin(0.5*(i+j)), i=1, size(a,1)), j=1, size(a,2))/), shape(a))
```

これなら、宣言文の要素数を変更するときに、この配列計算式を変更する必要はありません。

6.6 配列の動的割り付け

配列を宣言するときには、整数定数を与えて要素数を明示しなければなりません。これは、その情報に基づいて、プログラムの動作開始時に計算で使用するメモリ量を確定するためです。しかし、最近のコンピュータではプログラムの実行中に必要になったメモリを確保したり、不要になったメモリを解放するという動作が頻繁に行われています。このような、プログラムの実行中に必要に応じてメモリを確保することを“動的割り付け”といいます。FortranもFortran90から動的割り付けが可能になり、プログラムの動作に必要な要素数を持つ配列を実行中に確保できるようになりました。動的割り付けを使うと、メモリ効率の良い汎用性のあるプログラムを作成することができます。ここでその方法を紹介します。

まず、サブルーチンの中で必要に応じた要素数の配列を確保する簡単な方法として、サブルーチンの引数を使った配列宣言があります。例えば、次のサブルーチンで `real` 宣言されている2次元配列 `b` がこれに相当します。


```

subroutine memory1(a,m,n)
  implicit none
  real a(m,n),b(m,n),x           ! 引数 m と n を使って宣言する
  integer m,n,i,j
  x = 10.0
  b(:, :) = x*reshape((/(i*j,i=1,m),j=1,n)/),shape(b))
  a(:, :) = b(:,:)*3
end subroutine memory1

```

配列 `b` は、このサブルーチンがコールされた時点で、引数 `m` と `n` の値を使って確保されます。このとき、`b(m,n)` のような単純な宣言だけでなく、`b(0:2*m,n-1)` のような、下限指定や計算式を使った宣言も可能です。ここで、配列 `a` も引数を使って宣言されていますが、`a` は引数に含まれているので 3.5 節で説明した整合配列です。整合配列は、引数配列の取り扱いを決める仕組みにすぎず、配列自体はサブルーチン側にありません。

これに対し、引数に含まれていない配列 `b` はローカル配列であり、宣言されたサブルーチンの一時メモリ領域に所属します。3.9.1 節で説明したように、一時メモリ領域に所属するローカル変数は、サブルーチン呼び出した時点で生成されます。引数の値はサブルーチン開始時に確定しているため、これを使って配列に必要なメモリを確保することが可能なのです。ただし、これはあくまでもサブルーチンで宣言する配列についてのみ使える機能であり、メインプログラムの配列やモジュール中のグローバル配列としては使えません。

そこで、メモリの動的割り付けを明示的に行う仕組みが用意されています。Fortran で動的割り付けを行うには、二つの手続きが必要です。一つは、割り付けたメモリを配列として使用するための配列名の宣言です。これは、割り付けたメモリの先頭アドレスを保持するためのメモリを用意することだと考えられます。もう一つは、その配列名に実際のメモリを割り付ける動作です。前者は宣言文なので非実行文、後者はプログラム実行中に割り付けるので実行文です。

メモリを割り付ける予定の配列名は、`allocatable` 属性を付けて型宣言をします³²。このとき、配列の次元情報をコロン“:”を使って示す必要があります。配列の次元はコロンの数で決まり、実行時に変更することはできません。例えば、

```

real, allocatable :: ab(:),z2(:, :)
integer, allocatable, dimension(:, :) :: km1, km2

```

のように宣言します。1 行目のように、名前の後ろに次元情報を付加しても良いし、2 行目のように、`dimension` 属性を使うことも可能です³²。この例の場合、`ab` は 1 次元実数型配列、`z2` は 2 次元実数型配列、`km1` と `km2` は 2 次元整数型配列として割り付けることができます。

配列の割り付けは `allocate` 文で行います。`allocate` 文は、次のように配列の要素指定をサブルーチンのかっこで囲んで記述します。

```

allocate ( ab(100), z2(0:m,-m:m), km1(k-1,2*k) )

```

かっこ内は、配列宣言と同じ形式です。`ab` や `km1` のように、要素数に整数だけを与えると、その次元の下限は 1 になるし、`z2` のように“:”を使って下限指定をすることも可能です。配列の数値型は型宣言の際に確定しているため、この例のように、一つの `allocate` 文で異なる数値型の配列を同時に割り付けることも可能です。`allocate` 文は実行文なので、`z2` や `km1` のように整数型変数や整数式の計算結果を使って割り付けることも可能です。一旦割り付けられた配列は、通常の宣言文で宣言した配列と同様に使用することができます。

ただし、むやみに割り付けを繰り返すと、コンピュータで利用できる容量を超える“メモリオーバー”になる可能性があります。そこで、不用になった配列のメモリ領域は `deallocate` 文で解放することが

³²配列に属性を付加する方法と `dimension` 属性については 3.9.1 節参照。

できます。deallocate 文は、次のように配列名だけをかっこで囲んで指定します。

```
deallocate ( ab, km1 )
```

deallocate 文も異なる数値型の配列を同時に解放することが可能です。例えば、先ほどのサブルーチン memory1 は、以下のように書き換えることができます。

```
subroutine memory1(a,m,n)
  implicit none
  real a(m,n),x
  real, allocatable :: b(:, :)          ! 割り付け用 2次元実数型配列
  integer m,n,i,j
  allocate (b(m,n))                    ! m×n の 2次元配列を割り付け
  x = 10.0
  b(:, :) = x*reshape((/(i*j,i=1,m),j=1,n)/),shape(b))
  a(:, :) = b(:, :)**3
  deallocate (b)                       ! メモリの解放
end subroutine memory1
```

なお、引数変数を使って宣言したローカル配列は一時メモリ領域に所属しますが、allocate 文で割り付けた配列は固定メモリ領域に所属します³³。サブルーチンにおける一時メモリ領域は固定メモリ領域より使用可能容量が少ない可能性があるため、要素数の大きな配列を割り付けるときは allocate 文を使う方が良いでしょう。

allocate 文を使った動的割り付けは、メインプログラムでも利用できるし、モジュールの中で配列名宣言をして、グローバル配列として使用することも可能です。ただし、allocate 文で割り付けられた配列を、解放しないで再度 allocate 文で割り付けようとするとう実行時エラーになります。また、割り付ける前にその配列を使用しようとしても実行時エラーになります。そこで、配列が割り付けられているか否かを確認する関数が用意されています。これが表 6.2 の最後にある関数 allocated です。allocated は論理型の値を返す関数で、allocatable 属性を持つ配列名を引数に与えると、その配列がすでに割り付けられていれば“真”，割り付けられていなければ“偽”を返します。論理型の関数は、そのまま if 文の条件として使うことができるので、例えば次のように利用することができます。

```
real, allocatable :: a(:)
integer i,n
.....
if (allocated(a)) then
  a(:) = /(100*i,i=1,n)/          ! 代入 1
else
  allocate ( a(n) )
  a(:) = /(200*i,i=1,n)/          ! 代入 2
endif
```

この場合、配列 a がすでに割り付けられていれば“代入 1”を実行し、割り付けられていなければ、allocate 文で割り付けてから“代入 2”を実行します。

メモリの動的割り付けや解放は、それほど時間がかかる処理ではありません。計算の途中で一時的に利用する配列は、必要なときに割り付けて、不要になったら解放するようにしておけば、メモリの利用効率が高くて汎用性のあるプログラムになります。

³³ただし、サブルーチン終了後も数値を保持するために配列を固定メモリ領域に所属させる場合には、配列名の宣言に save 属性を加えて、配列名も固定メモリ領域に所属させなければなりません。

付録 A gfortran を用いたコンパイルから実行までの手順

gfortran は Fortran90 の文法で書いたプログラムをコンパイルできる代表的なフリーの Fortran です。多くの Linux ディストリビューションに付属しているし、Windows 版や Mac OS 版も配布されているので、お手持ちのパソコンにインストールすれば、手軽に Fortran を利用することができます。ここでは gfortran を使用したコンパイルから実行までの手順について述べます。

まず、Fortran プログラムを作るときには、ファイル名の最後に “.f90” を付けます。つまり、

```
文字列.f90
```

という名前にします。このファイル名の最後に付加した “ドット (.)+文字” の部分を拡張子と呼びます。作成したプログラムファイルを計算機で実行させるには “コンパイル” と “リンク” という二つの過程が必要です。

コンパイルというのは、Fortran や C 言語など、人間が理解できる言語で書かれたプログラムをコンピュータが理解できる機械語に翻訳することです。コンパイルするアプリケーションを “コンパイラ” といいます。フリー Fortran コンパイラの代表が gfortran です。

gfortran でプログラムをコンパイルをするときは、gfortran コマンドを使って、

```
gfortran プログラムファイル名
```

と入力します。例えば、test1.f90 というファイル名のプログラムをコンパイルするときは

```
gfortran test1.f90
```

と入力します。コンパイル時に文法エラーなどが見つかったら、エラーメッセージを出力して終了します。

gfortran コマンドは、コンパイルが成功すると、引き続きリンクを行います。プログラムはコンパイルしただけでは実行できません。コンパイラはプログラミング言語を機械語に翻訳するだけであり、複数のルーチンを結合して OS 上で起動可能な形式にする作業までは行わないからです。この結合処理がリンクです。単にプログラム中のルーチンのリンクだけではありません。入出力文の read や write、組み込み関数の sin や log 等は、標準ライブラリルーチンとして用意されているので、必要に応じてこれらのリンクも行います³⁴。

リンクにも成功すると、最後に “a.out” という名前のファイルが作成されます。これが OS 上で直接実行させることができる機械語で書かれたファイルで、これを実行形式ファイルといいます。もしリンクに失敗した場合は、エラーメッセージを出力して終了します。この時 a.out は作成されません。

実行形式ファイルは、一般のコマンドのようにファイル名を入力することでプログラムを実行することができます。ただし、次のように頭に “./” を付ける必要があります³⁵。

```
./a.out
```

プログラムが正常に動作すれば、それに書かれた一連の計算を行って終了します。計算の途中で print 文の記述があれば、結果を画面に出力するし、write 文の記述があれば、指定したファイルに書き出します。また、標準入力 of read 文の記述があれば、その時点でプログラムが一時停止して入力待ちの状態になります。この状態で必要な数値をキーボードから入力すれば計算は再開します。

以上が gfortran を用いた、もっとも単純なコンパイルからプログラム実行までの手順ですが、gfortran コマンドにファイル名を与えただけの命令では、どんなプログラムをコンパイルしても a.out という同じ名前の実行形式ファイルになってしまうので不便です。また、Fortran プログラムは計算速度が重要

³⁴ コンパイラである gfortran 自体はリンクする機能を持っていませんが、gfortran 内部から OS に付属しているリンク用アプリケーション (リンカ) を呼び出すことができるので、リンクも実行することが可能なのです。

³⁵ “./” は “現在のディレクトリにあるファイル” という意味です。このあたりの詳細は、Linux などの解説書を見て下さい。

なので、最適化をしてできるだけ高速に処理する方が良いでしょう。さらに 1.2.2 節で述べたように数値計算をするときは倍精度実数を使うべきなので、自動倍精度化オプションも必要です。

このため、gfortran でプログラムをコンパイル・リンクする時は、以下のようなオプションを付けることをお勧めします。

```
gfortran -O -fdefault-real-8 test1.f90 -o xtest1
```

-O が最適化のオプション、-fdefault-real-8 が自動倍精度化のオプションです。また、-o のオプションの次の文字列 (ここでは xtest1) は実行形式ファイル名指定です。大文字の-O と小文字の-o を間違わないようにして下さい。

この場合、コンパイルとリンクが正常に終了すると、-o オプションで指定した “xtest1” という名の実行形式ファイルが作成されます。このため、プログラムを実行するには、

```
./xtest1
```

と入力することになります。

また、付録 F で紹介している数値計算ライブラリやグラフィックライブラリを使用したプログラムをコンパイル・リンクする場合は、それらのライブラリファイルをオプションとして付加する必要があります。例えば、FFTW を含んだプログラムの場合は、

```
gfortran -O -fdefault-real-8 test1.f90 -o xtest1 -L/usr/local/lib -lfftw3
```

のように入力します。ここで、-L オプションが指定している “/usr/local/lib” は、FFTW ライブラリ (libfftw3.a) の入ったディレクトリを示すオプションなので、使っている環境に応じて書き換える必要があります。

付録 B エラー・バグへの対処法

プログラムを開発する際の最大の問題は“エラー (Error)”の発生です。コンパイル・リンク・実行という一連の過程の中でそれぞれエラーが発生する可能性があります。エラーはプログラム開発にとって付き物であり、避けては通れません。エラーの出ないように慎重にプログラムを書くことはもちろんですが、出たら出たでそれらに如何に素早く対処できるかが腕の見せ所です。ここではエラーをまとめて、それぞれの対処法を示します。なお、具体的なエラーメッセージは gfortran の出力を利用していますが、他のコンパイラでも形式が違うだけで出力情報の内容はあまり変わらないと思います。

(1) コンパイルエラー

コンパイルとは、作成したプログラムを文法に従って機械語に翻訳することです。このため、“コンパイルエラー”とは文法的な間違いのことです。プログラムに文法的間違いがあれば、コンパイラがエラーメッセージを出力して終了します³⁶。コンパイラのエラーメッセージ形式はコンパイラによって異なりますが、gfortran では次のような形式で出力されます。この例は、test1.f90 というプログラムファイルをコンパイルしたときのエラー出力です。

```
test1.f90:15.7:  
  do i = 1, 100  
    1  
Error: Symbol 'i' at (1) has no IMPLICIT type
```

以下に、この出力の読み方を示します。

- (1) 1行目はエラーのあるプログラムファイル名(この例では、test1.f90)とエラーの位置
この例では、エラーの位置が15.7となっていますが、15はプログラムの行番号、7は左から数えた文字の位置を示す数値です。
- (2) 2行目はエラーのあるプログラム文のコピー
- (3) 3行目は“1”を上向き矢印のように使って、2行目の文のエラーが起きている位置を指定
- (4) 4行目は3行目が示した部分の具体的なエラーの説明
このメッセージを頼りにエラーを見つけてプログラムを修正します。

この例の4行目のメッセージは「3行目の1が示している変数*i*は、暗黙の型がない」という意味です。これは、変数*i*が宣言されていないことが原因です。

コンパイルエラーは、エラーメッセージに従って修正すればいいので、それほどの手間ではありません。ただし、ある文に文法エラーがあると、その文が存在しない状態で引き続きコンパイルを行うため、その波及効果で下方のプログラムがエラーになることがあります。例えば宣言文に誤りがあると、その宣言文で宣言している変数がコンパイラにとっては宣言されていないことになり、その変数を使っている文が全てエラーになります。エラーメッセージが多い時は、一度に全部チェックしないで、ある程度修正したら再度コンパイルして、エラーが出るかどうかを確認した方が良いでしょう。

³⁶コンパイラが出すメッセージには“Warning”(警告)と“Error”(エラー)があります。Warningは「このままでも実行できるけど、ひょっとすると予期しない結果が起こる可能性があります」という意味なので、必ずしも修正する必要はありません。このため、メッセージがWarningだけのときは強制終了せずにリンクの作業に移ります。しかし、文法的には間違いがなくても、作成者の意図とは違ったプログラムになっている可能性があるため、念のためにプログラムを確認した方が良いでしょう。

(2) リンクエラー

リンクとは、別々に翻訳されたルーチンを結合して実行形式ファイルにする作業のことです。よって、リンクエラーは用意されていないサブルーチンや関数を使ったときに起こります。例えば、

```
program test1
  implicit none
  real x,y
  x = 5
  y = 100
  call subr(x,y)      ! subr が用意されていないとリンクエラーになる
  print *,x,y
end program test1
```

のようなプログラムを作って、これだけを単独でコンパイル・リンクすると

```
/tmp/ccy3lcd.o: In function 'MAIN_':
test1.f90:(.text+0x6d): undefined reference to 'subr_'
collect2: ld returned 1 exit status
```

のようなメッセージが出力されます。以下に、この出力の読み方を示します。

- (1) 1行目はエラーが起こっているルーチンの表示
この例では“MAIN”，すなわちメインプログラム中で起こっていることが示されています。
- (2) 2行目はエラーの原因
このメッセージを頼りにエラーを見つけてプログラムを修正します。

この例の2行目のメッセージには「subr という名への参照が定義されていない」とあります。これは、メインプログラム中でコールしている subr という名のサブルーチンが用意されていないのが原因です³⁷。

標準の組み込み関数が用意されていないことはまずないので、リンクエラーのほとんどはプログラム中で呼び出したサブルーチン名や関数名の書き間違いが原因です。リンクエラーのメッセージにはコンパイルエラーのようなエラー行の表示はありませんが、エラーメッセージの中に見つからなかったサブルーチンや関数の名前が表示されているので、エディタの検索機能を使えば、見つけるのはさほど困難ではありません。

(3) 実行時エラー

一番厄介なのが実行時のエラーです(こういうプログラムは“バグ(虫)”があるといいます)。なぜなら、コンパイルエラーやリンクエラーのようなエラーの場所を特定する情報が出力されないのです。どこでエラーが起こっているのかを探す作業が大変だからです。実行時エラーを探すには、プログラムを詳細にチェックするしかありません(この作業を「バグを取る」という意味でデバッグするといいます)。

それでも、答えが合っているか否かは別として、プログラムがなんとか終了すれば良いのですが、バグによっては実行した後でうんともすんともいわなくなってしまう場合があります。これは無限ループに入ってしまったか、暴走したかのどちらかだと考えられます。通常、このような状態で実行を強制的に終了させるには、コントロールキーを押しながらCキーを押します。まず間違いなく止まります。

もう一つ、以下のようなメッセージを出して強制的に終了する場合があります。

```
Segmentation fault
```

これはプログラムではなく OS が出力しているメッセージで、主として実行中のプログラムがそれ以外の実行中のプログラムのメモリ領域を破壊しようとした時に起こります。例えば、次のように配列宣

³⁷メッセージ中では MAIN_ とか、 subr_ のように名前後ろにアンダースコアが付いていますが、これはコンパイラが自動的に付加するものなので無視して下さい。リンクは gfortran がするのではなく、ld というリンカがするのですが、リンカは OS によって異なるので、gfortran を使っていてもこの形式でエラーメッセージが出るとは限りません。

言の範囲外の要素に値を代入しようとするとき起こることがあります³⁸.

```
real a1(10)
integer i
do i = 1, 100000
  a1(i) = i
enddo
```

この例では、10個しか用意されていない配列に、100000番目まで値を代入しようとしているのですから、どこのメモリに値を代入しようとしているのか不明なのが問題なのです。

また、サブルーチンの引数に型の合っていない数値を与えたり、戻り値を与える引数に定数を与える、などの不注意なサブルーチンコールでも Segmentation fault が起こる可能性があります。Segmentation fault は、実行時に問題が出た段階で起こるので、適当に print 文を入れて、どこまでが出力されて、どこからが出力されないかを確認することでエラーの起こっている場所を特定することができます。

しかし、このような実行時エラーはコンピュータの動作が正常でないという形で表面化するので、まだましです。原因を特定するのに時間がかかるかもしれませんが、修正しなければ動かないのだから実害はありません。実を言うと、一番やっかいなエラーとは、ちゃんと実行してはいるのだけど、計算結果が何となくおかしい、というものです。

筆者は何年も Fortran のプログラムを書いてきましたが、実行時のエラーほど見つけにくいものはありません。それでも見つかれば良い方で、場合によってはバグの存在を知らずに結果を出し、学会直前に気がついた、なんてこともありました。大規模なシミュレーションプログラムを作るとき、長いプログラムを書くのに時間がかかるのはもちろんですが、取りあえず動いたプログラムの計算結果が正しいかどうかを確認する作業にもかなりの時間が必要です。本書で紹介した“エラーの出にくいプログラムの書き方”は、この確認作業にかかる時間を少しでも減らすためのものです。しかし、最後に必要なのは“地道な努力と忍耐”なのです。

³⁸実際にはこの程度の簡単なプログラムでは起こるとは限らないのですが、だからといってこんなプログラムを書いてはいけません。

付録 C 自動倍精度化オプション

コンピュータシミュレーションでは大量の数値を使って複雑な計算を繰り返し、必要に応じてそれらを集積する作業を行います。このとき問題となるのが、1.4.3節で述べた“桁落ち”です。桁落ちを減らすための工夫をいくつか紹介しましたが、究極的には演算精度を上げるしかありません。

Fortran で取り扱える実数には、単精度実数型と倍精度実数型がありますが、数値計算をするには倍精度実数型を使うべきです。倍精度実数計算は単精度実数計算と比べてそれほど実行時間はかかりません。これは最近の CPU が倍精度実数計算用のハードウェアを装備しているからです。

Fortran の仕様では、デフォルトの実数型が単精度なので、`real` 宣言をした実数は単精度実数型です。このため、基本仕様のままで倍精度実数計算をするには変数宣言をする時に `real*8` や `real(8)` などの 8byte 指定が必要です³⁹。

例えば、デフォルト仕様のままでも、

```
real*8 x,a1(10)
```

と宣言すれば、変数 `x` や配列 `a1` は倍精度になります。しかし、宣言文を変更するだけでは不完全です。1.0 や 1.23e5 などの実定数もデフォルト仕様では単精度なので、倍精度実定数に書き換える必要があります。

そもそも倍精度で計算するのを基本にするのに、常に倍精度を意識した書式を利用しなければならないのは、面倒だけでなく間違いに気がつかないまま精度の悪い計算をしている可能性もあります。そこで最近の Fortran コンパイラのほとんどが自動倍精度化オプションを装備していることを幸いに、本書では単精度実数型と倍精度実数型を使い分ける文法の説明は省略しました³⁹。自動倍精度化オプションを付けてコンパイルすれば、`real` だけの宣言文で倍精度実数変数を宣言できるし、1.0 や 1.23e5 のような定数はそのまま倍精度実定数を意味します。

表 C.1 に筆者が使っている Fortran コンパイラの自動倍精度化オプションを示します。この他のコンパイラについてはそれぞれのマニュアルをチェックしてもらえば、大抵はあると思います。

表 C.1 コンパイラに応じた自動倍精度化オプション

コンピュータ	コンパイラ	ベンダー	自動倍精度化オプション
パソコン	<code>gfortran</code>	GNU	<code>-fdefault-real-8</code>
パソコン	<code>ifort</code>	Intel	<code>-r8</code>
パソコン	<code>f95</code>	Absoft	<code>-N113</code>
スパコン	<code>sxf90</code>	NEC	<code>-Wf,-A dbl4</code>

自動倍精度化は、コンパイルする時にオプションを加えれば良いだけなので、プログラムを変更することから考えれば楽なものです。ぜひ利用して下さい。

³⁹精度指定の詳細は付録 D を参照。

付録 D 数値型の精度指定

1.2.2 節で、基本的な数値型には整数型と実数型があり、実数型には単精度実数型と倍精度実数型があるという話をしました。Fortran のデフォルト実数型は単精度ですが、コンパイラの自動倍精度化機能(付録 C)を使えばデフォルトを倍精度実数型に変更できるので、本書では 2 種類の実数型を区別せず、単に“実数型”と表現しています。しかし、大量の実数型データをバイナリ形式で保存するときには、保存容量を圧縮するために精度を落とすことが考えられますし、“4 倍精度実数”という倍精度よりも有効桁数の多い実数型を使って、より高精度の計算をすることも可能です⁴⁰。ここでは、定数や変数の精度を変更する書式について説明します。

変数の精度を指定するには、以下の 3 種類の方法があります。ここでは、基本的な宣言文だけを紹介しますが、3.9.1 節で述べた属性や数値代入を含む拡張宣言も可能です。

```
型指定*精度数 変数 1, 変数 2, ...
型指定(精度数) 変数 1, 変数 2, ...
型指定(kind=精度数) 変数 1, 変数 2, ...
```

ここで、“精度数”のところには、数値型の byte 数を整数で指定します。実数型の場合、単精度なら 4、倍精度なら 8、4 倍精度なら 16 です。例えば、単精度実数型変数を宣言するには、

```
real*4 xs1,ys1,as1(100)
real(4) xs2,ys2,as2(100)
real(kind=4) xs3,ys3,ds3(100,100)
```

などと書きます。また、通常の整数型は 4byte ですが、最近では“倍精度整数”という 8byte の整数型(使用可能な範囲は $-2^{63} \sim 2^{63} - 1$)を使うことができるので、これを利用するときには、

```
integer*8 n81,m81,k81(100)
integer(8) n82,m82,k82(100)
integer(kind=8) n83,m83,ka83(100,100)
```

などと書きます。

3 種類の書式の中でどれが良いかは一概には言えません。一番目の“*”を使った書式は、古い Fortran から使われているので、コンパイラが古い場合や、昔のプログラムを継承する可能性がある場合には覚えておいた方が良いでしょう。しかし、これからプログラムを書き始める場合には、二番目か、精度数の意味を明記した三番目の方が良いでしょう。これらは、次のように `parameter` 変数(3.9.2 節)を使って精度数を指定することができます。

```
integer, parameter :: kp=16
real(kp) xq2,yq2,aq2(100)
real(kind=kp) xq3,yq3,dq3(100,100)
```

これに対し、“*”を使った書式では必ず数字を使って指定しなければなりません。`parameter` 変数で精度指定をしておけば、計算機環境に応じて精度を変更する必要があるときに便利です。

ただし、高精度の計算をする場合は、定数も高精度にしなければなりません。例えば、“1.23”と書けばデフォルトの実数型になるので、本書の暗黙指定では倍精度実数型です。よって、このまま 4 倍精度の計算に使うと精度が落ちてしまいます。 $A \times 10^B$ で表される実数を 4 倍精度で表現するときは“ AqB ”と書きます。このため、“1.23”は“1.23q0”と書かなければなりません⁴¹。

しかし、`parameter` 変数で変数の精度を指定しているときに、`e` や `q` のような文字を使って定数の精度を指定していると、精度変更の時に定数の変更が別途必要になります。そこで、数値の後にアンダースコ

⁴⁰ただし、4 倍精度実数型は、必ず実装しなければならない規格ではないようで、コンパイラによっては使えない場合があります。gfortran でも、バージョン 4.8 では使えますが、4.2 では使えませんでした。

⁴¹ちなみに、デフォルト実数が単精度の時、倍精度実定数は“ AdB ”と書きます。例えば、“1.2”は“1.2d0”です。

ア “_”と精度数を付けて定数の精度を指定することができます。例えば、倍精度の“1.23”は“1.23_8”と書け、4倍精度の“1.23q0”は、“1.23_16”と書くことができます。この精度指定には parameter 変数を使うことができるので、次のように書くことができます。

```
integer, parameter :: kp=16
real(kp) xx,yy
xx = 1.23_kp
yy = 1.2345e-15_kp*xx**3
```

この例の yy に代入している実数の指数指定は e を使っていますが、精度数が 16 なので、4倍精度定数になります。

付録 E Big Endian と Little Endian

大量のデータを精度を落とさずにファイルに保存するときは、4.5 節で説明した書式なし `write` 文を使ってバイナリ形式で保存する方が良いと思います。これは計算機のメモリに保存された内部形式そのままファイルに保存するため、書式付き `write` 文を使ったテキスト形式出力よりも出力データ量が小さくなるからです。

しかし、バイナリ形式で保存するときは注意が必要です。テキスト形式で保存したファイルは、我々でも直接読むことができるのですから、どんなコンピュータでも同じ様に読むことができます。このため、大型計算機で計算した結果をテキスト形式で出力しておけば、そのままパソコンのプログラムから読み込んで処理をすることができます。しかしバイナリ形式で出力すると必ずしもうまくいきません。バイナリ形式は計算機が直接計算を実行するためのものですから、計算機のアーキテクチャによって違っている可能性があるからです。しかし、現在はバイナリ形式も標準化されていて、ほとんどの計算機で整数型も実数型も統一した形式が採用されています。このため、バイナリ形式を使って大型計算機が出力したデータをパソコンのプログラムで処理することができます。

ただやはりいくつかの制約は残っています。その一つが Endian 問題です。コンピュータが扱う数値の内部形式は整数型で 4byte、倍精度実数型で 8byte ですが、この byte の並びが上位の桁からのコンピュータと、下位の桁からのコンピュータがあるのです。前者を Big Endian、後者を Little Endian といいます。イメージ的に表現すれば、1234 という数字を保存すると、Big Endian では 01234 の順番で保存するのにに対し、Little Endian では 43210 の順番で保存するようなものです。これらの違いは CPU のアーキテクチャによるものなのでユーザーが変更することはできません。

最近のほとんどのパソコンは、CPU に Intel かその互換チップを使っていますが、これらは Little Endian です。これに対し、以前の Macintosh で使われていた Motorola の CPU や NEC のスパコンは Big Endian です。このため、スパコンの Fortran で計算して書式なし `write` 文で出力すると、そのままではパソコンの Fortran で読むことができません。

しかし、要はデータ入出力時の形式の問題だけなので、最近の Fortran コンパイラの多くはコンパイラオプションや環境変数などで Big から Little へ、あるいはその逆への変換をしてから入力または出力をすることができるようになっていきます。例えば、NEC SX の場合にはプログラム実行前に環境変数 `F_UFMTENDIAN` を指定することで変換できます。

```
setenv F_UFMTENDIAN 10,20      # C シェル系
```

環境変数で指定するのは書式なし `write` 文の装置番号です。この例の場合、`write(10)` と `write(20)` が Little Endian に変換して出力されます。この変換を使うと、メモリ上で Big な形式を Little にして出力するので若干計算時間にロスが出ます。しかし、byte 並びを逆にするだけですからテキスト変換よりは楽です。

これに対し、入力側で変換するやり方もあります。例えば、`gfortran` の場合には `-fconvert=big-endian` というオプションを加えれば、書式なし `read` 文で入力するときに Big Endian から Little Endian に変換してくれます。最近の Fortran コンパイラならば大抵変換機能が付いているようなので、マニュアルで確認すれば良いでしょう。

筆者はどちらかといえばスパコン側で変換する方が良いと思います。スパコンで行う計算は元々時間がかかるものであり、それに比べれば出力時のデータ変換による時間増加はわずかです。これに対してパソコン側で変換するのは時間もかかるし、変換するかどうかを問題に応じて使い分けるのも面倒です。

ただし、スパコンでの計算には時間制限があるので、データ変換のために CPU 時間を削るのはもったいないとも考えられます。スパコンはデータ変換に時間がかかるという話もありますので、どちらが良いかはスパコン管理者に確認した方が良いでしょう。

付録 F 数値計算を補助するフリーのサブルーチンライブラリ

複雑な計算アルゴリズムを使ったプログラムを一から全て書くのは結構大変です。汎用性のあるアルゴリズムを使ったプログラムは、インターネット上にサブルーチンライブラリとして公開されているものがあるので、これらを利用してプログラムを作るのもスマートな方法です。ここでは筆者が使ったことのあるライブラリの中からお勧めをいくつか紹介します。

F.1 高速フーリエ変換

偏微分方程式の解法やデータ解析で良く使う高速フーリエ変換 (Fast Fourier Transform, FFT) は、アルゴリズムが複雑で、効率の良いプログラムを自作するのは大変です。筆者はもっとも簡単なレベルのプログラムを作ったことはありますが、最近はネットで公開されているものを使っています。お勧めなのは、FFTW (Fastest Fourier Transform in the West) です。

```
http://www.fftw.org/
```

このライブラリには、Fortran から使用できるサブルーチンが入っています。

F.2 多倍長計算

Fortran での倍精度実数は 8byte で、有効数字は 15 桁程度です。4 倍精度実数は 16byte で、有効数字が 33 桁程度に増加しますが、それ以上は無理だし、コンパイラによっては 4 倍精度実数が使えないこともあります。筆者は多項式の根を計算するときに高精度計算が必要になったことがあります。数十次の多項式を計算するときは、最大次数の項と最小次数の項との数値の差が非常に大きくて、桁落ちが深刻な問題になるからです。

このときに便利なのが多倍長計算ライブラリです。これは、任意の精度で実数計算をすることができるものです。通常の計算より処理に時間がかかりますが、精度優先のときにはやむを得ません。ネット上には何種類か公開されていますが、筆者が使って便利だったのは、FMLIB です。

```
http://myweb.lmu.edu/dmsmith/FMLIB.html
```

これは Fortran のモジュールになっていて、四則演算を “+−*/” の記号で書けるし、sin や cos などの関数も充実しています。本書では説明していない構造体の知識が少しだけ必要ですが、変数宣言程度なのでそれほど難しくはありません。筆者は、倍精度計算用に作った多項式の根を計算するプログラムを、それほど大きな変更をすることなく多倍長で動作させることができました。

F.3 長周期乱数発生

計算機シミュレーションには、時間発展の方程式を解くことで未来を予測する決定論的手法と、サイコロを振って確率的に将来の可能性を探る確率論的手法があります。後者の確率論的手法で、サイコロに相当する働きをするのが乱数です。Fortran には、乱数発生用の組み込みサブルーチン `random_number` が用意されているので、

```
call random_number(x)
```

のように書けば、乱数 `x` を得ることができます (問題 (3-10) 参照)。しかし、乱数は周期長が問題で、周期の短い乱数を使うと良いシミュレーション結果が得られません。`random_number` では周期長が不足するときにお勧めなのは、周期が非常に長い乱数として有名な、Mersenne Twister です。これを考案した松本真さんが以下のサイトでプログラムを公開されています。

<http://www.math.sci.hiroshima-u.ac.jp/~m-mat/MT/mt.html>

松本さんが公開されているのはC言語で書かれたものですが、リンク集の中に Fortran 用の公開先が入っています。

F.4 数値計算ライブラリ

連立方程式の解法や数値積分など、一般的な数値計算ライブラリがあれば、数値計算のプログラムを作るのが楽になります。ただ、残念ながら筆者はフリーの良いライブラリを知りません。ネットで公開されていて使えそうなものに、GSL (GNU Scientific Library) があります。

<http://www.gnu.org/software/gsl/>

ただし、基本的にC言語用なので、Fortran で使うには若干テクニックがいるようです。また、ちょっと使ってみました。必ずしも速くありません。ベッセル関数などの特殊関数は、その高速バージョンを公開しているサイトからダウンロードした方が良いと思います。筆者は、数値計算の教科書を読んで原理から自作するか、参考文献 [5,6] などに掲載されているプログラムをコピーして使うことが多いです。

F.5 グラフィックライブラリ

計算機シミュレーションは物理的に起こっている現象を模擬するものですが、出力される数値だけで現象を読み取るのは容易ではありません。そこで必要不可欠なのが可視化ソフトです。フリーの可視化ソフトとして有名なものに gnuplot があります。

<http://www.gnuplot.info/>

プログラムの結果をファイルに保存しておけば、gnuplot からそれを読み込んで、様々なグラフに加工して表示することができます。

最近のパソコンは基本操作が全てビジュアルであり、グラフィック出力機能は OS に組み込まれています。Fortran にはグラフィック出力に関する文法はありませんが、他の言語で書かれたグラフィックライブラリとリンクすれば、Fortran プログラムからサブルーチンをコールする形式で図を描くことが可能です。その一つに、筆者が作った Frames があります。

<http://www.pp.teen.setsunan.ac.jp/frames/>

Frames は、Fortran で使うためのグラフィックツールとして開発してきたもので、本書で説明した知識だけで簡単に2次元や3次元のグラフを描くことができます。また、描画した図形をイメージやアニメーション形式で出力してプレゼンテーションに使うことも可能です。良かったら試してみてください。

参考文献

- [1] “入門 Fortran90 実践プログラミング”, 東田幸樹・山本芳人・熊沢友信, ソフトバンク, 1994.
- [2] “数値計算の常識”, 伊理正夫・藤野和建, 共立出版, 1985.
- [3] “コンピュータシミュレーションの基礎”, 岡崎 進, 化学同人, 2000.
- [4] “CIP 法”, 矢部 孝・内海隆行・尾形陽一, 森北出版, 2003.
- [5] “Numerical Recipes in Fortran 77, Second Edition”, W. H. Press, S. A. Teukolsky,
W. T. Vetterling, B. P. Flannery, Cambridge University Press, 1996.
- [6] “Numerical Recipes in Fortran 90”, W. H. Press, S. A. Teukolsky, W. T. Vetterling,
B. P. Flannery, Cambridge University Press, 1996.